

Menyelesaikan persoalan Matematika dan Fisika
tingkat SMU
dengan Maple

Surjo Abadi¹

13 Juni 2009

¹Dosen FT UKI

Daftar Isi

1	Berkenalan dengan Maple	4
1.1	Apa itu Maple	4
1.2	Memulai Maple	5
1.3	Kaidah Penulisan Maple	6
1.4	Menggunakan Paket - Paket Maple	7
1.5	Memakai Fasilitas Help	8
1.6	Menyimpan Lembar Kerja	9
2	Operasi Dasar Matematika dengan Maple	10
2.1	Operasi Aritmatika	10
2.2	Fungsi Trigonometri	15
2.3	Logaritma	16
2.4	Exponensial	17
2.5	Bilangan Kompleks	18
2.6	Bilangan Prima	19
2.7	Memanggil hasil Perhitungan sebelumnya	19
2.8	Soal soal dan penyelesaiannya	20
2.9	Soal Latihan	21
3	Menyelesaikan Aljabar dengan Maple	23
3.1	Membuat Variabel	23
3.2	Urutan operasi hitung	25
3.3	Perintah subs	26
3.4	Perintah expand	28
3.5	Perintah factor	29
3.6	Perintah simplify	30
3.7	Perintah combine	31
3.8	Identitas Trigonometri	32
3.9	Membuat fungsi	33
3.9.1	Fungsi Single Variabel	34
3.9.2	Fungsi Lebih dari Satu Variabel	35
3.9.3	Fungsi Komposisi	35
3.10	Soal soal dan penyelesaiannya	37
3.11	Soal soal latihan	41

4	Penyelesaian Persamaan dan Pertidaksamaan	42
4.1	Perintah solve	42
4.2	Perintah fsolve	45
4.3	Menyelesaikan Pertidaksamaan	48
4.4	Soal - Soal dan Penyelesaiannya	49
4.5	Soal soal latihan	52
5	Membuat Grafik fungsi	53
5.1	Grafik Dimensi Dua	53
5.1.1	Fasilitas Tambahan Grafik pada Maple	54
5.1.2	Grafik Fungsi Parametrik	56
5.1.3	Fungsi yang Mempunyai Nilai Singular	57
5.1.4	Menampilkan lebih dari satu grafik dalam satu gambar	58
5.1.5	Mengekspor gambar	61
5.1.6	Koordinat Polar	61
5.2	Grafik Dimensi Tiga	66
5.3	Membuat Animasi	67
5.3.1	Melihat bagaimana sebuah kurva tergambar	69
5.3.2	Mengekspor animasi	70
5.4	Soal - soal dan penyelesaiannya	70
5.5	Soal - soal latihan	75
6	Kalkulus	77
6.1	Limit Fungsi	77
6.2	Hitung Diferensial	80
6.2.1	Gradien garis	81
6.2.2	Menentukan harga maksimum dan minimum	82
6.2.3	Turunan kedua	85
6.3	Hitung Integral	86
6.3.1	Integral tak tentu	87
6.3.2	Integral tertentu	88
6.3.3	Luas Area	89
6.4	Paket Student	93
6.4.1	Volume benda putar	94
6.4.2	Volume benda putar yang dibatasi dua fungsi	97
6.4.3	Panjang Kurva	98
6.5	Soal - Soal dan Penyelesaiannya	100
6.6	Soal - Soal Latihan	106
7	Matrik dan Vektor	109
7.1	Matrik	109
7.1.1	Definisi	109
7.1.2	Membangun Matrik	110
7.1.3	Determinan suatu Matrik	111
7.1.4	Inverse suatu Matrik	111
7.1.5	Mentranspose matrik	113
7.1.6	Penjumlahan dan Pengurangan Matrik	114
7.1.7	Perkalian Matrik	115

7.2	Penyelesaian sistem persamaan linear	117
7.3	Program Linear	118
7.3.1	Menggambarkan daerah pertidaksamaan	118
7.3.2	Paket Simplex	122
7.4	Vektor	125
7.4.1	Memvisualkan Vektor	126
7.4.2	Aljabar Vektor	130
7.4.3	Perkalian Vektor	133
7.4.4	Sudut antara dua Vektor	134
7.5	Soal - soal dan penyelesaiannya	135
7.6	Soal - soal latihan	140
8	Aplikasi pada Fisika	142
8.1	Kasus Gerak Lurus	142
8.2	Gerak Vertikal	145
8.3	Gerak Parabola	148
8.4	Hukum Newton kedua	150
8.5	Katrol tetap	152

Bab 1

Berkenalan dengan Maple

1.1 Apa itu Maple

Maple adalah satu perangkat lunak (*software*) yang dapat memecahkan persoalan Matematika dengan komputer. Program Maple pertama kali diluncurkan tahun 1981 oleh satu perusahaan yang bernama *Waterloo Maple, Inc.* Keunggulan dari Maple adalah, program ini dapat menyelesaikan persoalan matematika baik secara simbolik, numerik, grafik, maupun kombinasinya. Sejarah Maple sendiri telah menunjukkan bahwa program ini banyak digunakan oleh: *scientist, engineer*, para pendidik baik guru maupun dosen, siswa, mahasiswa, dan juga oleh para peneliti (*researcher*). Bagi pembaca yang ingin mengetahui lebih lanjut mengenai apa dan bagaimana Maple itu, dapat mengunjungi web-site Maple pada :

<http://www.maplesoft.com/>.

Menggunakan Program Maple, tidak ubahnya seperti menggunakan sebuah kalkulator, hanya saja fungsi - fungsi yang ditawarkan oleh Maple jauh lebih lengkap ketimbang sebuah kalkulator. Sebagai ilustrasi, jika kita ingin menghitung dengan kalkulator, hasil perkalian 3×4 , maka pada kalkulator tersebut kita tekan tombol 3, tanda kali (\times), tombol 4, dan terakhir tombol sama dengan (=).



Kalkulator akan menampilkan hasil perkalian tersebut yaitu **12**. Untuk menghitung operasi perkalian tersebut dengan Maple, maka perintah yang kita tuliskan pada lembar kerja Maple adalah :

```
[> 3 * 4;
```

kemudian kita tekan tombol *enter*, dan Maple akan menampilkan hasilnya :

12

Keunggulan lain dari Program Maple adalah dalam hal pembuatan grafik. Fasilitas pembuatan grafik yang dipunyai Maple sangat membantu sekali bagi guru - guru matematika untuk

memperagakan grafik - grafik yang sulit dibuat secara manual (grafik tiga dimensi, misalnya). Sebagai contoh, untuk membuat grafik dari dua fungsi $y = \sin(x)$, dan $y = \cos(x)$, dengan batas dari 0 sampai 2π , secara bersamaan, pada lembar kerja Maple kita tuliskan perintah :

```
[>plot([sin(x),cos(x)],x=0..2*Pi);
```


Hasilnya seperti yang ditampilkan pada gambar(1.1a). Program Maple dapat juga membantu kita dalam membuat grafik tiga dimensi, secara mudah. Apabila kita hendak menggambar sebuah fungsi dengan dua variabel $z = f(x, y)$,

$$z = x.e^{-x^2-y^2},$$

dengan batas x dari -2 sampai 2 , dan y dari -2 sampai 2 . Pada lembar kerja Maple kita masukkan perintah seperti :

```
[> plot3d(x*exp(-x^2-y^2), x=-2..2, y=-2..2, color=x);
```

Hasil dari plot ini, yaitu grafik tiga dimensi diperlihatkan pada gambar(1.1b).




Gambar 1.1: Panel (a) memperlihatkan gambar $y = \sin(x)$ dan $y = \cos(x)$ dalam satu gambar. Panel (b) adalah grafik tiga dimensi dari fungsi $z = x.e^{-x^2-y^2}$.

Pada Bab selanjutnya akan dibahas secara rinci, bagaimana Maple dapat digunakan untuk memecahkan persoalan Matematika, seperti: Aljabar, Trigonometri, Calculus lengkap dengan pembuatan grafiknya.

1.2 Memulai Maple

Dalam subbab ini, kita akan belajar beberapa prinsip dasar Maple. Yang akan dibahas disini adalah beberapa perintah dasar Maple. Diasumsikan bahwa program Maple telah terinstalasi dengan

baik pada komputer pembaca. Program Maple mulai dari versi 10 keatas memberikan kepada pemakai dua opsi tampilan lembar kerja (*worksheet*) yakni; tampilan klasik, dan tampilan bukan klasik. Bentuk lembar kerja dari kedua tampilan tersebut dapat dilihat pada gambar 1.2 dan gambar ???. Walaupun tampilan lembar kerja antara klasik dan non-klasik berbeda, namun prinsip kerja keduanya sama.



Gambar 1.2: Tampilan lembar kerja Maple klasik (a), dan tampilan lembar kerja yang bukan klasik (b).

Jika komputer yang kita pakai menggunakan Windows XP sebagai sistem operasinya, untuk menjalankan program Maple kita bisa mulai dengan mengklik 'Start', kemudian 'Program', lalu cari 'Maple 11', atau versi lainnya, kemudian klik *Classic Worksheet Maple*, untuk menampilkan Maple dengan tampilan klasik. Cara yang paling mudah untuk menjalankan Maple, adalah dengan menggunakan klik ganda tepat di-ikon Maple pada layar monitor komputer. Perintah ini akan membawa kita kepada lembar kerja Maple, seperti yang telah diperlihatkan dalam gambar (1.2). Pada lembar kerja inilah perintah - perintah Maple dimasukkan. Kembali pada gambar (1.2) baik (a) maupun (b), perhatikan pada samping kiri atas lembar kerja, kita akan melihat kursor yang berbentuk:

[> .

Jika sudah berhasil memunculkan kursor Maple seperti diatas, maka kita sudah berhasil membuka program Maple. Untuk selanjutnya tepat disebelah kanan kursor tersebut perintah - perintah Maple dijalankan.

1.3 Kaidah Penulisan Maple

Sebelum kita memulai Maple, ada baiknya kita mengetahui dahulu kaedah penulisan (*syntax*) yang dimengerti oleh Maple. Beberapa hal yang harus selalu diingat untuk memasukkan perintah pada lembar kerja (*worksheet*) Maple adalah :

1. Fungsi fungsi terpasang (*built-in function*) yang disediakan oleh Maple, harus selalu diikuti oleh tanda kurung biasa (*..*). Operasi matematik yang mempunyai group seperti pada op-

erasi vektor dan matriks digunakan tanda kurung siku [...].

Contoh : Maple tidak akan menjalankan fungsi seperti dibawah ini,

$\sin \pi$.

Perintah baru bisa dimengerti oleh Maple jika kita menambahkan tanda kurung biasa setelah fungsi sinus tersebut, seperti:

$\sin(\pi)$.

2. Setiap perintah pada program Maple akan dieksekusi jika diakhiri dengan tanda semicolon (;) atau colon (:). Tanda ini harus diletakkan pada setiap akhir perintah, sebelum tombol *enter* ditekan. Jika tanda semicolon digunakan, maka Maple akan menghitung sekaligus menampilkan hasil perhitungan. namun Maple tidak akan menampilkan hasil hitungan jika tanda colon yang digunakan, walaupun Maple memproses perhitungan yang kita berikan. Maple tidak akan memproses perintah kita jika tanda colon (:) atau semi colon (;) tidak dituliskan. Jika itu yang terjadi, maka Maple akan memberikan peringatan sebagai berikut :

`Warning, premature end of input.`

Beberapa contoh :

`[> 2 + 3;`

5.

Jika kita menggunakan tanda colon maka :

`[> 2 + 3:`

Walau Maple memproses hitungan diatas, namun Maple tidak akan menampilkan hasil hitungan di atas.

1.4 Menggunakan Paket - Paket Maple

Pada saat kita memulai dengan Maple, yang perlu kita perhatikan adalah bahwa tidak semua fungsi dari Maple dapat kita gunakan secara langsung. Beberapa fungsi Maple disimpan dalam paket - paket tertentu. Pengguna Maple yang telah berpengalaman bahkan dapat membuat paket - paket sendiri sesuai dengan kebutuhannya. Fungsi - fungsi yang tersimpan dalam paket tersendiri ini biasanya fungsi - fungsi untuk tingkat lanjut. Untuk menjalankan fungsi - fungsi ini, terlebih dahulu kita harus memanggil paket-nya. Perintah untuk memanggil paketnya adalah **with**. Bentuk penulisan perintah **with** adalah :

`with(nama paket).`

Topik bahasan buku ini, tidak akan membahas secara detail penggunaan fungsi - fungsi lanjut. Buku ini akan lebih banyak membahas fungsi - fungsi dasar pada Maple.


1.5 Memakai Fasilitas Help

Fasilitas **Help** yang diberikan oleh Maple, memudahkan kita apabila kita menemukan kesulitan atau terlupa dengan kaidah penulisan perintah tertentu. Hal yang paling sering terjadi adalah; setelah kita selesai menuliskan suatu perintah dan menjalankannya, ternyata Maple tidak dapat mengeksekusi perintah yang kita berikan dan menampilkan pesan **error**. Dengan menggunakan fasilitas **Help** kita akan dengan mudah mencari kesalahan - kesalahan tersebut, untuk selanjutnya diperbaiki.

Cara menggunakan menu **Help** adalah dengan mengklik menu **Help**, selanjutnya kita klik **Topik Search**. Kemudian kita masukkan kata kunci dari topik yang akan kita cari. Setelah mencari dalam *library*-nya, Maple akan memberikan informasi yang lengkap dari topik yang kita cari. Informasi yang akan diberikan oleh Maple adalah :

1. Tata cara penulisan perintah tersebut;
2. Contoh penggunaan perintah tersebut.

Dengan kedua informasi tersebut, kita akan mudah bekerja dengan Maple, karena kita tidak perlu menghafal diluar kepala semua perintah atau fungsi yang terdapat pada Maple. Bagi yang benar - benar pemula untuk program Maple, dapat memilih *New User's Tour* dibawah menu **Help**. Kemudian Maple akan menuntun pembaca kepada **Quick Tour** seperti tampilan pada gambar 1.3.



Gambar 1.3: Tampilan Quick Tour pada Maple, yang ditujukan bagi pemula.

1.6 Menyimpan Lembar Kerja

Lembar kerja pada Maple yang digunakan dapat disimpan dan sewaktu waktu dapat dipanggil kembali. Perintah untuk menyimpan file lembar kerja Maple adalah : **Save**. Caranya adalah : klik **File** → **Save**. Maple akan menyimpan file tersebut dengan ekstensi `mws`. Sedang untuk memanggil file yang kita simpan, dapat dilakukan dengan mengklik **File** kemudian **Open** lalu pilih file : `namafile.mws`. Untuk memanggil kembali file yang telah tersimpan, kita pakai perintah : **File** → **open**, lalu ketik nama file yang mau kita panggil. Sedang untuk keluar dari Maple, perintah yang digunakan adalah : **File**, kemudian **exit**.

Bab 2

Operasi Dasar Matematika dengan Maple

Seperti yang telah disinggung diatas, bahwa menggunakan Program Maple tak ubahnya seperti menggunakan sebuah kalkulator ilmiah (*scientific calculator*). Sama dengan kalkulator, sebelum menggunakannya, kita harus mengetahui dahulu fungsi - fungsi yang ada pada kalkulator tersebut. Dalam bab ini pembaca akan diperkenalkan kepada fungsi - fungsi dasar Maple yang sering digunakan dalam proses perhitungan dasar matematika.

2.1 Operasi Aritmatika

Aritmatika adalah cabang matematika yang mempelajari operasi matematika yang mendasar seperti: penjumlahan, pengurangan, perkalian, pembagian, pemangkatan, dan penarikan akar. Dalam subbab ini akan diperlihatkan bagaimana Program Maple dapat digunakan untuk mengerjakan persoalan aritmatika. Setiap operasi aritmatika dalam Program Maple diberikan simbol. Pada umumnya simbol yang digunakan oleh Maple hampir sama dengan simbol yang digunakan sehari - hari. Tabel 2.1 memperlihatkan simbol - simbol operasi aritmatika pada Maple.

Keterangan	Penulisan Standard	Penulisan Maple
Penjumlahan	$a + b$	$a + b$
Pengurangan	$a - b$	$a - b$
Perkalian	$a \times b$	$a * b$
Pembagian	$\frac{a}{b}$	a / b
Pemangkatan	a^b	$a ^ b$
Faktorial	!	!
\sqrt{a}	$\text{sqrt}(a)$	

Tabel 2.1: Cara penulisan operasi dasar matematika pada Maple.

Dari tabel 2.1, yang perlu diperhatikan adalah, operasi perkalian. Simbol operasi perkalian pada Maple tidak ditulis (\times), melainkan menggunakan simbol asterisk ($*$). Operasi pemangkatan, Maple menggunakan simbol ($^$). Simbol yang digunakan untuk faktorial adalah sama yaitu $!$. Khusus untuk akar pangkat dua, Maple menggunakan perintah `sqrt`, *square root*. Bentuk dari perintah `sqrt` ini adalah :

`sqrt(bilangan);`

Contoh 2.1.1. Hitunglah dengan menggunakan Maple :

1. $125 + 475;$
2. $4569 - 1789;$
3. $\frac{225}{15};$
4. $-235 \times 14.$

Pada lembar kerja Maple dimasukkan perintah seperti dibawah ini,

```
[> 125 + 475;
```

600.

Soal selanjutnya,

```
[> 4569 - 1789;
```

2780,

Operasi pembagian $\frac{225}{15}$, pada Maple ditulis dalam bentuk :

```
[> 225/15;
```

15.

Operasi perkalian, -235×14 , pada lembar kerja Maple ditulis sebagai berikut :

```
[> -235*14;
```

-3290.

Contoh 2.1.2. Hitunglah operasi pemangkatan dibawah ini :

1. $\sqrt[3]{9^2};$
2. $\sqrt{81}.$

Bentuk $\sqrt[3]{9^2}$ dapat ditulis sebagai $9^{2/3}$,

```
[> 9^(2/3);
```

$9^{(2/3)};$

```
[> sqrt(81)
```

9.

Operasi pemangkatan $a^{m/n}$ ditulis $a^{(m/n)}$. Untuk $m/n = 1/2$, maka bisa dipakai perintah `sqrt`.

Contoh 2.1.3. Hitunglah $\sqrt{27}$.

```
[> sqrt(27);
```

$$3\sqrt{3}.$$

Program Maple dilengkapi dengan kemampuan menghitung bilangan kompleks. Sebagai contoh, memasukkan akar pangkat dua dari satu bilangan negatif,

$$\sqrt{-4}.$$

Maka dengan Maple akan diperoleh hasil seperti :

```
[> sqrt(-4);
```

$$2I.$$

Simbol I pada jawaban diatas menyatakan bilangan imajiner, $(\sqrt{-1})$. Perhatikan contoh dibawah ini

Contoh 2.1.4. Hitunglah nilai dari : $23 \times 85 + 14/43$

Jawaban dari soal di atas adalah :

```
> 23*85 + 14/43;
```

$$\frac{63310}{43}.$$

Sejauh ini terlihat Maple selalu berusaha menjawab hasilnya dengan bilangan pecahan (bukan desimal). Sekarang coba lagi lakukan perhitungan dengan dua contoh berikut ini,

Contoh 2.1.5. Hitunglah nilai dari $2 - 3 + 4/5 \times 6^7$ dan $1/3 + 1/2$.

```
[> (2-3)+(4/5)*(6^7);
```

$$\frac{119739}{5}.$$

```
[>1/3 + 1/2;
```

$$\frac{5}{6}$$

Dari dua contoh tersebut, hasil yang diperoleh adalah bilangan *real*. Selama Maple diberikan bilangan dalam bentuk *integer* (bukan bilangan desimal), maka Maple akan mengeluarkan hasil dalam bilangan *real*. Jika hasil suatu perhitungan diinginkan bilangan desimal, maka angka yang dimasukkan pada lembar kerja Maple harus ditulis dengan bilangan desimal juga. Sebagai contoh, bila angka 14, pada soal $23 \times 85 + 14/43$ ditulis menjadi 14.0, seperti,

```
[> 23*85 + 14.0/43;
```

maka hasilnya adalah :

1955.325581.

Hal yang sama untuk operasi akar pangkat dua. Misalnya, ingin diketahui nilai dari $\sqrt{5}$. Jika pada lembar kerja Maple dituliskan seperti :

```
[> 5^(1/2);
```

atau

```
[> sqrt(5);
```

maka hasil dari dua pernyataan tersebut adalah,

$$\sqrt{5}.$$

Agar operasi akar tersebut menghasilkan bilangan desimal, maka angka 5 diganti menjadi 5.0, seperti :

```
[>(5.0)^(1/2);
```

atau

```
[>sqrt(5.0);
```

maka hasil yang diperoleh,

2.236067977.

Cara lain agar setiap operasi matematik menghasilkan bilangan desimal, maka pada operasi perhitungan digunakan perintah **evalf**. Kaedah penulisan dari perintah **evalf** adalah :

`evalf(pernyataan matematika).`

Contoh 2.1.6. Hitunglah nilai dari $23 \times 85 + 14/43$ dalam bentuk desimal.

```
[> evalf( 23*85 + 14/43 );
```


2.2 Fungsi Trigonometri

Argumen atau sudut dalam trigonometri biasanya dinyatakan dalam derajat atau radian. Hal mendasar yang perlu diketahui dalam menggunakan Program Maple adalah, sudut selalu dinyatakan dalam **radian**. Nilai π dalam radian setara dengan 180° dalam derajat, (*degree*). Untuk menyatakan π dalam Program Maple dapat dilakukan dengan dua cara yaitu : **pi** dan **Pi**. Program Maple menyediakan fasilitas yang dapat mengkonversi besaran sudut dari radian menjadi derajat (*degree*), atau sebaliknya. Perintah yang dipakai adalah `convert`. Kaedah penulisan perintah `convert` ini adalah :

$$\text{convert}(\text{bentuk1}, \text{bentuk2}).$$

Contoh 2.2.1. Ubahlah nilai $1/2\pi$ radian menjadi derajat, kemudian nyatakan 30° derajat menjadi radian.

Untuk mengubah $1/2\pi$ radian menjadi derajat,

```
[> convert(1/2*Pi, degrees);
```

$$90\text{degrees}.$$

Sedang untuk proses sebaliknya yaitu dari besaran derajat menjadi radian seperti contoh dibawah ini.

Contoh 2.2.2. Ubahlah 30° derajat menjadi radian.

```
[>convert(30*degrees, radians);
```

hasilnya adalah :

$$\frac{\pi}{6}.$$

Contoh 2.2.3. Hitunglah nilai dari $\sin(90^\circ)$ dan $\cos(30^\circ)$.

Untuk menghitung nilai tersebut, langkah pertama adalah mengubah nilai 90° dan 30° kedalam radian yaitu sebesar $1/2\pi$ dan $1/6\pi$. Dalam lembar kerja Maple kemudian ditulis :

```
[> sin(Pi/2);
```

$$1,$$

```
[> cos(Pi/6);
```

$$\frac{\sqrt{3}}{2}.$$

Catatan 2.2.1. Program Maple sifatnya *case sensitive*, artinya Maple akan membedakan perintah yang ditulis dengan huruf besar dan huruf kecil. Nilai π akan dinyatakan sebagai 3.14... jika ditulis **Pi** (diawali dengan huruf kapital), dan akan dinyatakan sebagai simbol π , jika ditulis dengan **pi** (diawali dengan huruf kecil).

Perhatikan contoh dibawah ini :

```
[> sin(pi);
```

akan menghasilkan:

$$\sin(\pi)$$

sedangkan pernyataan dibawah ini:

```
[> sin(Pi);
```

akan menghasilkan

$$0.$$

Fungsi - fungsi Maple yang telah tersedia (*built-in function*) untuk persoalan Trigonometri diperlihatkan pada tabel 2.2.

Penulisan Standard	Penulisan Maple	Penulisan Standard	Penulisan Maple
$\sin(x)$	$\sin(x)$	$\sec(x)$	$\sec(x)$
$\cos(x)$	$\cos(x)$	$\operatorname{cosec}(x)$	$\operatorname{csc}(x)$
$\tan(x)$	$\tan(x)$	$\cot(x)$	$\cot(x)$
$\arcsin(x)$	$\arcsin(x)$	$\sinh(x)$	$\sinh(x)$
$\arccos(x)$	$\arccos(x)$	$\cosh(x)$	$\cosh(x)$
$\arctan(x)$	$\arctan(x)$	$\tanh(x)$	$\tanh(x)$

Tabel 2.2: Fungsi fungsi Trigonometri yang telah disediakan oleh Maple, termasuk kaedah penulisannya.

2.3 Logaritma

Pernyataan logaritma bilangan x dengan dasar a , pada program Maple dinyatakan :

$$\log[a](x).$$

Sedangkan untuk logaritma dengan dasar bilangan natural e , digunakan :

$$\ln(x),$$

dimana x adalah bilangan yang hendak dicari nilai logaritmanya.

Contoh 2.3.1. Hitunglah nilai ${}^2\log(8)$

Pemecahan persoalan diatas dengan Maple adalah :

```
[>log[2](8);
```

dan jawabannya adalah :

3.

Contoh 2.3.2. Hitunglah logaritma 25 dengan dasar bilangan natural, e .

Pemecahan pada Maple adalah :

```
[> ln(25);
```

$2 \ln(5)$.

Agar hasil perhitungan diatas dinyatakan dalam bentuk desimal, maka angka 25 diganti menjadi bilangan desimal, 25.0.

```
[> ln(25.0);
```

3.218875825.

2.4 Exponensial

Fungsi ekponensial e^x , dalam Maple dinyatakan dengan :

$\exp(x)$.

Untuk menghitung e^5 misalnya, maka pada lembar kerja Maple ditulis :

```
[> exp(5);
```

dan hasilnya adalah :

e^5 .

Agar hasilnya dalam bentuk desimal, maka angka 5 diatas kita tulis 5.0,

```
[> exp(5.0);
```

atau bisa juga memakai perintah `evalf`, seperti :

```
[> evalf(exp(5));
```

kedua pernyataan diatas akan menghasilkan :

148.4131591.

Beberapa contoh penggunaan fungsi - fungsi trigonometri, eksponensial dan logaritma akan ditampilkan dibawah ini.

Contoh 2.4.1. Untuk menghitung $e^{-5} = \frac{1}{e^5}$:

```
[>evalf(exp(-5));
```

0.006737946999.

Persoalan $\ln(e^3)$ dapat dipecahkan Maple dengan :

```
[> evalf(ln(exp(3)));
```

3.

Hitunglah $e^{\ln 4}$,

```
[>exp(ln(4));
```

4.

2.5 Bilangan Kompleks

Bilangan kompleks (*complex number*) adalah bilangan yang berbentuk $a + bi$, dimana a dan b adalah bilangan real, dan $i^2 = -1$. Penulisan bilangan kompleks secara umum :

$$z = x + yi.$$

Sedangkan yang dimaksud konjugasi \bar{z} adalah :

$$\bar{z} = x - yi.$$

Seperti telah diperlihatkan pada beberapa contoh soal diatas, Maple akan mendefinisikan bilangan imajiner sebagai I (huruf kapital). Untuk penulisan $z = 6 + 3i$ pada Maple harus ditulis :

```
[> z:= 6 + 3*I;
```

$$z := 6 + 3I.$$

Contoh 2.5.1. Hitunglah besarnya (a). i^{17} , (b). $\frac{3-2i}{2-3i}$.

Penyelesaian soal (a) :

```
[> I^(17);
```

I .

Sedangkan penyelesaian soal (b) dengan Maple adalah:

```
[> (3-2*I)/(2-3*I);
```

$$\frac{12}{13} + \frac{5}{13}I.$$

2.6 Bilangan Prima

Bilangan prima didefinisikan sebagai bilangan asli yang lebih besar dari 1, hanya dapat dibagi oleh 1, dan bilangan itu sendiri. Contoh bilangan prima : 2, 3, 5, 7, 11, ... Maple menyediakan fasilitas untuk mencari faktorisasi bilangan prima. Fasilitas ini dinyatakan dengan perintah `ifactor`. Bentuk perintah dari `ifactor` ini adalah :

```
ifactor(bilangan)
```

Contoh 2.6.1. Carilah faktorisasi prima pada bilangan 120

Untuk menjawab persoalan diatas, pada lembar kerja Maple dituliskan :

```
[> ifactor(120);
```

```
(2)3(3)(5).
```

2.7 Memanggil hasil Perhitungan sebelumnya

Ada kalanya saat kita melakukan satu perhitungan kita memerlukan hasil hitungan yang sebelumnya. Maple menyediakan fasilitas untuk memanggil hasil perhitungan yang sebelumnya dengan simbol persen (%). Prinsip ini sama dengan beberapa kalkulator scientific yang menyediakan fasilitas `ans` untuk memanggil hasil perhitungan yang sebelumnya.

Contoh 2.7.1. Hitunglah hasil $22^3 + 89$.

Jika 22^3 dihitung lebih dahulu, kemudian hasilnya ditambahkan dengan 89, maka penulisan pada lembar kerja Maple adalah sebagai berikut:

```
[> 22^3;
```

```
1068.
```

Kemudian hasil tersebut ditambahkan dengan 89, maka cara cepatnya tanpa harus menuliskan angka 1068,

```
[> % + 89;
```

```
10737.
```

Kita dapat meuliskan pada lembar kerja Maple :

```
[> 3*% + %^2;
```

```
113411848,
```

untuk menyatakan $3 \times 1068 + 1068^2$.

2.8 Soal soal dan penyelesaiannya

1. Sederhanakan pernyataan dibawah ini:

$$\sqrt{2\sqrt{221} + 30}.$$

```
[> simplify(sqrt(2*sqrt(221)+30));
```

$$\sqrt{13} + \sqrt{17}$$

2. Hitung $(2 + 5i) + (1 - i)$

```
[> (2 + 5*I)+(1 - I);
```

$$3 + 4I.$$

3. Hitung nilai dari $\frac{1+i}{3-2i}$.

```
[> (1+ I ) / (3-2*I);
```

$$\frac{1}{13} + \frac{5}{13}I.$$

4. Carilah faktor bilangan prima dari 60.

```
[> ifactor(60);
```

$$(2)^2(3)(5)$$

5. Hitunglah $\sqrt{7} + \sqrt[3]{25}$ dalam bentuk desimal.

```
[> evalf(sqrt(7) + 25^(1/3));
```

$$5.569769049$$

6. Hitunglah nilai dari :

(a) $\sin(35^\circ)$

(b) $\tan(58^\circ 48')$

Untuk soal a.

```
[> convert(35*degrees, radians);
```

$$\frac{7\pi}{36}$$

```
[> evalf(sin(7*Pi/36));
```

0.5735764363.

Untuk soal b.

$$58^{\circ}48' = 58 + \left(\frac{48}{60}\right)$$

```
[> 58 + 48/60;
```

58.8

```
[> convert(58.8*degrees, radians);
```

```
[> a:=convert(58.48*degrees, radians);
```

$$a := 0.3266666667\pi$$

```
[>evalf(tan(a));
```

1.651196259.

2.9 Soal Latihan

Buatlah soal - soal dibawah ini dengan menggunakan Program Maple.

1. Hitunglah

(a) $\sqrt[3]{81.72}$

(b) $\frac{0.3168^2 \times \sqrt{31.40^3}}{0.2645^4}$

(c) $\sqrt{54.85^2 + 75.64^2}$.

(d) $2^{2.5}$.

2. Hitunglah

(a) ${}^8 \log 512$.

(b) ${}^2 \log \sqrt{6} + \frac{1}{2} {}^2 \log 3$.

3. Tentukan hasil dari penjumlahan dan pengurangan dibawah ini dalam bentuk i .

(a) $\sqrt{-100} - \sqrt{-81}$

(b) $-1/2 + \sqrt{-2/3} - 1/2 + \sqrt{-24/9}$

(c) $\sqrt{4} + \sqrt{-32} + \sqrt{-8}$

(d) $1/7 + \sqrt{-1/8} + 2/7 - \sqrt{-1/2}$

(e) $\sqrt{50} + \sqrt{-2} + \sqrt{200} + \sqrt{-50}$

4. Sederhanakan pernyataan pernyataan dibawah ini :

(a) $3i + 2i^3$

(b) $i + i^3 + i^5$

(c) $i + i^2 + i^3 + i^4$

(d) $i - i^5$

(e) $4i + 5i^8 + 6i^3 + 2i^4$

5. Luas persegi panjang adalah $536.8 [m^2]$, dan lebarnya adalah $20.8 [m]$. Hitunglah kelilingnya.

6. Berapakah nilai $\log \log 5.1243$

7. Volume sebuah bola adalah $1.171 [cm^3]$. Berapakah jari - jarinya, dan berapakah luas permukaannya.

8. Periode T dari sebuah bandul yang berayun adalah

$$T = 2\pi \sqrt{\frac{l}{g}}$$

Dimana l adalah panjang tali dan g adalah gravitasi. Berapakah besarnya T , jika $l = 281.3 [cm]$, dan $g = 981 [cm/s^2]$.

Bab 3

Menyelesaikan Aljabar dengan Maple

3.1 Membuat Variabel

Jika sebuah bilangan secara berulang dipakai dalam suatu perhitungan, maka akan lebih praktis bilamana bilangan tersebut didefinisikan menjadi sebuah variabel. Dalam program Maple, sebuah variabel dapat dinyatakan secara langsung. Untuk mendefinisikan suatu variabel harus ditulis dengan tanda colon dan sama dengan (:=). Sebagai contoh bila ingin mendefinisikan variabel f adalah 25, pada lembar kerja Maple dimasukkan :

```
[> f:=25;
```

Maple akan menyimpannya harga f adalah 25. Demikian pula dengan :

```
[>h := 2 + 2;
```

Maka Maple akan menyimpan variabel $h = 4$. Atau bisa juga menamakan variabel dengan kata yang mudah diingat seperti :

```
[>jawaban := 56;
```

Ini bearti **jawaban** mempunyai nilai 56. Hal yang perlu diperhatikan dalam pemakaian kata untuk membuat variabel adalah, penulisan nama dari suatu variabel harus dimulai dengan huruf, setelah itu dapat dikombinasikan baik dengan angka atau huruf. Untuk melihat kembali variabel yang telah didefinisikan, cukup mengetikkan kembali nama variabelnya seperti :

```
[>h;
```

4,

```
[>jawaban;
```

56,

```
[>f;
```

25.

Variabel yang telah dibuat dapat didefinisikan ulang dengan cara mengetik ulang nilainya. Sebagai contoh, nilai $f = 25$ hendak diubah menjadi $f = 750$,

```
[>f;
```

25

```
[>f:= 750;
```

$f := 750.$

Sebuah variabel yang telah dibuat dapat dihapus. Untuk menghapus variabel digunakan perintah `unassign`. Kaedah penulisan perintah ini adalah :

```
unassign(var1, var2, ...).
```

Dimana `var1`, `var2` adalah variabel variabel yang hendak dihapus. Sebagai contoh untuk menghapus variabel $p = 25$, maka langkah yang harus dilakukan adalah,

```
[> unassign('p');
```

Dalam bekerja dengan Maple, penggunaan banyak variabel harus disikapi dengan hati - hati. Kesalahan umum yang sering terjadi adalah, pengguna Maple lupa bahwa ada variabel yang telah didefinisikan pada satu persoalan, namun tanpa disadari variabel tersebut digunakan untuk persoalan lain. Disamping itu menggunakan banyak variabel akan menyita memori komputer. Jika hal ini terjadi, Maple dapat melakukan salah hitung atau melaksanakan perintah secara tidak akurat. Perintah `restart` digunakan untuk mengatasi masalah ini. Perintah tersebut akan menghapus semua variabel yang ada pada *internal memory*. Untuk itu disarankan memasukkan perintah `restart` pada awal lembar kerja. Hal ini untuk meyakinkan bahwa semua variabel dari pekerjaan kita yang terdajulu terhapus. Variabel - variabel yang telah didefinisikan, oleh Maple dapat dioperasikan secara simbolik.

Contoh 3.1.1. Jika $p = 3$, dan $q = 11$, hitunglah nilai

1. $p + q$

2. $p \cdot q$

3. p/q

```
[>p:= 3;
```

$p = 3,$

```
[>q:= 11;
```

$$q = 11.$$

[> p + q;

14,

[> p*q;

33.

[> p/q;

$\frac{3}{11}$.

Maple memberikan kebebasan kepada kita dalam membuat nama variabel. Namun ada beberapa kata atau huruf yang telah menjadi *default*-nya Maple, kata atau huruf ini tidak dapat dipakai lagi menjadi nama sebuah variabel. Kata - kata atau huruf tersebut antara lain :

eval	perintah ini dikenal Maple untuk mengevaluasi suatu fungsi
Factor	dikonal Maple untuk memfaktorkan pernyataan aljabar
expand	dikonal Maple untuk mengekspansikan pernyataan aljabar
Diff	dipakai Maple untuk mediferensialkan satu fungsi secara simbolik
Simplify	dikonal Maple sebagai perintah untuk menyederhanakan satu pernyataan aljabar
Pi	adalah nilai sebesar π yaitu : 3.14159 ...
I	bilangan imajiner $\sqrt{-1}$

Jika tetap kata atau huruf diatas dipakai sebagai nama variabel, maka Maple akan mengeluarkan pernyataan sbb :

Error, attempting to assign to which is protected.

3.2 Urutan operasi hitung

Operasi hitung pada Maple mengikuti urutan hirarki operasi bilangan. Sebagai contoh dalam menghitung $\frac{x+5}{2}$ pada lembar kerja Maple tidak dapat ditulis seperti : $x + 5/2$. Pernyataan tersebut, akan menghasilkan $x + \frac{5}{2}$. Tanda kurung biasa (...) harus dikenakan untuk menjelaskan urutan operasi yang diinginkan. Jadi pernyataan $\frac{x+5}{2}$ harus ditulis dalam lembar kerja Maple ($x + 5$)/2.

Demikian halnya pernyataan : $(1+x)^{1/x}$ akan dibaca oleh program Maple sebagai $\frac{(1+x)^1}{x}$. Sekarang bandingkan jika ditulis seperti : $(1+x)^{(1/x)}$, perhatikan tanda kurung. Maka pernyataan tersebut akan dibaca oleh Maple sebagai : $(1+x)^{\frac{1}{x}}$. Sama halnya dengan pernyataan $x^3 * x$, ini akan menghasilkan $x^3 * x = x^4$. Lain halnya jika ditulis $x^{(3*x)}$, hasilnya adalah x^{3x} .

Kesalahan yang sering terjadi pada penulisan pada lembar kerja Maple adalah dalam menterjemahkan pernyataan simbolik kedalam lembar kerja Maple. Pernyataan perkalian dalam aljabar $2xy$ ($2 \times x \times y$), pada lembar kerja Maple sering ditulis sebagai $2xy$. Hal ini keliru, yang benar adalah $2*x*y$, perhatikan tanda asterisk diantaranya.

3.3 Perintah subs

Perintah subs digunakan untuk mensubstitusi satu variabel dengan bilangan atau simbol. Perintah subs mempunyai format dasar :

```
subs( x= nilai1, y = nilai2, ..., pernyataan);
```

Contoh 3.3.1. Jika $a = 5, b = -5$ dan $c = 2$, hitunglah $y = ax^2 + bx + c$

Langkah pertama adalah dengan membuat dahulu persamaan y ,

```
[>y:= a*x^2 + b*x + c;
```

$$y = ax^2 + bx + c.$$

Kemudian substitusikan nilai a, b, c ke persamaan y , dengan perintah

```
[>subs(a=5, b=-5, c=2, y);
```

dan hasilnya adalah :

$$y = 5x^2 - 5x + 2.$$

Contoh 3.3.2. Hitunglah nilai $k = z^2 + 3$ dimana $z = x + y$.

Perintah subs dapat digunakan untuk mensubstitusi simbol.

```
[>k := z^2 + 3;
```

$$k := z^2 + 3.$$

Substitusikan $z = x + y$ ke k ,

```
[>subs(z=x+y, k);
```

hasilnya adalah :

$$(x + y)^2 + 3.$$

Dalam persoalan tertentu perintah subs tidak memberikan hasil secara numerik, seperti contoh dibawah ini.

Contoh 3.3.3. Hitunglah harga $\sin(x)$ untuk $x = \pi$.

```
[> subs(x = Pi, sin(x));
```

$$\sin(\pi).$$

Bila tetap diinginkan berapa nilai dari $\sin(x)$ dimana $x = \pi$ secara numerik, perintah Maple yang dipakai adalah `eval(evaluated)`. Perintah ini sifatnya hampir sama dengan `subs`, hanya bentuk penulisannya yang sedikit berbeda. Kaedah penulisan dari perintah `eval` adalah :

`eval(f(x), x=a).`

Kembali kepada contoh soal 3.3.3, kali ini persoalan tersebut dikerjakan dengan memakai perintah `eval`.

```
[> eval(sin(x), x=Pi);
```

0.

Contoh 3.3.4. Hitunglah :

$$f(x) := \frac{x^3 + 2x^2 - x - 2}{x^3 + x^2 - 4x - 4}$$

untuk $x = 4$, $x = -3$, $x = 1/2$.

Langkah pertama, buat fungsi f ,

```
[> f := (x^3 + 2*x^2 - x - 2) / (x^3 + x^2 - 4*x - 4);
```

$$f := \frac{x^3 + 2x^2 - x - 2}{x^3 + x^2 - 4x - 4}.$$

Substitusikan $x = 4$ ke f ,

```
[> subs(x=4, f);
```

$\frac{3}{2}$,

```
[> subs(x= -3, f);
```

$\frac{4}{5}$,

```
[> subs(x= 1/2, f);
```

$\frac{1}{3}$.

Sebagai pembanding, dapat juga kasus diatas dikerjakan dengan perintah `eval`,

```
[> eval(f, x=1/2);
```

$\frac{1}{3}$

Sekarang bagaimana jika $x = 2$ disubstitusikan ke persamaan di atas,

```
[> subs(x=2, f);
```

perintah ini tidak dapat dieksekusi oleh Maple. Karena jika $x = 2$ disubstitusikan ke dalam persamaan f , maka nilai penyebut f akan menjadi 0, sehingga nilai pecahan tersebut menjadi tidak terdefinisi. Dalam kasus ini Maple akan mengeluarkan peringatan,

Error, numeric exception: division by zero.

Contoh 3.3.5. Hitunglah persamaan $f(x) = \frac{x^2-4}{x^3+5}$, untuk $x = 3$.

Supaya tidak mengetik ulang, persamaan diatas diberikan nama dahulu, dengan menjadikannya variabel. Dalam hal ini *pers*.

```
[> pers:=(x^2 - 4)/(x^3 + 5);
```

$$pers := \frac{x^2 - 4}{x^3 + 5}.$$

Sekarang substitusikan nilai $x = 3$ ke *pers*, dengan cara :

```
[> subs(x=3, pers);
```

hasilnya,

$$\frac{5}{32}.$$

3.4 Perintah expand

Perintah ini digunakan untuk mengekspansikan (menguraikan) suatu pernyataan aljabar. Format penulisan perintah ini adalah :

```
expand(pernyataan);
```

Contoh 3.4.1. Uraikan pernyataan $(x + 2)^2$.

Jika dalam lembar kerja Maple pernyataan tersebut ditulis seperti ini :

```
[> (x+2)^2;
```

maka Maple akan mengeksekusi pernyataan tersebut sebagai :

$$(x + 2)^2.$$

Namun bila perintah *expand* diaplikasikan, seperti dibawah ini :

```
[> expand((x+2)^2);
```

maka hasilnya adalah :

$$x^2 + 4x + 4.$$

Contoh 3.4.2. Ekspansikan pernyataan ini $(x + y)^2(3x - y)^3$

```
[> expand((x+y)^2*(3*x-y)^3);
```

$$27x^5 + 27x^4y - 18x^3y^2 - 10x^2y^3 + 7xy^4 - y^5$$

Perintah `expand` dapat juga digunakan untuk mengekspansi fungsi - fungsi Trigonometri. Sebagai contoh pernyataan $\sin(2x)$ dapat diuraikan menjadi $2 \sin(x) \cos(x)$. Untuk mencarinya dengan Maple, maka pada lembar kerja Maple dimasukkan :

```
[>expand(sin(2*x));
```

$$2 \sin(x) \cos(x).$$

Contoh 3.4.3. Uraikan bentuk $\cos(3x)$ menjadi bentuk $\cos(x)$.

```
[>expand(cos(3*x));
```

$$4 \cos(x)^3 - 3 \cos(x).$$

3.5 Perintah `factor`

Jika pada sub-bab 3.4 terlihat bagaimana Maple dapat mengekspansikan, atau menguraikan suatu pernyataan aljabar, maka perintah `factor` adalah fungsi sebaliknya. Perintah `factor` digunakan untuk mencari faktor dari sebuah pernyataan aljabar. Bentuk perintah `factor` adalah :

```
factor(pernyataan);
```

Contoh 3.5.1. Faktorkan $12x^2 + 27xy - 84y^2$.

Secara manual kita dapat mencari faktor dari $12x^2 + 27xy - 84y^2$ yaitu: $3(x + 4y)(4x - 7y)$. Dengan Maple penyelesaian persoalan aljabar diatas dapat ditulis :

```
[> factor(12*x^2+27*x*y-84*y^2);
```

$$3(x + 4y)(4x - 7y)$$

Ada kalanya sebuah pernyataan aljabar tidak dapat difaktorkan. Untuk kasus ini, Maple tidak akan menampilkan hasilnya. Program Maple akan menampilkan kembali pernyataan aljabar tersebut.

Contoh 3.5.2. Faktorkan (a). $x^3 - 5$, dan (b). $x^3 + 1$.

Untuk persoalan (a), pada lembar kerja Maple dituliskan,

```
[>factor(x^3-5);
```

hasilnya adalah :

$$x^3 - 5.$$

Hal diatas menunjukkan bahwa $x^3 - 5$ tidak dapat difaktorkan. Sedang untuk soal (b):

```
[>factor(x^3+1);
```

hasilnya adalah :

$$(x + 1)(x^2 - x + 1).$$

3.6 Perintah `simplify`

Dalam persoalan aljabar, sering dijumpai penyederhanaan bentuk dari sebuah pernyataan, khususnya untuk pernyataan yang berbentuk pecahan. Program Maple menyediakan perintah `simplify` untuk keperluan penyederhanaan tersebut. Kaedah penulisan perintah tersebut adalah :

`simplify(pernyataan).`

Contoh 3.6.1. Buktikan bahwa pernyataan $\frac{x^3-y^3}{x^4-y^4}$ dapat disederhanakan menjadi : $\frac{x^2+xy+y^2}{x^3+x^2y+xy^2+y^3}$.

```
[>simplify((x^3-y^3)/(x^4-y^4));
```

$$\frac{x^2 + xy + y^2}{x^3 + x^2y + xy^2 + y^3}$$

Perintah `simplify` dapat juga diaplikasikan untuk perhitungan logaritma.

Contoh 3.6.2. Sederhanakanlah (1). $\ln(x^3)$, dan (2). $\ln(xy)$

1. `[>simplify(ln(x^3));`

$$3 \ln(x)$$

2. `[>simplify(ln(x*y));`

$$\ln(x) + \ln(y)$$

Contoh 3.6.3. Sederhanakanlah bentuk penjumlahan dari pecahan dibawah ini:

$$\frac{2}{x^2} - \frac{x^2}{2}$$

```
[>simplify(2/x^2 - x^2/2);
```

$$-\frac{-4 + x^4}{2x^2}$$

Contoh 3.6.4. Sederhanakan bentuk pecahan dibawah ini :

$$\frac{a^2 - c^2 - b^2 + 2bc}{a^2 - b^2 + 2ac + c^2}$$

```
[>simplify((a^2-c^2-b^2 + 2*b*c)/(a^2 - b^2 + 2*a*c + c^2 ));
```

$$\frac{a + b - c}{a + c + b}$$

Perintah `simplify` dapat digunakan untuk menyederhanakan persamaan Trigonometri. Jika perintah `simplify` ini hendak digunakan untuk aplikasi trigonometri, maka bentuk perintahnya seperti berikut :

`simplify(Pernyataan, trig).`

Pemakaian `trig` pada perintah `simplify` menandakan bahwa dalam proses penyederhanaan persamaan Maple akan memakai prinsip - prinsip trigonometri. Prinsip - prinsip tersebut antara lain $\sin^2(x) + \cos^2(x) = 1$, atau $\cosh(x)^2 - \sinh(x)^2 = 1$.

Contoh 3.6.5. Maple tidak akan langsung mengerjakan $\sin(x)^2 + \cos(x)^2 = 1$. Identitas trigonometri tersebut baru dapat dikerjakan oleh Maple bila perintah `simplify` diaplikasikan.

```
[>simplify(sin(x)^2 + cos(x)^2, trig);
```

1

Contoh 3.6.6. Ubahlah bentuk $\sin(3x) \cos(4x)$ dalam bentuk $\sin(x)$ dan $\cos(x)$.

```
[>simplify(sin(3*x)*cos(4*x), trig);
```

$$\left(-1 + 32 \cos(x)^6 - 40 \cos(x)^4 + 12 \cos(x)^2\right) \sin(x)$$

Contoh 3.6.7. Sederhanakan bentuk trigonometri ini $\cos(x)^5 + \sin(x)^4 + 2 \cos(x)^2 - 2 \sin(x)^2 - \cos(2x)$

Supaya tidak menulis pernyataan yang terlalu panjang, ada baiknya pernyataan diatas dibuat dalam sebuah variabel. Misalkan saja nama variabelnya adalah `pers1`.

```
[>pers1 := cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos(2*x);
```

$$pers1 := \cos(x)^5 + \sin(x)^4 + 2 \cos(x)^2 - 2 \sin(x)^2 - \cos(2x)$$

```
[>simplify(pers1, trig);
```

$$\cos(x)^5 + \cos(x)^4$$

3.7 Perintah `combine`

Perintah `combine` digunakan untuk mengubah pernyataan dari bentuk perkalian menjadi bentuk penjumlahan. Perintah `combine` ini sangat banyak gunanya untuk memecahkan persoalan trigonometri. Bentuk penulisan perintah `combine` pada program Maple adalah :

`combine(Pernyataan).`

Contoh 3.7.1. Ubahlah pernyataan $\cos(x)^3$ dalam bentuk penjumlahan.

```
[> combine(cos(x)^3);
```

$$\frac{1}{4} \cos(3x) + \frac{3}{4} \cos(x).$$

Contoh 3.7.2. Ubahlah pernyataan dibawah ini menjadi bentuk penjumlahan :

$$\sin(3x) \cos(4x).$$

```
[> combine(sin(3*x)*cos(4*x));
```

$$\frac{1}{2} \sin(7x) - \frac{1}{2} \sin(x).$$

3.8 Identitas Trigonometri

Dengan menggunakan perintah `simplify(trig)`, Maple dapat menyelesaikan persoalan identitas Trigonometri.

Contoh 3.8.1. Buktikan bahwa

$$(\tan(x) - \sec(x))(\sin(x) + 1) = -\cos(x)$$

Untuk memecahkan persoalan diatas, pisahkan persamaan tersebut menjadi dua bagian. Bagian pertama adalah persamaan yang terletak pada sebelah kiri tanda sama dengan. Sedangkan persamaan yang kedua, adalah persamaan yang terletak disebelah kanan tanda sama dengan. Masing - masing persamaan tersebut ada baiknya diberi nama *kiri* untuk menyatakan persamaan yang berada pada sebelah kiri,

```
[>kiri := (tan(x) - sec(x))*(sin(x)+1);
```

$$kiri := (\tan(x) - \sec(x))(\sin(x) + 1).$$

Persamaan yang kedua diberi nama *kanan*, karena letaknya berada pada sebelah kanan tanda sama dengan,

```
[>kanan := -cos(x);
```

$$kanan := -\cos(x).$$

Seperti diketahui bahwa $\tan(x) = \frac{\sin(x)}{\cos(x)}$ dan $\sec(x) = \frac{1}{\cos(x)}$. Substitusikan kedua persamaan ini ke persamaan *kiri*,

```
[> subs( {tan(x) = sin(x)/cos(x), sec(x) = 1/cos(x)}, kiri);
```

maka persamaan sebelah kiri akan berubah menjadi;

$$\left(\frac{\sin(x)}{\cos(x)} - \frac{1}{\cos(x)}\right)(\sin(x) + 1).$$

Langkah selanjutnya adalah menyederhanakan bentuk persamaan diatas dengan perintah,

```
[>simplify(%, trig);
```

perhatikan kita memakai tanda % untuk memanggil hasil perhitungan yang sebelumnya. Hasil dari penyederhanaan tersebut adalah :

$$-\cos(x).$$

Hasil ini identik dengan persamaan yang berada pada sebelah kanan. Dengan demikian telah terbukti bahwa persamaan yang berada pada sebelah kiri adalah **sama** dengan persamaan yang berada pada sebelah kanan.

Dalam memecahkan persoalan identitas Trigonometri, kadang kita perlu melakukan perubahan dari satu bentuk ke bentuk lain. Misalnya perubahan dari $\frac{\sin(\alpha)}{\cos(\alpha)}$ menjadi bentuk $\tan(\alpha)$, hal ini tidak dapat dikerjakan secara langsung oleh Maple. Untuk mengubah bentuk ini, Maple menyediakan perintah `convert` (lihat subbab 2.2). Bentuk penulisan `convert` adalah :

```
convert(bentuk1, bentuk2)
```

Pengertian bentuk1, bentuk2 pada perintah diatas adalah bentuk - bentuk perubahan yang diinginkan. Berbagai tipe bentuk1, bentuk2 antara lain : "sincos", "tan". Bentuk sincos mengubah pernyataan Trigonometri kedalam bentuk sin dan cos. Sedangkan bentuk tan mengubah pernyataan Trigonometri dalam bentuk tan.

Contoh 3.8.2. Ubahlah bentuk $\tan(x)$ menjadi $\sin(x)$ dan $\cos(x)$

```
[>convert(tan(x), sincos);
```

$$\frac{\sin(x)}{\cos(x)}.$$

Contoh 3.8.3. Buktikan bahwa :

$$\frac{\cos(a+b)}{\cos(a)\cos(b)} = \tan(a) + \tan(b).$$

Diperlukan dua langkah untuk membuktikan persoalan diatas. Pertama ruas sebelah kiri tanda sama dengan di-expand, dengan perintah

```
[>expand(sin(a+b)/(cos(a)*cos(b)));
```

$$\frac{\sin(a)}{\cos(a)} + \frac{\sin(b)}{\cos(b)}.$$

Langkah selanjutnya adalah mengubah pernyataan diatas dalam bentuk tan. Hal ini dikerjakan dengan

```
[>convert(%, tan);
```

$$\tan(a) + \tan(b).$$

Catatan 3.8.1. Tidak semua persoalan identitas Trigonometri dapat dipecahkan oleh Maple

3.9 Membuat fungsi

Relasi dari himpunan **A** ke himpunan **B** yang menghubungkan setiap anggota himpunan **A** dengan satu anggota **B**, disebut fungsi. **B** sering juga disebut sebagai daerah hasil, dan **A** disebut sebagai daerah asal atau domain. Secara matematik fungsi dapat dinyatakan,

$$f : A \rightarrow B.$$

Dengan Program Maple sebuah fungsi dapat dibuat untuk keperluan sendiri. Sebuah fungsi pada Maple harus dinyatakan dengan tanda panah \rightarrow . Tanda panah dapat dibuat dengan menggunakan tombol " - " dan " > " pada keyboard komputer. Tanda panah (\rightarrow) dapat juga dibuat dengan menggunakan perintah `unapply`.

3.9.1 Fungsi Single Variabel

Cara membuat fungsi yang mempunyai single variabel pada Maple adalah :

$f := x \rightarrow$ pernyataan di dalam x .

Contoh 3.9.1. Buatlah sebuah fungsi dari single variabel yakni $f(x) = 3x + 5$. Kemudian hitunglah : $f(2)$, $f(5)$, dan $f(-10)$.

Langkah pertama adalah membuat fungsi yang dimaksud yaitu :

```
[>f := x -> 3*x + 5;
```

$$f := x \rightarrow 3x + 5;$$

Atau dapat juga menggunakan perintah `unapply` sebagai ganti tanda panah.

```
[> f:=unapply(3*x+5,x).
```

$$f := x \rightarrow 3x + 5;$$

Setelah fungsi tersebut jadi, sekarang digunakan untuk menghitung nilai $f(2)$, $f(5)$, $f(-10)$:

```
[>f(2);
```

11.

```
[>f(5);
```

20.

```
[>f(-10);
```

25.

Contoh 3.9.2. Buatlah fungsi dengan satu variabel seperti dibawah ini :

$$f(x) = \frac{x}{x^2 + 1}.$$

Kemudian hitunglah $f(3)$, $f(a)$.

Sama seperti contoh diatas, fungsi $\frac{x}{x^2+1}$ dibuat dahulu dengan menggunakan Maple,

```
[> f:= x -> x/(x^2+1);
```

$$f := x \rightarrow \frac{x}{x^2 + 1}.$$

Untuk menghitung nilai $f(3)$;

```
[>f(3);
```

$\frac{3}{10}$.

```
[>f(a);
```

$\frac{a}{a^2 + 1}$.

3.9.2 Fungsi Lebih dari Satu Variabel

Pada prinsipnya untuk membuat fungsi yang mempunyai dua variabel $f(x, y)$ adalah sama dengan membuat fungsi yang mempunyai satu variabel. Kaedah penulisannya pada Maple adalah :

$$f := (x, y) \rightarrow \text{ekspresi dalam } (x, y)$$

Contoh 3.9.3. Buatlah sebuah fungsi yang terdiri dari dua variabel (x, y)

$$g(x, y) = \sin(x) \cos(y) + xy.$$

Kemudian hitunglah nilai $g(x, y) = (\pi/2, \pi)$.

Pertama fungsi $g(x, y)$ dibuat dahulu dengan Maple;

```
[> g := (x,y) -> sin(x)*cos(y) + x*y;
```

$$g := (x, y) \rightarrow \sin(x) \cos(y) + xy;$$

Untuk menghitung $g(x, y) = (\pi/2, \pi)$,

```
[>g(Pi/2,Pi);
```

$$-1 + \frac{1}{2}\pi^2.$$

3.9.3 Fungsi Komposisi

Apabila ada dua buah fungsi yaitu $f : A \rightarrow B$ dan $g : B \rightarrow C$. Fungsi dari A ke C yang memetakan setiap anggota $a \in A$ ke $g[f(a)] \in C$ disebut fungsi *komposisi* dari f ke g , $f \circ g$.

$$(g \circ f)(x) = g[f(x)].$$

Simbol yang digunakan Maple untuk fungsi komposisi adalah @. Sedang kaedah penulisan untuk membuat fungsi komposisi adalah :

$$(f1@f2@f3 \dots @fn)(x).$$

Contoh 3.9.4. Diketahui $f(x) = x^2 - 3x + 2$ dan $g(x) = 2x + 1$, tentukanlah :

1. $(f \circ g)(x)$.
2. $(f \circ g)(2)$.

Untuk mencari $(f \circ g)(x)$, pada lembar kerja Maple dimasukkan,

```
[>(f@g)(x);
```

$$(2x + 1)^2 - 6x - 1$$

Bentuk diatas dapat disederhanakan menjadi

[>simplify((f@g)(x));

$$4x^2 - 2x.$$

Nilai dari $(f \circ g)(2)$ dapat dicari dengan,

[>(f@g)(2);

$$12.$$

Contoh 3.9.5. Diketahui $f(x) = 3x + 2$, $g(x) = x^2$, dan $h(x) = 1/x$ tentukanlah :

1. $[(f \circ g) \circ h](x)$.
2. $[f \circ (g \circ h)](x)$.
3. $[(f \circ g) \circ h](2)$

Untuk mencari $[(f \circ g) \circ h](x)$,

[>((f@g)@h)(x);

$$\frac{3}{x^2} + 2.$$

Untuk mencari $[f \circ (g \circ h)](x)$

[>(f@(g@h))(x);

$$\frac{3}{x^2} + 2.$$

Ternyata $[(f \circ g) \circ h](x)$ sama dengan $[f \circ (g \circ h)](x)$. Untuk menentukan $[(f \circ g) \circ h](2)$

[>((f@g)@h)(2);

$$\frac{11}{4}.$$

3.10 Soal soal dan penyelesaiannya

1. Definisikan $a = 5/2$, dan $b = 6$. Maka hitunglah nilai nilai dibawah ini:

(a) $c = ab$

(b) $d = c^2$

(c) $e = d - 20$

(d) $y = ax^2 + b * x + c$

```
[>a := 5/2;
```

$$a := \frac{5}{2}.$$

```
[>b := 6;
```

$$b := 6.$$

```
[>c := a*b;
```

$$c := 15.$$

```
[> d := c^2;
```

$$d := 225.$$

```
[>e := d - 20;
```

$$e := 205.$$

```
[> y := a*x^2 + b*x + c;
```

$$y := \frac{5}{2}x^2 + 6x + 15.$$

2. Faktorkan pernyataan $a^3 - b^3$.

```
[>factor(a^3 - b^3);
```

$$(a - b) (a^2 + ab + b^2)$$

3. Faktorkan pernyataan $a^9 - b^9$.

```
[>factor(a^9 - b^9);
```

$$(a - b) (a^2 + ab + b^2) (a^6 + a^3b^3 + b^6)$$

4. Jabarkan pernyataan trigonometri ini, $\sin(x + y)$.

```
[> expand(sin(x+y));
```

$$\sin(x) \cos(y) + \cos(x) \sin(y).$$

5. Faktorkan pernyataan : $x^5 - x^4 - 7x^3 + x^2 + 6x$.

```
[>poli:=x^5 -x^4 -7*x^3 + x^2 + 6*x;
```

$$poli := x^5 - x^4 - 7x^3 + x^2 + 6x.$$

```
[>factor(poli);
```

$$x(x - 1)(x - 3)(x + 2)(x + 1).$$

6. Ubahlah pernyataan $\cos(x)^5$ dalam bentuk penjumlahan.

```
[>combine(cos(x)^5);
```

$$\frac{1}{16} \cos(5x) + \frac{5}{16} \cos(3x) + \frac{5}{8} \cos(x)..$$

7. Sebuah fungsi $y = x^2 + 11x + 24$, berapakah nilai y untuk $x = 1$ dan $x = t - 3$.

```
[> y := x^2+11*x+24;
```

$$y := x^2 + 11x + 24.$$

Untuk $x = 1$, substitusikan $x = 1$ ke persamaan y , dengan perintah `subs`,

```
[> subs(x = 1, y);
```

36.

Kemudian soal berikutnya untuk $x = t - 3$, substitusikan ke persamaan y ,

```
[> subs(x = t-3, y);
```

$$(t - 3)^2 + 11t - 9,$$

Pada jawaban diatas masih terlihat faktor $(t - 3)^2$, dan halitu masih bisa disederhanakan, dengan perintah :

```
[> simplify(%);
```

$$t^2 + 5t.$$

8. Bentuk sederhana dari :

$$\frac{\sin(a + b)}{\cos(a) \cos(b)}$$

```
[>expand(sin(a+b)/(cos(a)*cos(b)));
```

$$\frac{\sin(a)}{\cos(a)} + \frac{\sin(b)}{\cos(b)}$$

9. Berapa nilai dari $\sin(1/2\pi + a) - \cos(\pi - a)$

```
sin(1/2*Pi+a)-cos(Pi-a);
```

$$2 \cos(a).$$

10. Jika $a = 13, b = 14, c = 15$ dan $s = \frac{1}{2}(a + b + c)$. Tentukan nilai $\sqrt{s(s - a)(s - b)(s - c)}$.

```
[>a:= 13:
>b:= 14:
>c:= 15:
>s:=1/2*(a+b+c):
>sqrt(s*(s-a)*(s-b)*(s-c));
```

84.

11. Sederhanakan bentuk

$$\sin(3\alpha) \cos(2\alpha) + \cos(3\alpha) \sin(2\alpha).$$

```
[>combine(sin(3*alpha)*cos(2*alpha)+
cos(3*alpha)*sin(2*alpha));
```

$$\sin(5\alpha)$$

12. Fungsi $f : R \rightarrow R$ dan $g : R \rightarrow R$ ditentukan dengan rumus $f(x) = x^2 + 1$ dan $g(x) = x + 3$. Tentukan :

- (a) $f \circ g$,
- (b) $g \circ f$,
- (c) $f \circ f$,
- (d) $g \circ g$.

Untuk soal (a).

```
[>(f@g)(x);
```

$$(x + 3)^2 + 1.$$

Untuk soal (b).

```
[>(g@f)(x);
```

$$x^2 + 4.$$

Untuk soal (c).

```
[>(f@f)(x);
```

$$(x^2 + 1)^2 + 1.$$

atau

[>simplify((f@f)(x));

$$x^4 + 2x^2 + 2.$$

Untuk soal (d).

[> (g@g)(x);

$$x + 6.$$

3.11 Soal soal latihan

1. Sederhanakan bentuk pecahan dibawah ini:

$$\frac{(a+b)^2 - 2c(a+b) - 3c^2}{(a+b)^2 - 6c(a+b) + 9c^2}.$$

2. Jabarkan pernyataan $(\cos(\theta) + i \sin(\theta))^5$.
3. Jika $f : R \rightarrow R$ dan $g : R \rightarrow R$, ditentukan $f(x) = x^2 + 3x + 2$ dan $g(x) = 2x + 5$. Hitunglah $(f \circ g)(2)$.
4. Hitung nilai dari $\sin(1/2\pi + a) + \sin(2\pi + a)$.
5. Faktorkan $x^5 - x^4 - 7x^3 + x^2 + 6x$.
6. Jabarkan $(1 + 2x)^6$.
7. Hitunglah $(x + y)^2(3x - y)^3$.
8. Buktikan bahwa :

$$\frac{\sin(a+b) + \sin(a-b)}{\cos(a+b) + \cos(a-b)} = \tan(a)$$
9. Diketahui $f(x) = x + 1$ dan $g(x) = \sqrt{x^2 - 1}$. Carilah rumus untuk $(f \circ g)(x)$.
10. Sederhanakan bentuk pecahan dibawah ini:

$$\frac{x^4 + x^3 + 4x^2 - x}{x^3 + 2x^2 + 8x}.$$

Bab 4

Penyelesaian Persamaan dan Pertidaksamaan

Dalam bab ini, kita akan melihat bagaimana program Maple dapat digunakan untuk memecahkan sebuah atau sistem persamaan. Selain itu, Maple dapat digunakan untuk mencari akar - akar dari persamaan polynomial (persamaan pangkat banyak). Pada bagian akhir bab ini, akan dibahas bagaimana Maple dapat memecahkan persoalan pertidaksamaan. Sebuah persamaan, oleh Maple didefinisikan sebagai :

$$\text{ruas kiri} = \text{ruas kanan.}$$

Dari persamaan $3x + 7 = 4$, yang dimaksud ruas kiri adalah $3x + 7$ sedang ruas kanannya adalah 4.

4.1 Perintah `solve`

Maple mempunyai perintah untuk memecahkan persamaan yaitu `solve`. Bentuk penulisan perintah `solve` pada lembar kerja Maple adalah:

```
solve(ruas kiri = ruas kanan, x);
```

dimana x adalah variabel yang dicari harganya.

Contoh 4.1.1. Carilah nilai x dari persamaan $3x + 7 = 4$.

```
[> solve(3*x+7=4,x);
```

–1

Untuk persamaan yang hanya terdiri dari satu variabel, penulisan `solve(3*x+7=4,x)` dan `solve(3*x+7=4)` akan memberikan hasil yang sama. Perintah `solve` ini dapat diaplikasikan untuk mencari akar dari sebuah persamaan polynomial. Hal ini diperlihatkan pada contoh [4.1.2](#), dibawah ini.

Contoh 4.1.2. Carilah nilai x yang memenuhi persamaan $x^2 + 3x - 10 = 0$.

Pada lembar kerja Maple dituliskan,

```
[> solve(x^2+3*x-10=0);
```

2, -5.

Penyelesaian $x^2 + 3x - 10 = 0$ memberikan hasil $x = 2$ dan $x = -5$.

Contoh 4.1.3. Carilah nilai x untuk persamaan

$$\frac{x^2 - 1}{x - 1} = 0.$$

Penyelesaian dengan Maple adalah :

```
[> solve((x^2-1)/(x-1)=0,x);
```

-1.

Maple dengan perintah `solve`-nya dapat digunakan untuk mencari akar - akar sebuah persamaan yang mengandung bilangan imajiner. Bilangan imajiner $\sqrt{-1}$, disimbolkan oleh Maple sebagai I .

Contoh 4.1.4. Carilah nilai x untuk persamaan

$$x^3 + x^2 + x + 1 = 0.$$

```
[> solve(x^3+x^2+x+1=0);
```

-1, I , $-I$.

Terlihat bahwa persamaan $x^3 + x^2 + x + 1 = 0$ mempunyai 3 akar, yaitu satu akar bilangan *real*, (-1) dan dua akar bilangan imajiner (I dan $-I$). Perintah `solve` seterusnya dapat dipakai untuk memecahkan sistem persamaan yang terdiri dari n persamaan dengan n bilangan yang tidak diketahui. Bentuk dari perintah `solve` untuk menyelesaikan sistem persamaan adalah :

```
solve([persamaan1,persamaan2 ...],{x,y, ...})
```

Dimana x, y, \dots adalah variabel yang tidak diketahui, dan hendak dicari harganya.

Contoh 4.1.5. Carilah himpunan penyelesaian dari dua persamaan dibawah ini :

$$\begin{aligned} 2x + 3y &= 7 \\ 4x - 5y &= 3. \end{aligned}$$

Persoalan diatas, mempunyai dua persamaan dengan dua bilangan yang tidak diketahui, x dan y . Untuk memecahkan persamaan tersebut dengan Maple, pada lembar kerja Maple dimasukkan :

```
[> solve({2*x+3*y = 7, 4*x-5*y = 3},{x,y});
```

$$x = 2, y = 1$$

Terlihat bahwa, sistem persamaan di atas menghasilkan solusi $x = 2$ dan $y = 1$.

Catatan 4.1.1. Perhatikan kedua persamaan tersebut ditulis didalam tanda kurung ($\{\dots\}$) dan dipisahkan oleh tanda koma (.). Penulisan $\{x, y\}$ dimaksudkan bahwa kita mencari nilai x dan y untuk kedua persamaan itu.

Perintah **solve** pada Maple dapat digunakan untuk memecahkan persamaan dalam bentuk simbolik, seperti yang diperlihatkan pada contoh dibawah ini:

Contoh 4.1.6. Carilah harga h sebagai fungsi r dari persamaan $v = \pi r^2/h$.

```
[> solve(v=Pi*r^2/h,h);
```

dan hasilnya :

$$\frac{\pi r^2}{v}.$$

Contoh 4.1.7. Tentukan nilai a pada persamaan $a^2 + b^2 = c^2$.

Sama seperti contoh 4.1.6, masukkan perintah dibawah ini pada lembar kerja Maple, kemudian perhatikan hasilnya.

```
[>> solve(a^2+b^2=c^2,a);
```

$$\sqrt{-b^2 + c^2}, -\sqrt{-b^2 + c^2}.$$

Contoh 4.1.8. Carilah nilai x dari persamaan dibawah ini :

$$ax^2 = 4,$$

Bilamana variabel yang hendak dicari adalah x , maka penulisan perintahnya adalah :

```
[> solve(a*x^2=4, {x});
```

$$\left\{x = \frac{2}{\sqrt{a}}\right\}, \left\{x = -\frac{2}{\sqrt{a}}\right\}.$$

Namun bilamana variabel yang dicari adalah a , maka pada lembar kerja Maple ditulis,

```
[> solve(a*x^2 = 4, {a});
```

hasilnya adalah :

$$\frac{4}{x^2}.$$

Contoh 4.1.9. Carilah nilai x dan y dalam bentuk a, b, c, d dari sistem persamaan dibawah ini:

$$\begin{aligned} ax + by &= c \\ dx + ey &= f \end{aligned} \tag{4.1}$$

Harga x dan y dapat ditentukan secara simbolik, seperti berikut ini :

```
[> solve({a*x+b*y=c, d*x+e*y=f}, {x,y})
```

$$\left\{ x = -\frac{-ec + fb}{-db + ae}, y = \frac{-dc + af}{-db + ae} \right\}$$

Sekarang akan dilihat, bagaimana perintah `solve` ini dipakai untuk memecahkan persamaan Trigonometri. Perhatikan contoh dibawah ini :

Contoh 4.1.10. Carilah harga x yang memenuhi persamaan : $\sin(x)^2 - 2\sin(x) - 3 = 0$.

```
[> solve(sin(x)^2 - 2*sin(x)-3=0,x);
```

$$\arcsin(3), -\frac{1}{2}\pi.$$

4.2 Perintah `fsolve`

Sampai sejauh ini penggunaan perintah `solve` yang telah dilihat diatas menghasilkan bilangan real dan simbolik. Namun bilamana solusi yang diinginkan adalah bilangan desimal, maka perintah yang dipakai : `fsolve`. Perintah ini tidak dapat digunakan untuk menyelesaikan persamaan secara simbolik. Nilai yang didapat dari perhitungan menggunakan `fsolve`, adalah pendekatan. Bentuk penulisan perintah ini adalah :

```
fsolve(ruas kiri = ruas kanan, x).
```

Dimana x adalah variabel yang akan dicari nilainya.

Contoh 4.2.1. Carilah harga x yang memenuhi persamaan $x^2 - x = 2025$. Gunakan perintah `solve` dan `fsolve` kemudian bandingkan hasilnya.

Dengan memakai perintah `solve`,

```
[> solve(x^2 - x = 2025);
```

hasilnya adalah :

$$\frac{1}{2} + \frac{\sqrt{8101}}{2}, \quad \frac{1}{2} - \frac{\sqrt{8101}}{2}.$$

Bandingkan sekarang, bilamana persamaan tersebut dipecahkan dengan perintah `fsolve`,

```
[> fsolve(x^2 - x = 2025);
```

maka hasilnya adalah

$$-44.50277769, \quad 45.50277769$$

Sekarang akan dilihat bagaimana perintah `fsolve` dapat digunakan untuk memecahkan sistem persamaan.

Contoh 4.2.2. Carilah nilai x dan y untuk sistem persamaan dibawah ini :

$$\begin{aligned}x^2 - 3xy - 4y^2 &= 0 \\ 3x^2 + 5xy - 19y^2 &= 49\end{aligned}$$

Untuk menghindari penulisan persamaan yang berulang, maka pertama kali kedua persamaan tersebut dimasukkan dalam satu variabel. Pada contoh diatas, kedua persamaan tersebut diberi nama f dan g .

```
[> f := x^2 - 3*x*y - 4*y^2 = 0;
```

$$f := x^2 - 3xy - 4y^2 = 0$$

```
[> g := 3*x^2 - 5*x*y - 19*y^2 = 49;
```

$$g := 3x^2 - 5xy - 19y^2 = 49$$

Dengan perintah `solve`,

```
solve({f,g},{x,y});
```

hasil yang diperoleh adalah :

$$\begin{aligned}\{y = 7\text{RootOf}(11Z + 1, \text{label} = L1), \\ x = -7\text{RootOf}(11Z + 1, \text{label} = L1)\}, \{y = 7/3, x = 28/3\}, \\ \{y = -7/3, x = -28/3\}.\end{aligned}$$

Sekarang bandingkan, penyelesaian dengan perintah `fsolve`.

```
[>fsolve({f,g},{x,y});
```

kita akan dapatkan :

$$\{x = -9.333333333, y = -2.333333333\}.$$

Contoh 4.2.3. Carilah nilai x dan y dari sistem persamaan dibawah ini :

$$\begin{aligned}3x^4y^2 &= 17 \\ x^3y - 5xy^2 - 2y &= 1\end{aligned}$$

Ada beberapa cara untuk memecahkan masalah tersebut diatas, cara pertama adalah setiap persamaan dimasukkan kedalam satu variabel. kemudian dipecahkan. Cara kedua adalah, kedua persamaan tersebut dimasukkan kedalam satu variabel dan kemudian dipecahkan.

```
[>f:=3*x^4*y^2=17:
[>g:=x^3*y-5*x*y^2-2*y=1:
[>fsolve({f, g} );
```

$$\{x = 2.118203038, y = .5305528603\}$$


Atau cara lain, kedua persamaan itu didefinisikan dahulu secara bersama,

```
[>polynomials:={3*x^4*y^2=17, x^3*y-5*x*y^2-2*y=1}:
[>fsolve(polynomials );
```

$$\{x = 2.118203038, y = .5305528603\}.$$

Contoh 4.2.4. Carilah penyelesaian dari $10x^5 - 30x + 10 = 0$ dengan batasan dari $x = -1.5$ sampai $x = 1.5$

Jika persamaan polynomial diatas digambarkan (cara membuat grafik akan dibahas pada bab selanjutnya), hasilnya dapat dilihat pada gambar 4.1.



Gambar 4.1: Gambar dari fungsi $10x^5 - 30x + 10$.

Dari gambar 4.1, dapat diketahui bahwa fungsi $10x^5 - 30x + 10$ memotong sumbu x di beberapa titik. Untuk mencari nilai titik potong dengan batasan $-1.5 \leq x \leq 1.5$, maka perintah `fsolve` diberi perintah tambahan $x = 1 \dots 1.5$, seperti yang terlihat pada pernyataan dibawah ini,

```
[>fsolve(10*x^5-30*x+10=0, x=1..1.5) ;
```

$$1.214648043.$$

Sama seperti perintah `solve`, perintah `fsolve` dapat diaplikasikan juga kedalam persamaan Trigonometri. Hasil yang diperoleh dari pemecahan persamaan trigonometri adalah dalam bentuk radian.

Contoh 4.2.5. Hitunglah nilai x untuk persamaan $\cos(x) - x = 0$.

```
[>fsolve(cos(x)-x=0, x);
```

$$\{x = 0.7390851332\}$$

4.3 Menyelesaikan Pertidaksamaan

Perintah `solve` dan `fsolve`, selain dapat digunakan untuk memecahkan persamaan, dapat juga digunakan untuk memecahkan pertidaksamaan. Dalam Maple ada sejumlah simbol yang dipakai untuk menyatakan pertidaksamaan. Simbol - simbol tersebut ditampilkan pada tabel 4.1.

Simbol Matematika	Simbol Maple	Arti
$<$	<code><</code>	lebih kecil
\leq	<code><=</code>	lebih kecil sama dengan
$>$	<code>></code>	lebih besar
\geq	<code>>=</code>	lebih besar sama dengan
\neq	<code><></code>	tidak sama dengan

Tabel 4.1: Simbol yang digunakan Maple untuk pertidaksamaan.

Contoh 4.3.1. Carilah harga - harga x yang memenuhi pertidaksamaan $x^3 - 2x^2 - x + 2 \geq 0$.

Kita memakai perintah `solve` untuk menyelesaikan pertidaksamaan diatas. Dalam lembar kerja Maple, kita tuliskan :

```
[>solve(x^3-2*x^2-x + 2 >= 0);
```

Dari Maple kita akan memperoleh jawaban :

$$RealRange(-1, 1), RealRange(2, \infty).$$

Jawaban Maple tersebut bearti nilai x yang memenuhi pertidaksamaan adalah x antara -1 dan 1 $-1 \leq x \leq 1$ dan x lebih besar sama dengan 2 , $x \geq 2$.

Contoh 4.3.2. Carilah domain $f(x)$ dari :

$$f(x) = \frac{\sqrt{-x^4 + 4x^3 + 4x^2 - 16x}}{x^2 - 2x - 3}$$

Ada dua syarat untuk mencari domain dari fungsi diatas yaitu:

1. Yang pertama adalah pembilang harus lebih besar sama denngan nol (0),
 $-x^4 + 4x^3 + 4x^2 - 16x \geq 0$.
2. Yang kedua adalah penyebut dari fungsi tersebut tidak sama dengan nol (0),
 $x^2 - 2x - 3 \neq 0$.

Untuk memecahkan kedua pertidaksamaan tersebut, kita tuliskan pada lembar kerja Maple :

```
[> solve( {-x^4+4*x^3+4*x^2-16*x>=0, x^2-2*x-3<>0}, x);
```

Perhatikan kedua pertidaksamaan tersebut dituliskan dalam tanda kurung kurawal {...}. Hasil dari perhitungan tersebut adalah :

$$\{-2 \leq x, x < -1\}, \{-1 < x, x \leq 0\}, \{2 \leq x, x < 3\}, \{3 < x, x \leq 4\}$$

4.4 Soal - Soal dan Penyelesaiannya

Dibawah ini diperlihatkan contoh - contoh persoalan matematika yang dapat diselesaikan oleh Maple.

1. Tentukan himpunan penyelesaian dari persamaan berikut ini :

$$x + \frac{3}{x} = \frac{3-2x}{x}$$

```
[> solve(x+3/x = (3-2*x)/x, x);
```

-2.

2. Tentukan himpunan penyelesaian dari persamaan berikut ini:

$$\sqrt{2x-3} - \sqrt{x-1} = 0.$$

```
[> solve(sqrt(2*x-3)-sqrt(x-1));
```

2.

3. Tentukan semua himpunan penyelesaian dari persamaan berikut ini :

$$\sqrt{x} + \sqrt{2x-2} - \sqrt{5x+4} = 0$$

```
[> solve(sqrt(x)+sqrt(2*x-2)-sqrt(5*x+4)=0, x);
```

9.

4. Carilah semua himpunan penyelesaian dari persamaan dibawah ini,

$$2x^4 + x^3 - 11x^2 + x + 2 = 0.$$

```
[>solve(2*x^4+x^3-11*x^2+x+2=0, x);
```

$$\frac{1}{2}, 2, \left(\frac{\sqrt{5}}{2} - \frac{3}{2}\right), \left(-\frac{3}{2} - \frac{\sqrt{5}}{2}\right).$$

5. Carilah nilai x_1 dan x_2 dari persamaan kuadrat $ax^2 + bx + c = 0$.

```
[> solve(a*x^2+b*x+c=0, x);
```

Hasilnya seperti kita ketahui bersama yaitu :

$$-\frac{b - \sqrt{b^2 - 4ac}}{2a}, -\frac{b + \sqrt{b^2 - 4ac}}{2a};$$

6. Dalam Fisika, persamaan gas ideal dapat dinyatakan sebagai :

$$PV = nRT.$$

Dimana P adalah tekanan gas [Pa], V adalah volume [m^3], n adalah jumlah mol [kilomol], R konstanta gas universal [J/kilomol.K], dan T adalah temperatur mutlak dalam kelvin [K]. Berapakah nilai T sebagai fungsi dari tekanan dan volume. Kita definisikan dahulu $PV = nRT$ sebagai *pers1*

```
[>pers1 := P*V = n*R*T;
```

$$pers1 := PV = nRT;$$

Untuk mencari harga T dari persamaan diatas, maka kita masukkan :

```
[>solve(pers1, T);
```

dan hasilnya adalah :

$$\frac{PV}{nR};$$

7. Carilah x dan y dari $x + y = 7$ dan $x^2 + y^2 = 29$ Penyelesaian soal ini adalah :

```
[>solve({x+y = 7, x^2+y^2=29}, {x, y});
```

$$\{x = 5, y = 2\}, \{y = 5, x = 2\}$$

8. Harga harga x mana saja yang memenuhi pertidaksamaan berikut :

$$\begin{aligned}10 - 3x &< 10 \\ 2x - 5 &< 10\end{aligned}$$

Penyelesaiannya adalah :

```
[>solve({10 - 3*x <10, 2*x-5 <10},x);
```

$$0 < x, x < \frac{15}{2}$$

9. Carilah penyelesaian dari persamaan ini :

$$x^2 = -4.$$

```
[>solve(x^2 = -4);
```

$$2I, -2I.$$

Hasil dari persamaan tersebut adalah bilangan imajiner.

10. Bandingkan hasilnya antara menggunakan perintah solve dan fsolve untuk memecahkan persamaan ini : $\cos(x) = 0$.

Dengan menggunakan solve :

```
[> solve(cos(x)=0,x);
```

$$\frac{\pi}{2}.$$

Dengan menggunakan perintah fsolve:

```
[> fsolve(cos(x)=0,x);
```

$$1.570796327.$$

4.5 Soal soal latihan

Buatlah soal - soal dibawah ini dengan cara manual, lalu buat juga dengan Program Maple. Kemudian Bandingkan hasilnya.

1. Carilah nilai x agar memenuhi sistem persamaan ini : $x^2 + y^2 = 10$ dan $y = 3x$.
2. Dari sistem persamaan linear dibawah ini :

$$\begin{aligned}a + 3b + c &= 10 \\2a - b + 72c &= 20 \\7a - 3b + 21c &= 30\end{aligned}$$

hitunglah nilai a, b, c agar memenuhi persamaan diatas.

3. Carilah nilai x agar memenuhi persamaan ini : $x - 1 = \sqrt{1 - 2x}$.
4. Tentukan himpunan penyelesaian x untuk pertidaksamaan dibawah ini :
 - (a) $2x + 6 \leq x + 10 \leq 2x + 1$
 - (b) $\frac{x+2}{2x+3} > 0$
 - (c) $\frac{x+1}{x+3} < \frac{x+5}{x+6}$
 - (d) $x + \frac{1}{x} < 2$
5. Grafik $y = ax^2 + bx + c$, melalui titik - titik : $(0,1)$, $(2,5)$, dan $(-1,4)$. Tentukanlah a, b , dan c .
6. Carilah titik potong dari $y = x^2$ dan $y = 8 - x^2$.
7. Carilah semua akar- akar dari persamaan ini :

$$x^3 - 5x^2 - 2x + 10 = 0.$$

Bab 5

Membuat Grafik fungsi

Dalam memahami matematika cara yang terbaik adalah bilamana persoalan tersebut dapat divisualkan. Dengan Program Maple, berbagai jenis grafik mulai dari dua dimensi sampai grafik tiga dimensi dapat dibuat dengan mudah. Maple dapat membuat grafik dari berbagai macam koordinat sistem, seperti cartesian, polar, silinder, dan sebagainya. Disamping kemampuannya membuat grafik, program Maple dapat membuat grafik yang beranimasi. Hasil grafik yang dibuat oleh program Maple ini dapat digunakan (*export*) oleh program lain seperti : Word, Power Point, dan sebagainya.


5.1 Grafik Dimensi Dua

Bilamana ada sebuah fungsi yang mempunyai satu variabel bebas, $y = f(x)$, hendak digambar (di-plot) dengan batas $[a,b]$, maka perintah Maple yang dipakai adalah `plot`. Kaedah penulisan perintah ini adalah :

```
plot(f(x), x=a..b);
```

Contoh 5.1.1. Gambarlah fungsi $y = \cos(x) - (1/3) \cos(3x)$ dengan batas $-\pi \leq x \leq \pi$.

```
[> plot(cos(x)-(1/3)*cos(3*x), x = -Pi..Pi );
```



Gambar 5.1: Grafik $y = \cos(x) - (1/3) \cos(3x)$ dengan interval $[-\pi, \pi]$

Hasil dari perintah diatas, dapat dilihat pada gambar 5.1. Bila batasan grafik $[a, b]$ tidak diberikan kepada Maple, maka Maple akan memberi batas secara otomatis (*default*) yakni dari -10 sampai 10 .


Contoh 5.1.2. Gambarlah dengan Maple fungsi,

$$f(t) = \frac{\sin(t)}{(1 + \exp(t)^2)}$$

sebagai fungsi dari waktu (t).

```
[>plot(sin(t)/(1 + exp(t)^2), t);
```

Dalam persoalan diatas, pada lembar kerja Maple tidak diberikan batas interval (t), Maple akan menggunakan nilai *default*-nya, yakni t dari -10 sampai 10 . Hasil plot tersebut dapat dilihat pada gambar 5.2.



Gambar 5.2: Grafik $f(t) = \frac{\sin(t)}{(1+\exp(t)^2)}$, sebagai fungsi dari t .

5.1.1 Fasilitas Tambahan Grafik pada Maple

Dalam pembuatan grafik, Maple menyediakan beberapa fasilitas tambahan (*option*). Hal ini sangat membantu agar tampilan grafik menjadi lebih representatif. Fasilitas - fasilitas tersebut antara lain :

1. Warna, untuk memberi warna pada grafik, digunakan perintah `color`. Bentuk penulisan perintah `color`, adalah

```
color = nama warna
```

Nama - nama warna yang dikenal oleh Maple antara lain:

```
black   blue   cyan   green   gray   magenta maroon orange
navy    pink   red    sienna  violet wheat  white  yellow
```

Sebagai contoh bila grafik yang diinginkan berwarna merah, maka perintah opsional cukup ditambahkan `color = red`.

2. Ketebalan garis dalam membuat grafik dapat diatur dengan menggunakan perintah `thickness`. Kaedah penulisan perintah `thickness` adalah :

```
thickness = n.
```

Dimana n adalah bilangan bulat yang menyatakan tebal garis. Semakin besar harga n ketebalan garis pun akan semakin tebal.

3. Ukuran grafik, yang muncul dapat dengan bebas didefinisikan dengan mengatur jendela pandangnya. Perintah yang digunakan untuk mengatur besarnya tampilan grafik adalah `view`. Bentuk perintah `view` adalah :

```
view=[xmin..xmax, ymin..ymax].
```

Dimana `xmin` dan `xmax` adalah harga minimum dan maksimum sumbu horizontal. Sedangkan `ymin` dan `ymax` adalah batas minimum dan maksimum sumbu vertikal.

4. Sebuah grafik dapat diberi judul (`title`) dengan perintah `title`. Yang perlu diperhatikan dalam pemberian nama grafik adalah; nama grafik harus ditulis antara dua tanda petik ganda. Bentuk penulisan perintah ini :

```
title = " Nama Grafik".
```

5. Maple menyediakan fasilitas legenda pada grafik yang dikenal dengan perintah `legend`. Sama seperti `title`, nama legend juga harus diapit oleh dua tanda petik ganda (" nama legend "). Kaedah penulisan `legend` adalah seperti berikut :

```
legend = "nama legend".
```

6. Pengguna Maple dapat juga memilih bentuk- bentuk garis yang dipakai oleh grafiknya. Perintah Maple yang dipakai untuk memilih jenis - jenis garis adalah `linestyle`. Format penulisan perintah `linestyle` adalah sebagai berikut :

```
linestyle = bentuk garis.
```

Bentuk bentuk garis yang dikenal oleh Maple antara lain : **solid** (untuk garis biasa—), **dot** (untuk titik titik· · ·), **dash** (untuk garis putus putus— — —), atau **dashdot** (untuk titik garis. — · — · —).

7. Selain dengan garis, grafik dapat juga diplot dalam bentuk simbol. Perintah yang dipakai untuk mengerjakan hal ini adalah `symbol`. Bentuk penulisan perintah `symbol` adalah :


symbol = bentuk simbol.

Bentuk - bentuk simbol yang dikenal Maple adalah : **circle** (\circ), **cross** (\times), **diamond** (\diamond), atau **point**(\bullet).

Penggunaan fasilitas - fasilitas tambahan (*option*) ini akan diilustrasikan pada contoh berikut ini.

Contoh 5.1.3. Gambarlah grafik fungsi $y = \sin(x)$ dengan $-2\pi \leq x \leq 2\pi$. Buatlah grafik sinus tersebut dengan warna hitam, menggunakan *garis titik (dashdot)*. Berilah judul gambar tersebut "Grafik Sinus", dan pakailah nama untuk legenda $\sin(x)$, dan ketebalan garis = 3.

```
[>plot(sin(x),x=-2*Pi..2*Pi,color=black,thickness=3,
title="Grafik Sinus",[linestyle=DASHDOT,legend=["sin(x)"]);
```



Gambar 5.3: Satu fungsi $y = \sin(x)$ diplot dengan menggunakan berbagai fasilitas dari Maple.

5.1.2 Grafik Fungsi Parametrik


Yang dinamakan fungsi parametrik adalah, bilamana dalam sebuah fungsi variabel x dan y tergantung dari variabel lain misalnya t . Variabel t yang menjadi variabel bebas dari variabel x dan y dinamakan *parameter*, sedang fungsi atau persamaannya dinamakan fungsi parametrik. Untuk menggambarkan fungsi ini, perintah `plot` pada Maple, masih dapat digunakan. Bentuk perintah `plot` untuk fungsi parametrik ini adalah :

```
plot([ekspresi-x, ekspresi-y, range parameter]).
```

Contoh 5.1.4. Gambarkanlah sebuah fungsi parametrik dengan $x = t^2$, dan $y = t^3$ dengan $-1 \leq t \leq 1$

```
[>plot([t^2, t^3, t = -1..1]);
```

Hasil dari perintah diatas diperlihatkan pada gambar 5.4(a).



Gambar 5.4: Dua buah fungsi parametrik. (a) $x = t^2$, $y = t^3$. (b). $x = \sin(t)$, $y = \sin(2t)$.

Contoh 5.1.5. Sebuah persamaan parametrik dengan

$$x = \sin(t) \quad y = \sin(2t), \quad 0 \leq t \leq 2\pi$$

Buatlah grafik gambar diatas.

```
[>plot([sin(t), sin(2*t), t = 0..2*Pi]);
```

Hasilnya dapat dilihat pada gambar 5.4(b).

5.1.3 Fungsi yang Mempunyai Nilai Singular


Hampir semua fungsi pada prinsipnya dapat digambarkan oleh Maple. Namun untuk menggambarkan fungsi yang mempunyai nilai singular, memerlukan teknik khusus untuk menampilkan-nya. Jika tidak maka gambar yang diperoleh terkadang menjadi sulit untuk dimengerti.

Contoh 5.1.6. Gambarkanlah grafik dari : $\frac{1}{(x-1)^2}$ dengan batas $-5 \leq x \leq 6$.

```
[>plot( 1/(x-1)^2, x=-5..6 );
```

Fungsi $\frac{1}{(x-1)^2}$ mempunyai nilai singular di titik $x = 1$. Sebab pada saat x mendekati 1 nilai fungsi menjadi sangat besar. Hal ini diperlihatkan pada gambar 5.5(a). Gambar ini sulit dimengerti, karena tampilan gambar tersebut hanya berupa garis lurus pada titik $x = 1$.

Tampilan grafik $\frac{1}{(x-1)^2}$ akan menjadi menarik, jika daerah hasilnya, yakni sumbu y diberi batasan, seperti dibawah ini.



Gambar 5.5: Fungsi $f(x) = \frac{1}{(x-1)^2}$ yang mempunyai nilai singular pada $x = 1$. Tampilan grafik sebelum sumbu y diberi batas (a), dan setelah sumbu y diberi batas (b).

```
[> plot( 1/(x-1)^2, x=-5..6, y = -1..7 );
```

Setelah sumbu y diberi batasan yakni $-1 \leq y \leq 7$, grafik $\frac{1}{(x-1)^2}$ lebih mudah dipahami, lihat gambar 5.5(b).

5.1.4 Menampilkan lebih dari satu grafik dalam satu gambar

Dalam persoalan matematika seringkali dijumpai dimana beberapa grafik dari beberapa fungsi perlu ditampilkan dalam satu gambar secara bersamaan. Untuk keperluan ini, perintah `plot` tetap digunakan, hanya perlu dimodifikasi sedikit cara penulisannya :

```
plot([f(1), f(2) ...], x = x1..x2, y = y1..y2)
```

$f(1), f(2), \dots$ adalah fungsi -fungsi yang akan di-plot. Perhatikan pernyataan diatas, beberapa fungsi yang hendak di-plot diletakkan bersama dalam tanda kurung siku ([. . .]). $x1$ dan $x2$ menyatakan batasan pada sumbu x sedang $y1$ dan $y2$ batasan untuk sumbu y .


Contoh 5.1.7. Tentukan titik potong dari persamaan fungsi kuadrat $y = x^2 + 3x - 20$, dan persamaan linear $y = 2x + 40$.

Titik potong kedua kurva tersebut dapat dicari dengan dua metode yaitu secara analitis dan grafis. Secara grafis, persoalan tersebut dipecahkan dengan mem-plot kedua fungsi tersebut dalam satu gambar, secara bersamaan. Hasil dari metode grafik diperlihatkan pada gambar 5.6.

```
[> plot([x^2+3*x-20, 2*x+40], x=-15..10, y=-30..120);
```

Untuk mengetahui titik potong kedua kurva tersebut secara eksak, Maple menyediakan perintah `solve` atau `fsolve`. Kedua perintah ini telah dibahas pada bab 4.2.

```
[> fsolve(x^2+3*x-20=2*x+40, {x});
```



Gambar 5.6: Fungsi $y = x^2 + 3x - 20$ dan fungsi $y = 2x + 40$ digambarkan dalam satu grafik. Secara visual dapat diperkirakan lokasi titik potong kedua fungsi tersebut.

$$\{x = -8.262087348\}, \{x = 7.262087348\}$$

Kedua kurva tersebut memotong pada dua titik yaitu $x = -8.262087348$ dan $x = 7.262087348$. Harga y dapat ditentukan dengan mensubstitusi kedua harga x tersebut ke salah satu persamaan di atas. Untuk keperluan ini, dipakai perintah `subs`, yang juga telah dibahas pada bab sebelumnya.

```
[>subs(x=-8.262087348, 2*x+40),
```

23.47582530.

```
[>subs(x=7.262087348, 2*x+40),
```

54.52417470.

Setelah dibulatkan sampai dua desimal, titik potong kedua kurva tersebut berada pada koordinat (x, y) , $(-8.26, 23.47)$ dan $(7.26, 54.52)$. Hasil perhitungan analitik ini dapat dibandingkan dengan gambar 5.6.


Cara lain untuk menggabungkan beberapa grafik dalam satu gambar adalah memakai perintah **display**. Perintah ini berada dalam paket `plots`. Jadi sebelum menggunakan perintah **display**, paket `plot` harus dipanggil terlebih dahulu dengan, **with(plots)**. Setelah paket **plots** dipanggil, lembar kerja Maple akan menampilkan seperti dibawah ini :

[Interactive, animate, animate3d, animatecurve, arrow, changecoords, complexplot, complexplot3d, conformal, conformal3d, contourplot, contourplot3d, coordplot, coordplot3d, cylinderplot, densityplot, display,

display3d fieldplot, fieldplot3d, gradplot, gradplot3d, graphplot3d, implicitplot, implicitplot3d, inequal, interactive, interactiveparams, listcontplot, listcontplot3d, listdensityplot, listplot, listplot3d, loglogplot, logplot, matrixplot, multiple, odeplot, pareto, plotcompare, pointplot, pointplot3d, polarplot, polygonplot, polygonplot3d, polyhedra_supported, polyhedraplot, replot, rootlocus, semilogplot, setoptions, setoptions3d, spacecurve. sparsematrixplot, sphereplot, surfdata, textplot, textplot3d, tubeplot].

Inilah fungsi - fungsi tambahan yang terdapat dalam paket **plots**. Perintah `display` terlihat ada pada salah satu fungsi - fungsi tambahan tersebut.

Contoh 5.1.8. Plot ke tiga fungsi ini: $y = \sin(x)$, $y = \cos(x)$, $y = \tan(x)$ dalam satu grafik. Interval domain ketiga fungsi tersebut adalah $-\pi \leq x \leq \pi$.



Gambar 5.7: Tiga fungsi yang diplot dalam satu gambar, untuk range dari $-\pi$ sampai π .

Langkah pertama yang dilakukan adalah mendefinisikan setiap fungsi kedalam bentuk variabel.

1. $p1 = \text{hasil plot untuk } \sin(x)$;
2. $p2 = \text{hasil plot untuk } \cos(x)$;
3. $p3 = \text{hasil plot untuk } \tan(x)$.

Kemudian ketiga hasil plot tersebut ditampilkan bersamaan dengan perintah `display`. Agar penampilan gambar lebih menarik, ketiga kurva tersebut akan dibedakan propertis-nya. Kurva pertama bewarna hitam, kurva kedua juga bewarna hitam, namun menggunakan garis putus - putus. Sedangkan kurva ketiga memakai warna abu - abu.


```
[>with(plots):
[>p1:=plot(sin(x),x=-Pi..Pi,color=black):
[>p2:=plot(cos(x),x=-Pi..Pi,linestyle=DASH,color=black):
[>p3:=plot(tan(x),x=-1..1,color=gray):
[>display(p1,p2,p3,view=[-Pi..Pi,-Pi..Pi], scaling=constrained);
```

Hasil dari grafik ini dapat dilihat pada gambar 5.7.

5.1.5 Mengekspor gambar

Gambar atau grafik yang dibuat oleh Maple dapat digunakan oleh program lain seperti *Power Point* atau *Word*. Cara yang paling mudah untuk menggunakan gambar tersebut pada program lain adalah dengan mengkopi gambar tersebut kemudian mem-*paste* pada program lain seperti; *Word* dan *Power Point*.

Hasil gambar yang dibuat oleh Maple dapat juga disimpan dalam bentuk file gambar tersendiri. Caranya adalah : dekatkan mouse pointer ke gambar tersebut, lalu klik kanan, pilih **Export as**, lalu pilih format gambar seperti *jpeg*, *eps*, *gif*, *bmp*, dan sebagainya, lihat gambar 5.8. Terakhir, beri nama file tersebut, dan simpan pada directory yang diinginkan.




Gambar 5.8: Tampilan menu yang keluar setelah gambar tersebut diklik kanan.

5.1.6 Koordinat Polar

Sampai sejauh ini contoh - contoh soal grafik masih menggunakan koordinat cartesian (x, y), untuk menggambarkan sebuah fungsi. Sistem koordinat lain yang sering digunakan dalam persoalan matematika adalah koordinat polar. Pada sistem koordinat ini, posisi sebuah titik dinyatakan oleh r dan θ . Dimana r adalah jarak dari pusat koordinat ke-titik yang dimaksud. Sedangkan θ adalah besarnya sudut yang diukur secara berlawanan arah jarum jam (*counterclockwise*) antara sumbu x dan garis yang melalui titik dan pusat koordinat, lihat gambar 5.9.

Secara matematik relasi antara koordinat polar dan cartesian dapat dinyatakan :

$$x = r \cos(\theta) \quad y = r \sin(\theta).$$




Gambar 5.9:

Atau sebaliknya yaitu :

$$r = \sqrt{x^2 + y^2} \quad \theta = \tan^{-1}\left(\frac{y}{x}\right).$$

Program Maple, selain dapat membuat grafik pada koordinat cartesian (x, y) , Maple juga dapat membuat grafik pada koordinat polar (r, θ) . Perintah yang dipakai oleh Maple untuk menggambar fungsi dalam koordinat polar adalah sama yakni `plot`. Hanya pada *option* perintah `plot` ditambahkan `coords = polar`, sebagai pertanda kalau grafik tersebut diplot dalam sistem koordinat polar.

Contoh 5.1.9. Gambarlah fungsi $y = \sin(x)$, dalam koordinat cartesian dan polar, untuk interval $x = 0..2\pi$.



Gambar 5.10: Fungsi $\sin(x)$ digambarkan dalam koordinat cartesian (a), sedang pada (b) terlihat $\sin(x)$ dinyatakan dalam koordinat polar.

Perintah yang digunakan pada Maple untuk menggambarkan fungsi tersebut dalam koordinat cartesian adalah :

```
[>plot(sin(x),x=0..2*Pi,thickness=3,color=black);
```

Hasil plot fungsi ini dapat dilihat pada gambar 5.10(a). Jika fungsi $y = \sin(x)$ tersebut ingin digambarkan dalam koordinat polar maka pada lembar kerja Maple dimasukkan :

```
[> plot(sin(x),x=0..2*Pi, coords=polar, thickness=3, color=black);
```


Perhatikan perintah `coords=polar`, yang dipakai oleh Maple untuk menandakan fungsi tersebut di-plot dalam koordinat polar. Hasil dari grafik ini dapat dilihat pada gambar 5.10(b).

Contoh 5.1.10. Gambarlah dalam koordinat polar dari :

$$y = \sin(4x)$$

untuk batas $x = 0$ sampai dengan 2π

```
[> plot([sin(4*x),x,x=0..2*Pi],coords=polar,thickness=3);
```



Gambar 5.11: Grafik $\sin(4x)$ dalam koordinat polar.

Contoh 5.1.11. Buatlah grafik dari fungsi - fungsi dibawah ini dalam koordinat kutub (polar). Untuk mudahnya berilah harga $a = 5$ dan $0 \leq \theta \leq 2\pi$.

1. $r = a + a \cos(\theta)$
2. $r = a + 2a \cos(\theta)$
3. $r^2 = a^2 \cos(2\theta)$
4. $r = a\theta$, untuk $\theta = 0$ sampai 10π .


Untuk soal no.1 :

```
[>plot(5+5*cos(theta),theta=0..2*Pi,coords=polar,thickness=3,color=black);
```

Hasilnya terlihat seperti gambar(5.12a). Untuk soal no.2 :

```
[>plot(5+10*cos(theta),theta=0..2*Pi,coords=polar);
```

Hasilnya terlihat pada gambar (5.12b).




Gambar 5.12:

Soal no.3 dapat dikerjakan dengan mudah dengan merubah $r^2 = a^2(\cos(2\theta))$ menjadi $r = a\sqrt{\cos(2\theta)}$ terlebih dahulu. Kemudian dengan mensubstitusi $a = 5$, pada lembar kerja Maple dituliskan :

```
[> plot(5*sqrt(cos(2*theta)), theta=0..2*Pi,coords=polar);
```

Hasil dari pernyataan diatas dapat dilihat pada gambar 5.13(a). Untuk soal no.4, dapat dengan mudah dikerjakan

```
[> a:= 5:
[> plot(a*theta, theta=0..10*Pi,coords=polar);
```



Gambar 5.13:

hasilnya terlihat pada gambar 5.13(b).


Cara lain untuk menggambar grafik dalam koordinat polar adalah dengan perintah `polarplot`. Sebelum menggunakan perintah ini, paket `plots` harus dipanggil terlebih dahulu. Hal ini dikarenakan perintah `polarplot` berada pada paket `plots`. Kaedah penulisan `polarplot` adalah :

```
polarplot(r, angle).
```

Pada perintah diatas, r merepresentasikan jari - jari, sedangkan `angle` adalah sudut, θ , lihat gambar 5.9.

Contoh 5.1.12. Gambarkan dalam koordinat polar sebuah fungsi dengan $r = \sin 3\theta$ dimana θ antara 0 sampai 2π .

```
[>with(plots):
[>polarplot( sin(3*theta), theta=0..2*Pi );
```



5.2 Grafik Dimensi Tiga

Dengan Program Maple, grafik tiga dimensi dapat dibuat dengan mudah. Perintah yang digunakan untuk membuat grafik tiga dimensi adalah :

```
plot3d(f(x,y), x=a..b, y=c..d
```

Dimana $f(x, y)$ adalah fungsi yang akan digambar, dengan batas sumbu x (a, b) dan sumbu y (c, d).

Contoh 5.2.1. Gambarlah satu grafik dari sebuah fungsi yang mempunyai dua variabel seperti dibawah ini :

$$z = x \exp(-x^2 - y^2).$$

Fungsi diatas dapat divisualisasikan kedalam grafik tiga dimensi dengan sumbu x dan y sebagai bidang horizontalnya dan sumbu z sebagai bidang vertikalnya. Maple akan membuat grafik tiga dimensi tersebut bila perintah dibawah ini ditulis pada lembar kerjanya,

```
[> plot3d(x*exp(-x^2-y^2), x=-2..2, y=-2..2);
```


Dengan mengambil batas sumbu x dan y antara -2 sampai dengan 2 , grafik fungsi tersebut dapat dilihat pada gambar 5.14(a).

Sama halnya dengan grafik dua dimensi yang dapat dipakai untuk menggambar beberapa grafik dalam satu gambar. Demikian halnya dengan pembuatan grafik tiga dimensi. Dengan program Maple, beberapa fungsi tiga dimensi dapat di-plot dalam satu gambar.


Contoh 5.2.2. Dua buah fungsi, $z = 4 - x^2 - 2y^2$ dan $z = 6 - 4y$. Gambarlah kedua fungsi tersebut dalam satu gambar tiga dimensi. Dengan batas $-4 \leq x \leq 4$, dan $-3 \leq y \leq 3$.

Tuliskan kedua fungsi tersebut didalam tanda kurung kurawal ($\{...\}$), seperti terlihat pada baris perintah dibawah ini:

```
[>plot3d({4-x^2-2*y^2, 6-4*y}, x=-4..4, y=-3..3 );
```



(a)



(b)

Gambar 5.14: (a). Grafik tiga dimensi dari fungsi $z = x \exp(-x^2 - y^2)$, (b). Grafik tiga dimensi dari dua buah fungsi $z = 4 - x^2 - 2y^2$ dan $z = 6 - 4y$

Hasilnya seperti yang diperlihatkan pada gambar 5.14(b).

Sudut pandang grafik tiga dimensi diatas, dapat kita ubah - ubah. Cara mengubah sudut pandang dari grafik tersebut adalah dengan cara; letakkan kursor (tanda panah) kepada objek gambar dilayar monitor, tekan tombol kiri pada mouse dan tahan sambil "memainkan" gerakan mouse. Kita akan melihat objek tiga dimensi tersebut akan berotasi sesuai dengan gerakan mouse. Sama seperti garfik dua dimensi, grafik tiga dimensi ini juga dapat digabung. Beberapa fungsi tiga dimensi dapat di-plot dalam dalam satu gambar.

5.3 Membuat Animasi

Salah satu kelebihan Maple dibandingkan dengan program aplikasi lainnya adalah, Maple memberikan kemudahan kepada penggunanya untuk membuat animasi. Perintah yang digunakan untuk membuat animasi adalah `animate`. Perintah ini tersedia dalam paket `plots`. Fungsi yang dapat dianimasikan haruslah mempunyai dua variabel bebas, dan satu diantaranya adalah variabel waktu, yang biasa disimbolkan t . Kaedah penulisan perintah `animate` ini adalah :

```
animate(F(x,t), x=a..b, t=p..q, options );
```


$F(x, t)$ adalah fungsi yang akan dianimasikan, $x = a..b$, adalah batasan x dalam hal ini (a, b) , sedang $t = p..q$ adalah batasan waktu dari p sampai q . Pada bagian *option* ini pengguna program Maple dapat memberi perintah tambahan untuk membuat tampilan animasi lebih baik. Perintah tambahan ini pada prinsipnya sama seperti membuat grafik dua dimensi, dan ini telah dibahas dalam subbab 5.1.1.

Contoh 5.3.1. Suatu gelombang berjalan yang diformulasikan sebagai

$$y = \sin(x.t).$$

Dimana x menyatakan posisi (tempat atau kedudukan), sementara t menyatakan waktu. Animasikan gelombang berjalan tersebut, dengan batas $-4 \leq x \leq 4$ dan waktu $1 \leq t \leq 4$.

```
[> with(plots):  
[> animate([x, sin(x*t),x=-4..4], t = 1..4, frames=100);
```



Gambar 5.15: Animasi dari gelombang $\sin(x.t)$. Panel (a) memperlihatkan gelombang pada saat $t = 1$ [s], Sedangkan panel (b) memperlihatkan bentuk gelombang pada saat $t = 2$ [s], dan panel (c) mengilustrasikan gelombang pada saat $t = 4$ [s].

Pernyataan `frames=100` dalam perintah diatas, adalah menyatakan banyaknya *frames* yang diinginkan persatuan waktu. Banyaknya *frames* dapat bebas ditentukan. Agar kualitas animasi menjadi lebih baik (halus, *smooth*), maka dibutuhkan lebih banyak *frames*. Namun hal yang perlu diingat adalah, bilamana *frames* dibuat dalam jumlah banyak, maka komputer akan bekerja lebih lama dan lebih berat. Terkadang pembuatan *frames* yang banyak akan mengakibatkan komputer menjadi *hang*. Hasil dari perintah diatas ditampilkan pada gambar 5.15 (a), (b), (c).

Pada saat perintah `animate` dijalankan, animasi tersebut tidak bisa langsung dilihat. Untuk melihatnya, klik mouse pada grafik yang muncul. Saat grafik tersebut di-klik, pada menu toolbar Maple akan tampil tombol - tombol seperti yang ditunjukkan gambar 5.16.



Gambar 5.16: Menu toolbar pada Maple saat kita klik gambar yang mengandung animasi.

Tombol - tombol ini sebenarnya sangat familiar bagi yang sering menggunakan tape recorder, video player dan sebagainya. Untuk menejalankan animasi, kita klik tombol `play` ► dan untuk menghentikannya kita klik tombol `stop` ■.

5.3.1 Melihat bagaimana sebuah kurva tergambar


Mungkin ada keinginan diantara pembaca untuk memakai Maple seperti menggambar grafik sinus di papan tulis. Maple menyediakan fasilitas ini dengan perintah `animatecurve`. Dengan perintah ini proses pembuatan grafik dapat dianimasikan. Bentuk perintah `animatecurve` adalah :

```
animatecurve( f(x), x=a..b, options )
```

$F(x)$ adalah fungsi yang hendak dianimasikan dengan batasan antara a dan b . Perintah tambahan `option` sama dengan perintah tambahan untuk `plot` dan `animate` (subbab 5.1.1).

Contoh 5.3.2. Buatlah grafik $\sin(x)$ dengan batasan $0 \leq x \leq 2\pi$.

```
[> with( plots ):  
[> animatecurve( sin(x), x=0..2*Pi, color=green, frames=100 );
```



(a) $x = 1/2\pi$

(b) $x = 1.2\pi$

(c) $x = 2\pi$


Gambar 5.17: Maple memperlihatkan bagaimana proses grafik $\sin(x)$ terbuat.

Untuk melihat proses bagaimana sebuah fungsi parametrik (lihat subbab 5.4) dibuat, contoh dibawah ini akan mengilustrasikannya.

Contoh 5.3.3. Animasikan $y = \sin(t)$ dan $x = \cos(t)$ dengan batasan $t = 0$ sampai 2π .

```
[>animatecurve( [sin(t),cos(t)],t=0..2*Pi],view=[-2..2,-2..2]);
```

Gambar 5.18 memperlihatkan sebuah fungsi parametrik digambarkan dalam koordinat sumbu cartesian. Baik sumbu x dan sumbu y pada fungsi ini tergantung pada variabel lain yaitu t . Hasil yang terlihat adalah bagaimana terjadinya sebuah lingkaran.



Gambar 5.18: Animasi dari $\sin(xt) \cos(xt)$, sebagai fungsi dari waktu.

5.3.2 Mengekspor animasi

Gambar atau grafik yang dibuat oleh Maple dapat kita export untuk keperluan presentasi ataupun sebagai "figure" pada satu atikel. Proses export akan menawarkan kita pilih format gambar, antara lain : dxf, eps, gif, jpeg/jpg, pov, windows bmp, dan wmf. Sama seperti gambar, animasi pun dapat kita export dalam format gif misalnya. Langkah pertama untuk mengekspor gambar adalah dengan mengklik kanan pada daerah gambar, lalu akan keluar menu **Export as** dan keluar pilihan format - format gambar. Pilih salah satu format tersebut dan beri nama, maka gambar kita telah diekspor, dan disimpan dalam direktori yang kita inginkan.

5.4 Soal - soal dan penyelesaiannya

1. Tentukanlah titik potong dua garis lurus yaitu $y = x + 2$ dan $y = -x + 2$. Periksalah jawabannya dengan grafik. Untuk menentukan titik potong kedua garis tersebut kita memakai perintah `solve` atau `fsolve`. Bila kita masukkan perintah ini kedalam lembar kerja Maple,

```
[>fsolve({y=x+2, y = -x +2}, {x,y});
```


hasil nya adalah :

$$\{x = 0, y = 2\}.$$

Jadi titik potong kedua garis itu ada pada titik $(x, y) = (0, 2)$. Dengan memakai grafik :

```
[>plot([x+2, -x+2], x = -5..5);
```


Kita bisa melihat dari gambar diatas, bahwa titik potong kedua garis tersebut jatuh pada koordinat $(0, 2)$.



2. Suatu persamaan kuadrat $y = ax^2 + bx + c$ dimana $a \neq 0$. Sekarang marilah kita selidiki apa pengaruh besaran a , b dan c pada persamaan tersebut. Step pertama kita jadikan a , b , c sebuah variabel yang bisa bervariasi dari -2 menjadi 2, untuk itu kita perlu perintah `animate`, seperti dibawah ini.

(a) Pengaruh besaran a .

```
[>with( plots ):
[> animate( a*x^2, x=-2..2, a=-2..2 );
```



(a) $x = -2$

(b) $x = -0.25$


(c) $x = 2$

Gambar 5.19: Pengaruh harga a dari persamaan $y = ax^2 + bx + c$.

(b) Untuk melihat pengaruh dari harga b . Kita buat harga $a = 1$ dan $c = 1$, sehingga persamaan kuadrat diatas menjadi $x^2 + bx + 1$. Nilai b kita variasikan dari -2 menjadi 2.

```
[>with(plots):
[>animate( x^2+b*x+1, x=-2..2, b=-2..2 );
```

(c) Untuk melihat pengaruh c terhadap persamaan kuadrat tersebut, maka persamaan kuadrat tersebut kita ubah menjadi $y = x^2 + x + c$. Disini nilai c dapat bervariasi antara -2 sampai 2 .




(a) $b = -2$

(b) $b = -0.25$

(c) $b = 2$

Gambar 5.20: Pengaruh harga b dari persamaan $y = x^2 + bx + 1$.

```
[>with(plots):
[>animate( x^2+x+c, x=-2..2, c=-2..2 );
```



(a) $c = -2$

(b) $c = -0.25$

(c) $c = 2$


Gambar 5.21: Pengaruh harga b dari persamaan $y = x^2 + x + c$.

3. Gambarlah tiga fungsi $f(t, s)$ ini dalam satu bidang 3 dimensi.

$$\begin{aligned} x &= 2t - 3s^2 \sin(t) \\ y &= st \\ z &= 2s - 3 \cos(t) \end{aligned}$$

```
[>plot3d([2*t-3*s^2*sin(t), s*t, 2*s-3*cos(t)], s=-2..2, t=-2..2);
```

hasilnya adalah :



4. Tentukan titik potong $y = x^2 - 8x + 15$ terhadap sumbu horizontal x . Kita dapat mencarinya secara eksak dengan menggunakan perintah solve atau fsolve dari persamaan : $x^2 - 8x + 15 = 0$.

```
[> solve(x^2 - 8*x + 15=0, x);
```

5,3.

Titik potong kurva tersebut dengan sumbu x adalah (5,0) dan (3,0). Kita dapat mengeceknya dengan grafik seperti ini:



```
[> plot(x^2-8*x+15, x = -1..6);
```

5. Gambarlah grafik $y = \sin(x)$ dan $y = 2 \cos(2x)$ dalam satu gambar, dengan batas $(0, 3\pi)$.

```
[> plot([sin(x), 2*cos(2*x)], x = 0..3*Pi);
```

6. Gambarlah grafik dalam tiga dimensi dari fungsi $z = \sin(x + \cos(y))$, untuk batas $0 \leq x \leq 4\pi$, dan $0 \leq y \leq 4\pi$.

```
[> plot3d(sin(x+cos(y)), x=0..4*Pi, y =0..4*Pi);
```



7. Gambarlah dalam koordinat polar $r = \frac{4\cos(3\theta)+\cos(13\theta)}{\cos(\theta)}$ dengan batasan $0 \leq \theta \leq 2\pi$. Gunakan perintah `polarplot` yang berada dalam paket **plots**.

```
[> with(plots):
[> polarplot( (4*cos(3*theta)+cos(13*theta))/cos(theta),
theta=0..2*Pi, colour=black, thickness=3 );
```



8. Animasikan fungsi parametrik dibawah ini dalam koordinat (x, y) .

$$x = r \cos(\theta)$$

$$y = r \sin(\theta)$$

Dengan batas $1 \leq r \leq 4$ dan $0 \leq \theta \leq 2\pi$


```
[> with( plots ):
[> animate( [r*cos(theta), r*sin(theta), theta=0..2*Pi],
r=1..4);
```




5.5 Soal - soal latihan

1. Tentukan titik potong garis $y = x + 1$ dengan parabola $y = x^2 - x - 2$. Gambarlah kedua fungsi tersebut.
2. Gambarlah permukaan $f(x, y) = x^4 - 2x^2 + y^2 - 1$ dalam bentuk tiga dimensi.
3. Gambarlah fungsi dari : $y = x \sin(3x)$ dengan batas : $0 \leq x \leq 2\pi$.
4. Gambarlah 2 fungsi ini yaitu $y = \sin(\theta)^3$ dan $y = \cos(\theta)^3$ dalam satu grafik. Dengan batasan $-\pi \leq x \leq \pi$.
5. Gambarlah grafi fungsi dari $y = \tan(x)$ dengan batasan $-2\pi \leq x \leq 2\pi$.
6. Buatlah grafik tiga dimensi dari $z = \frac{xy^2}{x^2+y^4}$ dengan batasan $-5 \leq x \leq 5$ dan $-5 \leq y \leq 5$.
7. Gambarlah dengan koordinat polar $r = \frac{4\cos(\theta)+\cos(9\theta)}{\cos(\theta)}$. Dengan batas $0 \leq \theta \leq 2\pi$.
8. Gambarlah fungsi parametrik dibawah ini pada koordinat cartesian (x, y) :


$$x = \frac{3t}{1+t^3}$$



(a) $r = 1$



(b) $r = 2$



(c) $r = 4$

$$y = \frac{3t^2}{1+t^3}$$

9. Buatlah grafik dibawah ini dalam tiga dimensi :

$$z = \frac{y^2 - x^2}{y^2 + x^2}$$

10. Animasikan dari fungsi berikut ini :

$$y(x, t) = A \sin(kx) \sin(\omega t).$$

Ambilah A , k , dan ω sembarang konstanta.

Bab 6

Kalkulus

6.1 Limit Fungsi

Topik *limit* biasanya dipakai sebagai bahan pengantar Kalkulus (hitung diferensial dan integral). Limit sendiri bukanlah Kalkulus, melainkan hanya pelengkap dasar - dasar Kalkulus. Pernyataan limit dapat ditulis :

$$\lim_{x \rightarrow a} f(x) = L,$$

yang bearti bahwa jika x mendekati a (tetapi $x \neq a$), maka nilai $f(x)$ akan mendekati L . Program Maple memberikan kemudahan untuk menghitung limit. Persoalan limit pada Maple dapat dipecahkan dengan perintah :

$$\text{limit}(f(x), x=a);$$

Perhatikan pernyataan x mendekati a ($x \rightarrow a$) ditulis pada Maple $x=a$.

Contoh 6.1.1. Hitunglah soal limit dibawah ini dengan Maple,

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x}.$$

Penyelesaiannya adalah; masukkan pada lembar kerja Maple seperti :

```
[>limit(sin(x)/x, x=0);
```

pernyataan diatas akan memberikan hasil :

1.

Dari sini bisa dilihat bahwa

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x} = 1.$$

Maple mengenal dua perintah untuk menghitung limit, yakni `limit` dan `Limit`(dimulai dengan huruf kapital). Perbedaan kedua perintah ini pada hasil akhirnya. Perintah `Limit` hanya akan mengasilkan bentuk pernyataan limit, dan tidak menghitung hasil limit tersebut. Sedangkan dengan perintah `limit`, Maple akan menghitung hasil limit tersebut. Contoh 6.1.1 akan digunakan lagi, kali ini dengan menggunakan perintah `Limit`,

```
[>Limit(sin(x)/x, x=0);
```

maka hasilnya akan seperti :

$$\lim_{x \rightarrow 0} \frac{\sin(x)}{x}.$$

Maple tidak menghitung limit diatas, melainkan hanya menuliskan bentuk limitnya saja, bandingkan dengan penyelesaiannya pada contoh 6.1.1.

Contoh 6.1.2. Hitunglah limit dibawah ini :

$$\lim_{x \rightarrow -\frac{9}{2}} \frac{2x^2 + 25x + 72}{72 - 47x - 14x^2}.$$

Penyelesaiannya adalah :

```
[> limit((2*x^2+25*x+72)/(72-47*x-14*x^2), x=-9/2);
```

$$\frac{7}{79}.$$

Contoh 6.1.3. Hitung nilai limit dari :

$$\lim_{x \rightarrow \infty} \left(1 + \frac{1}{x}\right)^x$$

Untuk membuat simbol tak terhingga (∞), gunakan perintah `infinity`. Jadi pada lembar kerja Maple dimasukkan :

```
[>limit((1+1/x)^x, x=infinity);
```

dan hasilnya :

$$e.$$

Selain pemecahan secara numerik, Maple dapat memecahkan limit secara grafis. Hal ini dapat dilihat pada contoh soal dibawah ini.


Contoh 6.1.4. Gunakan grafik untuk memecahkan limit dari

$$\lim_{x \rightarrow 1} \frac{x^7 - 2x^5 + 1}{x^3 - 3x^2 + 2}$$

Untuk menginvestigasi limit dari contoh 6.1.4, plot $\frac{x^7 - 2x^5 + 1}{x^3 - 3x^2 + 2}$ dengan batas antara 0 sampai 1.5. Agar tidak mengetik ulang, kita definisikan dahulu fungsi diatas dengan `f`. Seperti yang terlihat dibawah ini,

```
[> f:=(x^7 - 2*x^5 + 1)/(x^3-3*x^2+ 2);
```

Kemudian plot fungsi `f` terhadap sumbu `x`,



Gambar 6.1: Grafik dari $\frac{x^7-2x^5+1}{x^3-3x^2+2}$.

```
[>plot(f,x=0..1.5);
```

Dari gambar 6.1 tersebut terlihat harga f akan menjadi 1, jika harga x mendekati 1. Jika harga $x = 1$ disubstitusikan kedalam fungsi diatas, akan didapat hasil *error*. Hal ini terjadi karena dengan mensubstitusikan $x = 1$, harga penyebut dari fungsi diatas menjadi 0. Ini akan membuat fungsi diatas menjadi tidak terdefinisi.

Cara lain untuk memecahkan limit pada contoh diatas adalah dengan menyederhanakan fungsi yang berbentuk pecahan. Untuk keperluan ini dipakai perintah `simplify`. Perintah `simplify` ini telah dijelaskan pada subbab 3.6. Setelah pecahan tersebut disederhanakan, langkah selanjutnya adalah mensubstitusikan $x = 1$ kedalam hasil penyederhanaan tersebut. Dalam lembar kerja Maple dituliskan perintah seperti ini:

```
[>g:= simplify((x^7-2*x^5+1)/(x^3-3*x^2+2));
```

Hasil penyederhanaan tersebut kita beri nama g , yang hasilnya sebagai berikut :

$$g := \frac{x^6 + x^5 - x^4 - x^3 - x^2 - x - 1}{x^2 - 2x - 2},$$

kemudian $x = 1$ disubstitusikan kedalam fungsi g , dengan perintah `subs`,

```
[>subs(x=1,g);
```

dan diperoleh hasil :

1.

Cara langsung untuk mencari limit dari fungsi diatas dengan Maple,

```
[>limit((x^7 - 2*x^5 + 1)/(x^3-3*x^2+ 2),x=1);
```

1.

Contoh 6.1.5. Hitung soal dibawah ini :

$$\lim_{x \rightarrow \infty} e^{-2x} \sqrt{x}$$

Pernyataan e^{-2x} pada Maple ditulis `exp(-2*x)`, jadi ekspresi diatas dapat ditulis pada lembar kerja Maple dengan cara :

```
[> limit(exp(-2*x)*sqrt(x), x=infinity);
```

dan menghasilkan :

0.

6.2 Hitung Diferensial

Untuk menghitung nilai turunan (*derivative*) pada Maple, kita memakai perintah `diff`. Bentuk dari perintah `diff` ini adalah :

$$\text{diff}(f, x_1, x_2, \dots),$$

dimana f adalah sebuah fungsi, dan x_1, x_2, \dots adalah variabel bebas.

Contoh 6.2.1. Carilah turunan pertama dari fungsi berikut : $\sqrt[5]{x} - \sqrt{x}$.

Pada lembar kerja Maple ditulis,

```
[> diff(x^(1/5) - sqrt(x), x);
```

dan hasilnya adalah :

$$\frac{1}{5x^{4/5}} - \frac{1}{2\sqrt{x}}$$

Contoh 6.2.2. Carilah turunan $\sin(x)$ terhadap x .

```
[> diff(sin(x), x);
```

$$\cos(x).$$

Sama halnya dengan perintah `limit`, perintah `diff` juga mempunyai dua bentuk yaitu `Diff` (diawali dengan huruf kapital) dan `diff` (diawali dengan huruf kecil). Untuk membedakan kedua perintah tersebut, perhatikan contoh dibawah ini.

Contoh 6.2.3. Turunkan $\tan(x)$ terhadap x .

Langkah pertama, turunkan harga $\tan(x)$ dengan perintah `Diff`,

```
[>Diff(tan(x), x);
```

hasilnya akan terlihat seperti ini :

$$\frac{d}{dx} \tan(x);.$$

Sekarang bandingkan dengan perintah `diff`

```
[>diff(tan(x), x);
```

hasilnya adalah :

$$1 + \tan(x)^2.$$

Dari kedua contoh diatas terlihat bahwa penggunaan perintah `Diff` hanya memberikan bentuk dari diferensial dalam bentuk teks. Maple tidak menghitung hasil dari turunan tersebut. Maple akan menghitung turunan suatu fungsi apabila digunakan perintah `diff`.

6.2.1 Gradien garis

Aplikasi dari diferensial adalah untuk mencari gradien dari sebuah fungsi. Sebuah fungsi $f(x)$, akan mempunyai gradien $f'(a)$ di titik $(a, f(a))$. Persamaan garis yang mempunyai gradien $f'(a)$ dapat dicari dengan menggunakan rumus :

$$y - f(a) = f'(a)(x - a). \quad (6.1)$$

Contoh 6.2.4. Tentukan persamaan garis lurus yang menyinggung sebuah fungsi parabola $y = -1/3x^2 + x + 2$ di titik $(3, 2)$.

Dari soal diatas diketahui $f(x)$ adalah $-1/3x^2 + x + 2$, $a = 3$ dan $f(a) = 2$. Untuk menentukan persamaan gradien garisnya, langkah pertama yang dilakukan adalah mencari besarnya $f'(x)$. Agar mudahnya kita beri nama variabel *gradien* untuk turunan pertama dari $-1/3x^2 + x + 2$,

$$\text{gradien} = \frac{d(-1/3x^2 + x + 2)}{dx}.$$

Pada lembar kerja Maple, dimasukkan

```
[>gradien := diff(-1/3*x^2 + x + 2, x);
```

$$\text{gradien} := \frac{2x}{3} + 1.$$

Substitusikan nilai a kedalam *gradien*, sehingga $f'(a)$ diperoleh. Kemudian pada lembar kerja Maple $f'(a)$ diberi nama sebagai variabel m ,

```
[>m := subs(x=3, gradien);
```

$$m = -1.$$

Untuk mencari persamaan garis singgung yang mempunyai gradien -1 dan melalui titik $(3, 2)$, aplikasikan rumus 6.1. Nilai $f'(a) = m = -1$, $(a, f(a))$ adalah $(3, 2)$ telah diketahui. Selanjutnya nilai - nilai ini disubstitusikan ke persamaan 6.1, sehingga :

```
[> y :=m*(x- a)+f(a):
[> subs(m=-1,a=3, f(a)=2,y);
```

$$-x + 5.$$


Jadi persamaan garis singgung yang melalui parabola $-1/3x^2 + x + 2$ pada titik $(3, 2)$ adalah

$$y = -x + 5.$$

Secara grafis, Maple dapat menunjukkan bahwa persamaan garis $y = -x + 5$ dan parabola $-1/3x^2 + x + 2$ menyinggung di titik $(3, 2)$, yaitu dengan cara memplot kedua fungsi tersebut dalam satu gambar.

```
plot([-x+5, -1/3*x^2+x+2], x = -10..10);
```

Hasilnya dapat dilihat pada gambar 6.2.



Gambar 6.2: Garis $y = -x + 5$ menyinggung $y = -1/3x^2 + x + 2$ pada titik $(3, 2)$.

6.2.2 Menentukan harga maksimum dan minimum

Aplikasi lain dari penggunaan hitung difensial adalah menentukan harga maksimum dan minimum. Seperti diketahui bahwa harga ekstrim sebuah fungsi terjadi bilamana :

$$f'(x) = 0. \tag{6.2}$$

Contoh 6.2.5. Apakah $f(x) = x^3 - 12x + 5$ dengan interval $[-1, 4]$ mempunyai harga maksimum atau minimum, tentukan harga tersebut.

Pertama cari turunan pertama dari fungsi tersebut, hasil dari turunan tersebut, kita beri nama *solusi*. Pada lembar kerja Maple dimasukkan :

```
[> y:=x^3-12*x + 5;  
[> solusi:=diff(y,x);
```

$$\text{solusi} := 3x^2 - 12.$$

Tentukan harga x supaya $\text{solusi} = 0$. Hal ini dapat dicari dengan memakai perintah *solve*, seperti yang diperlihatkan dibawah ini:

```
[>solve(solusi=0,x);
```

$$2, -2.$$

Harga pembuat $\text{solusi} = 0$ terjadi pada saat $x = 2$ dan $x = -2$. Nilai $x = -2$ diabaikan karena nilai -2 berada diluar interval, yakni $[-1,4]$. Untuk mengetahui harga maksimum fungsi tersebut, maka $x = 2$ disubstitusikan ke persamaan y . Hal ini dikerjakan oleh Maple dengan memakai perintah *subs*, seperti,


```
[>subs(x = 2, y);
```

$$-11$$

Untuk memverifikasi jawaban diatas, fungsi diatas dapat di-plot sehingga dapat diketahui apakah nilai -11 tersebut adalah maksimum ataukah minimum. Untuk memplotnya, pada lembar kerja Maple dimasukkan perintah :


```
[>plot(y, x = -1..4)
```

Hasil dari plot tersebut dapat dilihat pada gambar 6.3. Dari gambar tersebut terlihat bahwa fungsi tersebut pada interval $[-1,4]$ mempunyai harga minimum yang bernilai -11.



Gambar 6.3: Fungsi dari $f(x) = x^3 - 12x + 5$ dengan interval $[-1,4]$, mempunyai harga minimum.

Contoh 6.2.6. Sebuah kotak yang terbuat dari karton, dibuka sisi-sisinya, seperti yang terlihat pada gambar 6.4. Panjang dan lebar kotak ini berturut turut adalah $150 \text{ [mm]} \times 240 \text{ [mm]}$. Jika sekarang karton tersebut pada bagian tertentu hendak ditekuk sepanjang x lihat gambar 6.4. Berapa harga x agar kotak tersebut mempunyai volume yang maksimum.



Gambar 6.4: Gambar soal 6.2.6.

Jika pada bagian sisi - sisi dari karton tersebut dipotong untuk dilipat sebesar x , maka panjang (p), lebar(l) dan tinggi(t) kotak tersebut adalah :

$$\begin{aligned} p &= 240 - 2x \\ l &= 150 - 2x \\ t &= x \end{aligned}$$

Untuk mencari volume (V) kotak tersebut adalah :

$$V = p.l.t$$

$$V = (240 - 2x)(150 - 2x)x$$

Seterusnya persoalan ini dapat dikerjakan oleh program Maple. Untuk mencari Volume dari kotak tersebut adalah :

```
[>V := (240 - 2*x)*(150-2*x)*x;
```

$$V := (240 - 2x)(150 - 2x)x.$$

Kemudian untuk mencari harga maksimumnya persamaan V diturunkan terhadap x , dan selanjutnya diberi nama `solusi`.

```
[> solusi:= diff(V,x);
```

$$solusi := -2(150 - 2x)x - 2(240 - 2x)x + (240 - 2x)(150 - 2x).$$

Untuk mendapatkan harga maksimum, maka nilai x harus ditentukan sehingga $solusi = 0$.

```
[>solve(solusi=0,x);
```

100, 30.

Terdapat dua harga x yang dapat menjadikan harga V menjadi maksimum. Harga $x = 100$, diabaikan, karena akan menjadikan lebar karton tersebut negatif ($150 - 2x$), dan ini tidak mungkin. Jadi harga $x = 30$ yang digunakan untuk mencari harga V maksimum. Hal ini dengan Maple dapat dicari dengan mensubstitusikan $x = 30$ ke persamaan V .

```
[> subs(x = 30, V);
```

hasilnya adalah :

486000.

Jadi volume maksimum yang terjadi adalah $486000 \text{ [mm}^3\text{]}$, untuk $x = 30 \text{ [mm]}$.

6.2.3 Turunan kedua

Program Maple dapat memecahkan persoalan differensial untuk turunan ke 2, atau lebih. Bentuk perintah yang digunakan adalah :

$$\text{diff}(f, x\$n),$$

dimana n merepresentasikan order turunan.

Contoh 6.2.7. Carilah turunan kedua dari $\sin(x)$,

$$y = \frac{d^2}{dx^2} \sin(x).$$

```
[> diff(sin(x), x$2);
```

dan hasilnya adalah :

$$-\sin(x).$$

Contoh 6.2.8. Carilah turunan pertama dan kedua dari : (a). $\tan(x)$ dan (b) $\sin^{-1}(x)$.

Untuk mencari turunan pertama :

```
[>diff(tan(x), x);
```

$$1 + \tan(x)^2.$$

Turunan kedua didapat dengan cara :

```
[>diff(tan(x), x$2);
```

$$2 \tan(x)(1 + \tan(x)^2).$$

Dengan cara yang sama kita kerjakan soal no (b).

```
[> diff(arcsin(x), x);
```

$$\frac{1}{\sqrt{1-x^2}}.$$

Untuk mencari turunan ke dua,

```
[> diff(arcsin(x), x$2);
```

$$\frac{x}{(1-x^2)^{3/2}}.$$

6.3 Hitung Integral

Jika turunan dari $F(x)$ adalah $f(x)$, ditulis :

$$F'(x) = f(x).$$

Fungsi $F(x)$ dapat dicari dengan mengintegalkan $f(x)$.

$$\int f(x) = F(x) + C,$$

dimana C adalah konstanta. Dalam Program Maple perintah integral dikenal dengan `int` atau `integrate`. Bentuk perintah tersebut adalah :

$$\text{int}(f, x).$$

atau

$$\text{integrate}(f(x), x).$$

6.3.1 Integral tak tentu

Jika dalam persoalan hitung integral, nilai konstanta C tidak diperlukan, maka persoalan integral dinamakan *integral indefinite* atau integral tak tentu.

Contoh 6.3.1. Hitung integral dari :

$$\int x^3 dx.$$

Pada lembar kerja Maple ditulis :

```
[> int(x^3,x);
```

$$\frac{x^4}{4}.$$

Jadi hasil integral $\int x^3 dx$ adalah $\frac{x^4}{4}$.

Contoh 6.3.2. Hitung integral dari :

$$\int x^n dx$$

```
[> int(x^n,x);
```

Hasilnya akan kita peroleh seperti :

$$\frac{x^{n+1}}{n+1}$$

Contoh contoh lain integral yang diselesaikan dengan memakai Program Maple :

Contoh 6.3.3. Hitung $\int \cos(x) dx$

```
[> int(cos(x),x);
```

$$\sin(x).$$

Contoh 6.3.4. Hitung integral dari $\frac{1}{x}$ terhadap x .

```
[> int(1/x,x);
```

$$\ln(x).$$

Contoh 6.3.5. Integralkan $\frac{1}{\sqrt{1-x^2}}$ terhadap x .

```
[> int(1/(1-x^2),x);
```

$$\arcsin(x).$$

Kadangkala dalam persoalan integral, suatu pernyataan ketika diintegrasikan tidak memberikan hasil jawaban. Jika hal ini terjadi, maka dengan program Maple juga tidak memberikan jawaban.

Contoh 6.3.6. Integralkan pernyataan dibawah ini:

$$\int (\exp(-x^2) \ln(x) dx.$$

```
[>int(exp(-x^2)*ln(x), x );
```

Hasil yang diberikan oleh Maple untuk hasil pengintegralan diatas adalah :

$$\int \exp(-x^2) \ln(x) dx.$$

6.3.2 Integral tertentu

Bilamana dalam persoalan hitung integral harga konstanta C diperlukan, maka hitung integral tersebut kita namakan *integral definite*, atau integral tertentu. Bentuk perintah integral tertentu pada lembar kerja Maple adalah

$$\text{int}(F, x=a..b).$$

Dimana $F(x)$ adalah sebuah fungsi x . Pernyataan $x = a$, dan $x = b$, adalah batas bawah dan atas fungsi yang akan diintegrasikan.

Contoh 6.3.7. Hitung integral dari x^3 terhadap x . Dengan batas $x = a$ dan $x = b$.

Pada lembar kerja Maple dimasukkan :

```
[>int(x^3, x = a..b);
```

$$\frac{b^4}{4} - \frac{a^4}{4}$$

Contoh 6.3.8. Hitunglah nilai integral :

$$\int_1^4 \frac{(x^2 + 1)}{\sqrt{x}} dx.$$

```
[>int(((x^2+1)/sqrt(x)), x=1..4);
```

$$\frac{72}{5}.$$

Contoh 6.3.9. Hitunglah nilai integral dari :

$$\int_0^{\sqrt{\pi/2}} x \cos x^2 dx.$$

```
[>int((x*cos(x^2)), x=0..sqrt(pi/2));
```

$$\frac{1}{2}.$$

Contoh 6.3.10. Hitunglah persoalan integral ini: $\int_0^{\infty} e^{-x^2} \ln(x) dx$.

```
[>int(exp(-x^2)*ln(x), x=0..infinity );
```

Hasil dari integral persamaan diatas adalah :

$$-\frac{1}{4}\sqrt{\pi}\gamma - \frac{1}{2}\sqrt{\pi}\ln(2).$$

Hasilnya agak sulit dipahami. Untuk mengetahui hasilnya secara numerik, maka Maple dapat diberi perintah untuk menghitung persoalan integral diatas dengan pendekatan *numerical integration*. Perintah tersebut adalah `evalf`. Untuk mencari harga numeriknya pada soal diatas, masukkan pada lembar kerja Maple :


```
[> evalf(int(exp(-x^2)*ln(x), x=0..infinity ));
```

Hasilnya akan :

$$-.8700577270$$

6.3.3 Luas Area

Salah satu aplikasi penggunaan hitung integral adalah mencari luas daerah dibawah kurva dari sebuah fungsi (lihat gambar 6.5) dengan interval $[a, b]$. Dalam gambar 6.5 $a = 2$ dan $b = 4$.



Gambar 6.5: Luas area dibawah grafik $f(x)$ dengan batas $a = 2$ dan $b = 4$.

Misalkan ada sebuah fungsi yang kontinu, $y = f(x)$ dengan batas $[a, b]$. Maka luas daerah A dibawah kurva fungsi tersebut dapat dicari dengan persamaan :

$$A = \int_a^b f(x). \quad (6.3)$$

Contoh 6.3.11. Carilah luas daerah dibawah kurva $y = 9 - 4x^2$ dengan batas $x = 0$ dan $x = 3/2$.

Persoalan diatas dapat dengan mudah dipecahkan oleh Maple, dengan memasukkan perintah `int` kedalam lembar kerja Maple,

```
[>int(9-4*x^2, x=0..3/2);
```

9.


Jadi luas daerah yang dibatasi oleh $y = 9 - 4x^2$, dan $x = 0$, $x = 3/2$ adalah 9. Agar lebih jelas, bagaimana bentuk geometri dari luas daerah yang dimaksud, kurva tersebut dapat diplot. Maple dapat memplot daerah yang dibatasi oleh dua buah kurva atau lebih. Dalam bab ini pembahasan hanya dibatasi pada daerah yang dibatasi oleh sebuah kurva dan sumbu x saja. Pada prinsipnya memplot sebuah daerah sama dengan memplot sebuah kurva, hanya pada bagian option diberi perintah tambahan yaitu `filled`. Untuk memplot daerah $f(x) = 9 - 4x^2$ dengan interval antara 0 dan $3/2$, dan dibatasi oleh sumbu x pada lembar kerja Maple ditulis :

```
[> f:= 9 - 4*x^2;  
[> plot(f,x=0..3/2, filled = true, color=green);
```

hasilnya dapat dilihat pada gambar 6.6(a). Gambar ini mungkin belum terlalu jelas, karena gambar ini hanya menggambar sebuah daerah yang dibatasi oleh sebuah fungsi dan sumbu x . Untuk lebih menjelaskan bahwa daerah tersebut berada dibawah sebuah kurva, maka kurva tersebut harus digambar secara bersamaan. Perintah `display` (lihat subbab 5.1.4) dapat digunakan untuk keperluan ini. Pada lembar kerja Maple ditulis :

```
[> with(plots):
[> f:= 9 - 4*x^2:
[> a:=plot(f,x=0..3/2, filled = true, color=grey):
[> b:=plot(f,x=0..3/2, color=black, thickness=3):
[> display(a,b, TEXT([0.8,8], 'f(x)=9-4x^2', FONT(TIMES,BOLD,12))];
```

hasilnya dapat dilihat pada gambar 6.6(b).




Gambar 6.6: (a). Gambar dari daerah dari $f(x) = 9 - 4x^2$, (b). Daerah $f(x) = 9 - 4x^2$ diplot secara bersamaan dengan kurva $f(x) = 9 - 4x^2$.

Catatan 6.3.1. Penambahan perintah `TEXT([0.8,8])` akan memberikan efek tampilnya text " $f(x) = 9 - 4x^2$ " pada koordinat (0.8,8) pada grafik. Sedangkan perintah `FONT(TIMES,BOLD,12)` membuat Maple menulis sebuah text dengan huruf jenis Times dan style tebal (bold) dengan ukuran 12.

Contoh 6.3.12. Carilah luas daerah yang dibatasi oleh $y = \sin(x)$ dan $y = \cos(x)$ dengan interval $(0, 2\pi)$.

Untuk mudahnya, gambar kedua kurva ($y = \sin(x)$ dan $y = \cos(x)$), dalam satu gambar sehingga dapat terlihat daerah yang dimaksud.

```
[>plot([sin(x),cos(x)],x=0..2*Pi,color=[green,red], thickness=3);
```



Gambar 6.7: Grafik $y = \sin(x)$ dan $y = \cos(x)$ untuk interval $(0, 2\pi)$.

Catatan 6.3.2. Seperti yang telah disinggung pada catatan 6.3.1, untuk menggambarkan daerah yang dibatasi oleh dua kurva yaitu $y = \sin(x)$, $y = \cos(x)$, cukup rumit. Hal ini tidak dibahas dalam buku ini.

Gambar (6.7) memperlihatkan dua grafik yang dimaksud pada soal diatas. Untuk menentukan batasan luas areanya, perlu diketahui dahulu titik - titik potong kedua grafik itu, caranya adalah dengan mensubstitusi kedua kurva tersebut $\sin(x) = \cos(x)$. Namun sebelum itu, Maple harus di set dahulu, sehingga Maple dapat mencari semua titik potong kedua kurva tersebut. Cara mensetnya :

```
[>_EnvAllSolutions := true:
```

Perintah `_EnvAllSolutions := true:` diartikan oleh Maple untuk mencari semua himpunan jawaban dari $\sin(x) = \cos(x)$. Langkah selanjutnya adalah mensubstitusi $\sin(x) = \cos(x)$ dengan perintah `solve`,

```
[>solve(sin(x)=cos(x),x);
```

kemudian menghasilkan :

$$\frac{1}{4}\pi + \pi_Z I.$$

Ini artinya perpotongan kedua grafik tersebut terjadi pada titik titik $1/4\pi$, $5/4\pi$, $9/4\pi$, ... Namun karena interval kedua kurva ini adalah $[0, 2\pi]$, maka titik - titik poting kedua kurva tersebut adalah $1/4\pi$, dan $5/4\pi$. Dari gambar 6.7 terlihat ada 3 daerah yang memenuhi, yaitu daerah dari $x = 0$ sampai $x = 1/4\pi$ kemudian dari $x = 1/4\pi$ sampai $x = 5/4\pi$ dan yang terakhir

daerah dari $x = 5/4\pi$ sampai $x = 2\pi$. Luas daerah L antara kedua grafik itu dapat dicari dengan mengintegrasikan bagian demi bagian,

$$L = \int_0^{\pi/4} [\cos(x) - \sin(x)]dx + \int_{\pi/4}^{5\pi/4} [\sin(x) - \cos(x)]dx + \int_{5\pi/4}^{2\pi} [\cos(x) - \sin(x)]dx.$$

Pernyataan diatas dapat disederhanakan menjadi :

$$L = 2 \int_{\pi/4}^{5\pi/4} [\sin(x) - \cos(x)]dx.$$

Untuk selanjutnya dalam lembar kerja Maple ditulis :


```
[>L:= 2*int(sin(x)-cos(x), x=Pi/4..5*Pi/4);
```

dan hasilnya adalah :

$$4\sqrt{2}.$$

Contoh 6.3.13. Tentukan luas daerah yang dibatasi oleh kurva $y = 4 - x^2$ dan sumbu x .

```
[>with(plots):
[> a:=plot(4-x^2, x=-2..2, filled=true, color=grey, view=[-5..5, -5..5]):
[> b:=plot(4-x^2, x=-5..5, color=black, view=[-5..5, -5..5]):
[> display(a,b,view=[-5..5,-5..5]);
```



Gambar 6.8: Area yang dibatasi oleh grafik $y = 4 - x^2$ dan sumbu x .

Dari gambar 6.8 terlihat, grafik $y = 4 - x^2$, memotong sumbu x . Titik potong kurva tersebut dengan sumbu x dapat dicari dengan perintah `solve` atau `fsolve`. Syarat grafik tersebut memotong sumbu x adalah jika harga $y = 0$. Pada lembar kerja Maple dituliskan :

```
[> solve(4 - x^2=0, x);
```

hasilnya adalah :

$$-2, 2.$$

Kurva $y = 4 - x^2$ memotong sumbu x pada titik $x = 2$ dan $x = -2$. Titik titik potong ini kemudian digunakan sebagai batas untuk mencari luas area antara grafik $y = 4 - x^2$ dan sumbu x . Luas daerah yang dimaksud selanjutnya dapat dicari dengan mengintegrasikan $y = 4 - x^2$ dengan batas $x = -2$ dan $x = 2$.

```
[> A := int(4-x^2, x = -2..2);
```

$$A := \frac{32}{3}$$

Jadi luas daerah yang dimaksud adalah : $32/3$.

6.4 Paket Student

Dalam subbab ini, pembaca akan diperkenalkan kepada paket Maple yang disebut paket Student. Paket ini sangat berguna untuk memecahkan persoalan - persoalan matematika yang sering dijumpai oleh siswa SMU dan mahasiswa tingkat pertama jurusan eksakta. Subbab ini akan memfokuskan pembahasan paket student yang menyangkut masalah kalkulus, lebih spesifik lagi adalah kalkulus variabel tunggal (*single variable*). Didalam paket student ini bukan hanya fungsi fungsi untuk memecahkan masalah kalkulus yang akan dijumpai, namun juga tutorial dari fungsi - fungsi tersebut. Dengan tuttorial ini siswa ataupun mahasiswa dapat melakukan kegiatan interaktif dengan Maple dalam memecahkan persoalan kalkulus.

Untuk memanggil paket student dengan modul kalkulus variabel tunggal digunakan perintah :

```
with(Student[Calculus1]);
```


Pernyataan `Calculus1` mengartikan kalkulus dengan variabel tunggal. Setelah perintah tersebut dieksekusi, maka pada lembar kerja Maple akan tampil fungsi fungsi yang dapat digunakan untuk memecahkan persoalan kalkulus. Perintah - perintah tersebut adalah :

[AntiderivativePlot, AntiderivativeTutor, ApproximateInt, ApproximateIntTutor, ArcLength, ArcLengthTutor, Asymptotes, Clear, CriticalPoints, CurveAnalysisTutor, DerivativePlot, DerivativeTutor, DiffTutor, ExtremePoints, FunctionAverage, FunctionAverageTutor, FunctionChart, FunctionPlot, GetMessage, GetNumProblems, GetProblem, Hint, InflectionPoints, IntTutor, Integrand, InversePlot, InverseTutor, LimitTutor, MeanValueTheorem, MeanValueTheoremTutor, NewtonQuotient, NewtonsMethod, NewtonsMethodTutor, PointInterpolation, RiemannSum, RollesTheorem, Roots, Rule, Show, ShowIncomplete, ShowSteps, Summand, SurfaceOfRevolution, SurfaceOfRevolutionTutor, Tangent, TangentSecantTutor, TangentTutor, TaylorApproximation, TaylorApproximationTutor, Understand, Undo, VolumeOfRevolution, VolumeOfRevolutionTutor, WhatProblem]

Dari sejumlah perintah perintah diatas, yang paling sering dijumpai pada persoalan matematika tingkat SMU adalah : panjang kurva (*Arch Length*), dan volume benda putar (*Volume of Revolution*). Pembahasan pada subbab ini akan memfokuskan pada kedua perintah tersebut.

6.4.1 Volume benda putar

Bilamana terdapat daerah **R** yang dibatasi oleh kurva $y = f(x)$, sumbu x , dan garis $x = a$, $x = b$, lihat gambar 6.9. Jika daerah **R** tersebut diputar dengan sumbu x sebagai sumbu putarnya maka daerah **R** akan membentuk sebuah bentuk solid yang dapat dihitung volumenya.



Gambar 6.9:

Volume benda putar dari $f(x)$ yang mengelilingi sumbu x dapat dicari dengan rumus :

$$V = \pi \int_a^b [f(x)]^2 dx.$$

Pada paket student, persoalan mencari volume benda putar dapat dipecahkan dengan perintah `VolumeOfRevolution`. Bentuk dari perintah tersebut adalah :

```
VolumeOfRevolution(f(x), x = a..b, opsi).
```

Dimana $f(x)$ adalah fungsi yang membuat volume benda putar, a dan b adalah batasan kurva pada sumbu x . Sedangkan pada opsi, pengguna Maple dapat memerintahkan Maple agar mengeluarkan hasilnya dalam bentuk plot, dalam bentuk integral, dan dalam bentuk nilai. Jika pengguna Maple ingin mendapatkan hasil perhitungan volume benda putar ini dalam bentuk plot, maka pada bagian opsi ditulis `output=plot`, jika pengguna ingin mendapatkan dalam bentuk integral, maka pada bagian opsi ditulis `output=integral`, sedangkan jika pengguna Maple hanya ingin mengetahui nilai dari volume benda putar tersebut, maka pada bagian opsi ditulis `output = value`.


Contoh 6.4.1. Hitung isi benda putar yang terbentuk karena perputaran daerah terhadap sumbu x . Daerah tersebut dibatasi oleh $y = x^2$, sumbu x , $x = 0$, dan $x = 2$.

Dengan mengaplikasikan perintah `VolumeOfRevolution`, pada lembar kerja Maple dituliskan :

```
[>with(Student[Calculus1]):  
[>VolumeOfRevolution(x^2, x=0..2, output=plot);
```

Dengan memasukkan `output=plot`, maka perintah diatas akan menghasilkan sebuah gambar seperti yang diilustrasikan pada gambar 6.10.

The Volume of Revolution Around the Horizontal Axis of
 $f(x) = x^2$
on the Interval $[0, 2]$



Gambar 6.10: Volume benda putar dari fungsi $y = x^2$ dengan batas $[0, 2]$, yang diputar terhadap sumbu x .

Jika sekarang kita ingin mengetahui volume benda putar tersebut dalam bentuk persamaan integral, maka perintah `output=integral` dimasukkan pada perintah diatas, sehingga :

```
[>with(Student[Calculus1]):  
[>VolumeOfRevolution(x^2, x=0..2, output=integral);
```

dan hasilnya dapat dilihat pada pernyataan dibawah ini:

$$\pi \int_0^2 x^4 dx.$$

Bila perintah `output=value` diletakkan pada bagian opsi, maka Maple akan menghitung nilai volume benda putar tersebut.

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(x^2, x=0..2, output=value);
```


$$\frac{32\pi}{5}.$$

Contoh 6.4.2. Kita gunakan lagi contoh 6.4.1. Hitung isi benda putar yang terbentuk karena perputaran daerah terhadap sumbu y . Daerah tersebut dibatasi oleh $y = x^2$, sumbu x , $x = 0$, dan $x = 2$.

Untuk menginformasikan kepada Maple bahwa volume benda putar yang dicari kali ini, memakai sumbu vertikal sebagai sumbu putarnya, maka pada perintah `VolumeOfRevolution` ditambahkan perintah `axis = vertical`. Jika pada perintah `axis` tidak dituliskan, maka Maple akan menghitungnya benda tersebut diputar terhadap sumbu x .

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(x^2, x=0..2, axis = vertical, output=plot);
```

The Volume of Revolution Around the Vertical Axis of
 $f(x) = x^2$
 on the Interval $[0, 2]$



Gambar 6.11: Volume benda putar dari fungsi $y = x^2$ dengan batas $[0,2]$, yang diputar terhadap sumbu y .

Dengan mengganti `output = plot` menjadi `output = integral`, Maple akan menampilkan volume yang terjadi dalam bentuk persamaan integral.

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(x^2, x=0..2, axis=vertical, output=integral);
```

$$\int_0^2 2\pi x^3 dx.$$

Sedang untuk mengetahui besarnya volume benda putar yang terjadi, pernyataan `output = value` diaplikasikan kedalam perintah `VolumeOfRevolution`.

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(x^2, x=0..2, axis=vertical, output=value);
```

hasilnya adalah :

$$8\pi.$$


6.4.2 Volume benda putar yang dibatasi dua fungsi

Jika sebuah daerah yang dibatasi oleh dua buah fungsi $f(x)$ dan $g(x)$ dengan interval $[a, b]$, diputar terhadap sumbu x , maka volume benda yang terjadi dapat dihitung dengan menggunakan rumus :

$$V = \pi \int_a^b \left[\{f(x)\}^2 - \{g(x)\}^2 \right] dx \quad (6.4)$$

Untuk menghitung persoalan ini, perintah `VolumeOfRevolution` dapat tetap diaplikasikan.

Contoh 6.4.3. Hitung volume benda putar yang terbentuk karena perputaran daerah yang dibatasi oleh $y = x^3$ dan $y = x$. Daerah tersebut terletak diatas sumbu x , dan diputar terhadap sumbu x .



Gambar 6.12: Daerah yang dibatasi oleh $y = x^3$ dan $y = x$, daerah ini kemudian diputar terhadap sumbu horisontal (x).

Langkah pertama adalah memvisualisasikan persoalan diatas, sehingga kita dapat dengan jelas melihat batasan - batasan dari benda tersebut. Agar melakukan penulisan ulang, kita definisikan $f(x) = x^3$ dan $g(x) = x$. Untuk mengilustrasikan daerah perpotongan antara $f(x)$ dan $g(x)$, pada lembar kerja Maple dituliskan :

```
[>with(plots):
[> fx=x^3:
[> gx = x:
[> a:=plot([fx,gx],x= 0..1.5, filled= true, color=[white,grey]):
[> b:=plot([fx,gx],x=0..1.5, color=[black,black]):
[> display(a,b);
```

Hasilnya dapat dilihat pada gambar 6.12. Dari gambar tersebut terlihat bahwa batasan lain dari benda tersebut adalah titik potong antara $y = x^3$ dan $y = x$. Perintah `solve` digunakan untuk mencari titik potong kedua kuva tersebut,

```
[>solve(fx=gx,x);
```

dan memberikan hasil,

0, 1, -1.

Karena daerah yang diinginkan adalah diatas sumbu x , maka titik potong yang dipakai adalah $x = 0$ dan $x = 1$. Selanjutnya volume benda yang dimaksud dapat diketahui dengan mengaplikasikan perintah `VolumeOfRevolution`. Untuk mengetahui volume benda tersebut dalam bentuk persamaan integral, maka pada lembar kerja Maple dituliskan :

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(fx^2-gx^2, x = 0..1, output=integral);
```

dan hasilnya adalah :

$$\int_0^1 \pi x^4 (x^4 - 1)^2 dx.$$

Untuk mengetahui besarnya volume benda tersebut, pernyataan `output=integral` diganti dengan `output=value` seperti dibawah ini,

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(fx^2-gx^2, x = 0..1, output=value);
```

dan hasilnya adalah :

$$\frac{32\pi}{585}.$$

Untuk melihat bentuk volume benda tersebut, pernyataan `output=value` diganti dengan `output=plot` seperti dibawah ini,

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(fx^2-gx^2, x = 0..1, output=plot);
```

gambar 6.13 memperlihatkan bentuk benda putar tersebut.


6.4.3 Panjang Kurva

Jika fungsi $f(x)$ dan turunannya $f'(x)$ kontinu dalam interval $[a, b]$, maka panjang busur s dari kurva $y = f(x)$ mulai dari titik $(a, f(a))$ sampai $(b, f(b))$ dapat dicari dengan :

$$s = \int_a^b \sqrt{1 + \left(\frac{dy}{dx}\right)^2} dx, \tag{6.5}$$

atau

The Volume of Revolution Around the Horizontal Axis of
 $f(x) = x^6 - x^2$
 on the Interval $[0, 1]$



Gambar 6.13: Volume benda putar yang dibatasi oleh $y = x^3$ dan $y = x$

$$s = \int_a^b \sqrt{1 + \left(\frac{dx}{dy}\right)^2} dy. \quad (6.6)$$

Panjang busur dapat dicari oleh Maple dengan menggunakan perintah `ArcLength`. Bentuk penulisan dari perintah ini adalah :

`ArcLength(f(x), batasan, opsional).`

Pada bagian optional dapat diganti dengan *output* dengan pilihan antara lain : *value*, *integral*.

Contoh 6.4.4. Hitung panjang busur $y = x^{2/3}$ dari titik (1,1) sampai titik (8,4).

Untuk jelasnya, kurva yang dimaksud dapat digambar dengan memakai perintah `plot`,


```
[>plot(x^(2/3), x = 0..10);
```

hasil dari perintah diatas dapat dilihat pada gambar 6.14.

Untuk menghitung panjang kurva tersebut, kita aplikasikan perintah `ArcLength`. Bila yang diinginkan hasilnya adalah panjang kurva dalam bentuk persamaan integral, maka pada bagian optional dimasukkan perintah `output = integral`, seperti yang dituliskan dibawah ini.

```
[>with(Student[Calculus1]):  
[>ArcLength(x^(2/3), x = 1..8, output=integral);
```

Perintah diatas akan menghasilkan panjang kurva $y = x^{2/3}$ dalam bentuk persamaan integral, seperti yang terlihat dibawah ini.



Gambar 6.14: Kurva $y = x^{2/3}$.

$$\int_1^8 \frac{\sqrt{4+9x^{(2/3)}}}{3} dx.$$

Bila yang diinginkan agar Maple menghitung nilai dari panjang kurva tersebut, maka pada bagian opsional perintah `output = integral`, diganti menjadi `output = value`.

```
[>with(Student[Calculus1]):
[>ArcLength(x^(2/3), x = 1..8, output=value);
```

dan hasilnya adalah :

$$-\frac{13\sqrt{13}}{27} + \frac{80\sqrt{10}}{27}.$$

6.5 Soal - Soal dan Penyelesaiannya

Kerjakan soal - soal dibawah ini dengan menggunakan Program Maple. Bandingkan hasilnya bila dikerjakan secara manual.

1. Tentukan nilai limit dibawah ini:

(a) $\lim_{x \rightarrow 2} (x^2 - x + 4)$.

(b) $\lim_{x \rightarrow 1} \sqrt{10 - x}$.

(c) $\lim_{x \rightarrow 2} \frac{x^2 - 4}{x - 2}$.

(d) $\lim_{x \rightarrow 1} \frac{x - 1}{\sqrt{x} - 1}$.

Untuk mendapatkan penyelesaian dari soal - soal diatas, pada lembar kerja Maple kita tuliskan sebagai berikut :

(a) $\lim_{x \rightarrow 2} (x^2 - x + 4)$.

[> limit(x^2-x+4, x=2);

6.

(b) $\lim_{x \rightarrow 1} \sqrt{10 - x}$.

[> limit(sqrt(10-x), x=1);

3.

(c) $\lim_{x \rightarrow 2} \frac{x^2 - 4}{x - 2}$.

[> limit((x^2-4)/(x-2), x=2);

4.

(d) $\lim_{x \rightarrow 1} \frac{x-1}{\sqrt{x}-1}$.

[> limit((x-1)/(sqrt(x)-1), x=1);

2.

2. Integralkan pernyataan ini : $\int \sqrt[3]{x} dx$. Kita ubah dahulu pernyataan $\sqrt[3]{x}$ menjadi $x^{1/3}$.

[> int(x^(1/3), x);

$$\frac{3x^{4/3}}{4}.$$

Maple tidak menambahkan konstanta untuk perhitungan integral tak tentu, oleh sebab itu janganlah lupa kalau hasil integral diatas perlu ditambah konstanta C.

3. Integralkan $\int (1 - x)\sqrt{x} dx$. Kita bisa ubah pernyataan $(1 - x)\sqrt{x}$ menjadi $(1 - x)x^{1/2}$. Kemudian pernyataan itu kita integralkan dengan menggunakan Maple seperti :

[> int((1-x)*x^(1/2), x);

$$-\frac{2(-5 + 3x)x^{3/2}}{15}$$

4. Integralkan $\frac{8x^2}{3(x^3+2)^3} dx$. Supaya tidak menulis terlalu panjang, kita beri nama dahulu $\frac{8x^2}{3(x^3+2)^3}$

```
[>f:=(8*x^2)/(3*(x^3+2)^3);
```

$$f := \frac{8x^2}{3(x^3 + 2)^3}.$$

Lalu kita integralkan menjadi :

```
[>int(f,x);
```

$$-\frac{4}{9(x^3 + 2)^2}.$$

5. Hitunglah turunan pertama dari fungsi trigonometri dibawah ini:

(a) $y = \tan(x)$

(b) $y = \sec(x)$

(c) $y = \cos(x)$

Jawaban soal (a)

```
[>diff(tan(x),x);
```

$$1 + \tan(x)^2.$$

Jabawan soal (b)

```
[>diff(sec(x),x);
```

$$\sec(x) \tan(x).$$

Jawaban soal (c)

```
[> diff(cos(x),x);
```

$$-\sin(x).$$

6. Hitunglah nilai dari integral tertentu $\int_0^1 x(1 - \sqrt{x})^2 dx$. Kita bisa menuliskan langsung pada Maple seperti :

```
[> int(x*(1-sqrt(x))^2, x=0..1);
```

dan hasilnya adalah :

$$\frac{1}{30}$$

7. Berapakah nilai dari $\int_{-6}^{-10} \frac{dx}{x+2}$ Dengan Maple kita dengan mudah mencari nilai dari pernyataan diatas :

```
[>int(1/(x+2), x= -6..-10);
```

Hasilnya adalah :

$$\ln(2).$$

8. Hitung daerah yang dibatasi oleh kurva parabola $y = 9 - x^2$ dan garis $x + y = 7$. Untuk mencari luas area yang dibatasi oleh dua buah kurva, maka dipakai persamaan :

$$A = \int_a^b (f(x) - g(x))dx. \quad (6.7)$$

Kurva parabola $y = 9 - x^2$ mewakilkan $f(x)$ dan persamaan garis $x + y = 7$ mewakilkan $g(x)$. Sedangkan a dan b adalah titik potong antara $f(x)$ dan $g(x)$ Bentuk $g(x) = x + y = 7$ dapat diubah bentuknya menjadi :

$$\begin{aligned}x + y &= 7 \\ y &= 7 - x\end{aligned}$$

Untuk mencari titik - titik potong kedua kurva tersebut, kita gunakan perintah `solve`, seperti dibawah ini:

```
[> solve([y=7-x, y=9-x^2], {x,y});
```


$$\{y = 5, x = 2\}, \{y = 8, x = -1\}.$$

Titik potong kedua kurva tersebut adalah (2,5) dan (-1,8). Untuk memvisualisasikan persoalan diatas, kita gunakan perintah `plot` seperti,

```
[>plot([7-x, 9-x^2], x=-4..4);
```

Dari gambar 6.15 terlihat daerah yang dimaksud mempunyai batasan pada $x = -1$ dan $x = 2$. Untuk mencari luas daerah tersebut, rumus 6.7, sehingga :

```
[> fx := 9 - x^2;
[> gx := 7 - x;
[> A:= int(fx-gx,x=-1..2);
```



Gambar 6.15: Gambar soal no.8, mencari luas daerah yang dibatasi oleh kurva $y = 9 - x^2$ dan $x + y = 7$.

Dari hasil perhitungan integral diatas, harga A diketahui

$$A := \frac{9}{2}.$$

9. Hitung volume benda putar yang terbuat dari kurva $f(x) = y = x - x^2$ dan dibatasi oleh sumbu x . Kurva tersebut diputar terhadap sumbu x .

Untuk memecahkan persoalan ini kita gunakan paket *student*. Namun sebelum itu untuk mudahnya, kita visualisasikan dahulu kurva tersebut dalam koordinat sumbu x dan y . Hal ini penting untuk mengetahui batasan - batasan dari kurva tersebut.

```
[>plot(x-x^2, x=0..1);
```


Dari gambar 6.16 terlihat bahwa kurva tersebut memotong sumbu x pada dua titik. Untuk mencari titik potong tersebut dapat digunakan perintah `solve`, seperti:

```
[>solve(x-x^2=0, x);
```

Hasilnya adalah :

0,1.

Artinya kurva tersebut memotong sumbu x pada titik potong 0 dan 1. Angka angka ini kemudian dijadikan batasan integral untuk menghitung volume benda putar. Selanjutnya



Gambar 6.16: Kurva $y = x - x^2$ yang akan diputar terhadap sumbu x .

kita gunakan paket student untuk menghitung dan memvisualisasikan volume benda putar tersebut. Untuk mencari besarnya volume benda putar yang terjadi, kita masukkan perintah dibawah ini pada lembar kerja Maple.

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(x-x^2, x=0..1, output=value);
```

dan hasilnya adalah :

$$\frac{\pi}{30}.$$

Untuk melihat bagaimana bentuk volume benda putar yang terjadi, kita gunakan perintah yang sama, hanya pada bagian `output=value`, diganti dengan `output=plot`. Lihat langkah - langkah dibawah ini.

```
[>with(Student[Calculus1]):
[>VolumeOfRevolution(x-x^2, x=0..1, output=plot);
```

Hasilnya dapat dilihat pada gambar [6.17](#).


10. Sebuah kurva,

$$y = \frac{x^4}{8} + \frac{1}{4x^2}.$$

Hitung panjang kurva tersebut dari $x = 1$ sampai $x = 2$.

Untuk mengerjakan soal diatas, kita gunakan perintah `ArcLength` yang berada pada paket Student. Pada lembar kerja Maple ditulis :

The Volume of Revolution Around the Horizontal Axis of
 $f(x) = x - x^2$
 on the Interval $[0, 1]$



Gambar 6.17: Hasil perputaran kurva $y = x - x^2$ terhadap sumbu x .

```
[>with(Student[Calculus1]):  
[>ArcLength(x^4/8 + 1/(4*x^2), x=1..2, output=value);
```

Perhitungan diatas menghasilkan :

$$\frac{33}{16}$$

Untuk melihat bentuk kurva tersebut, fungsi diatas dapat di-plot,

```
[>plot(x^4/8 + 1/(4*x^2), x=1..2);
```

6.6 Soal - Soal Latihan


1. Gunakan metode grafik dan aljabar untuk menghitung persoalan limit dibawah ini:

(a) $\lim_{x \rightarrow 5} \frac{x^3 - 125}{x^5 - 5}$

(b) $\lim_{x \rightarrow 0} \frac{\sqrt{1+x+x^2}-1}{x}$

(c) $\lim_{x \rightarrow 0} \frac{\sqrt{1+x}-\sqrt{1+x^2}}{\sqrt{1-x^2}-\sqrt{1-x}}$

2. Hitung turunan pertama dari :



Gambar 6.18: Panjang busur dari kurva $y = x^4/8 + 1/4x^2$, dengan batas $x = 1$ sampai $x = 2$.

(a) $y = 4x^5 - 5/2x^4 - 10x^3$.

(b) $y = \sqrt{e^{2x} + e^{-2x}}$.

(c) $y = (1 + 1/x)^x$.

3. Hitung integral dari :

(a) $\int \frac{dx}{x\sqrt{x}}$.

(b) $\int (\sqrt{2x} - \frac{1}{\sqrt{2x}}) dx$.

(c) $\int_0^\pi e^{2x} \sin^2(2x) dx$.

(d) $\int_0^{\sqrt{\pi/2}} x \cos(x^2) dx$.

4. Tentukan luas area yang dibatasi oleh kurva $y = 12 - x - x^2$, sumbu x dan garis $x = -3$ dan $x = 2$.

5. Hitunglah luas area yang dibatasi oleh kurva $y = x^2 - 5x + 6$, $x = 0$ dan $x = 4$.

6. Sebuah massa digantungkan pada sebuah pegas, mempunyai persamaan perpindahan jarak x fungsi dari waktu t sebagai berikut :

$$x(t) = 40 + 5e^{-5t} \cos(\pi t - \frac{\pi}{2}).$$

(a) Buatlah grafik x fungsi dari t dengan interval $[0,2]$.

(b) Carilah persamaan kecepatan v dari persamaan diatas. Kita tahu bahwa kecepatan adalah :

$$v = \frac{d}{dt}x(t).$$

7. Jika sebuah benda mempunyai kecepatan gerak sebesar :

$$v(t) = 1 - e^{-t^2}.$$

Tentukan persamaan jaraknya x sebagai fungsi dari waktu t .

8. Tentukan luas sebuah daerah yang dibatasi oleh sebuah kurva $x = y^2$ dan sebuah garis $y = x - 2$
9. Hitunglah sebuah volume benda putar, yang dibatasi oleh kurva $y = x^2$, sumbu x , dan garis $x = 2$. Benda ini diputar dengan memakai sumbu x sebagai sumbu putarnya.
10. Carilah persamaan garis lurus yang menyinggung parabola ($y = x^2$) pada titik (2,4).
11. Sebuah pagar yang panjangnya 20 [m] akan dipakai untuk memagari sebuah tanah yang berbentuk segitiga siku - siku. Berapa ukuran segitiga tersebut agar luas tanah tersebut maksimum.

Bab 7

Matrik dan Vektor

Pada bab ini akan diperlihatkan bagaimana Maple dapat diaplikasikan untuk menyelesaikan persoalan Aljabar Linear. Topik yang dibahas pada bab ini adalah tentang Vektor dan Matrik. Perintah - perintah untuk mengoperasikan vektor dan matrik berada dalam paket *Linear Algebra*, yang disingkat `linalg`.

Sebelum memakai perintah - perintah yang dapat digunakan menyelesaikan persoalan Matrik dan Vektor, pastikan dahulu, bahwa paket `linalg` telah dipanggil. Untuk memanggil paket `linalg`, digunakan perintah `with(linalg)`. Setelah paket ini terpanggil, maka pada lembar kerja akan tampil, perintah - perintah apa saja yang terdapat dalam `linalg` yang bisa digunakan untuk memecahkan soal Aljabar Linear.

[*BlockDiagonal, GramSchmidt, JordanBlock, LUdecomp, QRdecomp, Wronskian, addcol, addrow, adj, adjoint, angle, augment, backsub, band, basis, bezout, blockmatrix, charmat, charpoly, cholesky, col, coldim, colspace, colspan, companion, concat, cond, copyinto, crossprod, curl, definite, delcols, delrows, det, diag, diverge, dotprod, eigenvals, eigenvalues, eigenvectors, eigenvects, entermatrix, equal, exponential, extend, ffgausselim, fibonacci, forwardsub, frobenius, gausselim, gaussjord, geneqns, genmatrix, grad, hadamard, hermite, hessian, hilbert, htranspose, ihermite, indexfunc, innerprod, intbasis, inverse, ismith, issimilar, iszero, jacobian, jordan, kernel, laplacian, leastsqrs, linsolve, matadd, matrix, minor, minpoly, mulcol, mulrow, multiply, norm, normalize, nullspace, orthog, permanent, pivot, potential, randmatrix, randvector, rank, ratform, row, rowdim, row space, rowspan, rref, scalarmul, singularvals, smith, stackmatrix, submatrix, subvector, sumbasis, swapcol, swaprow, sylvester, toeplitz, trace, transpose, vandermonde, vecpotent, vectdim, vector, wronskian*].

7.1 Matrik

7.1.1 Definisi

Matrik adalah kumpulan angka yang ditulis dalam bentuk kolom dan baris. Satu Matrik dikatakan berorde $m \times n$ artinya matrik tersebut mempunyai m baris dan n kolom.

$$\begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22} & \dots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \dots & a_{mn} \end{pmatrix}$$

Angka - angka pada a_{ij} ($1 \leq i \leq m, 1 \leq j \leq n$) dinamakan elemen dari matrik. Pada notasi ini i adalah index baris, dan j adalah index kolom.

7.1.2 Membangun Matrik

Untuk membangun sebuah matrik yang berdimensi $m \times n$ pada Maple digunakan perintah :

```
matrix(m,n[elemen matrik])
```

Contoh 7.1.1. Buatlah suatu matrik yang berdimensi 4×4 , seperti dibawah ini :

$$\begin{pmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \\ 9 & 10 & 11 & 12 \\ 13 & 14 & 15 & 16 \end{pmatrix}$$

untuk membuat matrik yang dimaksud, pada lembar kerja Maple ditulis seperti :

```
[> A := matrix(4,4,[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16]);
```

dan hasilnya akan terlihat seperti dibawah ini:

$$A := \begin{bmatrix} 1 & 2 & 3 & 4 \\ 5 & 6 & 7 & 8 \\ 9 & 10 & 11 & 12 \\ 13 & 14 & 15 & 16 \end{bmatrix}.$$

Maple memberikan juga fasilitas untuk membangun sebuah matrik dalam bentuk simbol seperti :

```
[> m:=matrix(2,2,[a,b,1/2*sqrt(2),d]);
```

$$m := \begin{bmatrix} a & b \\ \frac{1}{2}\sqrt{2} & d \end{bmatrix}.$$

Matrik yang ditulis secara simbolik ini dapat dioperasikan sama seperti matrik - matrik yang elemennya numerik.

7.1.3 Determinan suatu Matrik

Untuk mencari determinan suatu matrik, kita gunakan perintah `det`. Kaedah penulisan perintah `det` adalah :

$$\text{det}(\text{Matrik})$$

Contoh 7.1.2. Hitunglah determinan dari matrik :

$$A := \begin{bmatrix} 1 & 2 \\ 5 & 6 \end{bmatrix}$$

Langkah pertama kita buat dahulu suatu matrik yang berdimensi 2×2 . Dengan memasukkan perintah dibawah ini kedalam lembar kerja Maple,

```
[>M:=matrix(2,2,[1,2,5,6]);
```

$$M := \begin{bmatrix} 1 & 2 \\ 5 & 6 \end{bmatrix}.$$

Untuk mencari determinannya kita gunakan perintah,

```
[> with(linalg):  
[> det(M);
```

-4

hasil determinan dari matrik M adalah -4.

7.1.4 Inverse suatu Matrik

Invers sebuah Matrix oelh Maple dapat dicari dengan menggunakan perintah :

$$\text{inverse}(\text{Matrik});$$

Contoh 7.1.3. Tentukan inverse dari matrik

$$A = \begin{pmatrix} 4 & 5 \\ 5 & 6 \end{pmatrix}.$$

Masukkan pada lembar kerja Maple :

```
[>A := matrix(2,2,[4,5,5,6]);
```

$$A := \begin{bmatrix} 4 & 5 \\ 5 & 6 \end{bmatrix}.$$

lalu kita cari inverse matrik A dengan cara :

```
[>with(linalg):
[> inverse(A);
```

$$A := \begin{bmatrix} -6 & 5 \\ 5 & -4 \end{bmatrix}.$$

Maple dapat juga digunakan untuk mencari inverse dari sebuah matrik dimana elemen matrik tersebut, berupa simbol.

Contoh 7.1.4. Carilah inverse dari matrik dibawah ini :

$$M = \begin{pmatrix} a & b \\ c & d \end{pmatrix}.$$

Langkah pertama, adalah membangun matrik M seperti;

```
[>M:=matrix(2,2,[a,b,c,d]);
```

$$M := \begin{bmatrix} a & b \\ c & d \end{bmatrix}.$$

Inverse dari matrik diatas dapat dicari dengan perintah **inverse**,

```
[> with(linalg):
[> inverse(M);
```

$$\begin{bmatrix} \frac{d}{ad-bc} & -\frac{b}{ad-bc} \\ -\frac{c}{ad-bc} & \frac{a}{ad-bc} \end{bmatrix}.$$

Contoh 7.1.5. Carilah invers dan determinan dari matrik M

$$M := \begin{pmatrix} 2 & 6 & 8 & 10 \\ 17 & 20 & 56 & 45 \\ 9 & 13 & 15 & 18 \\ 19 & 17 & 25 & 3 \end{pmatrix}$$

Langkah pertama adalah membangun Matrik **M** itu sendiri dengan cara :

```
[>M:=matrix(4,4,[2,6,8,10,17,20,56,45,9,13,15,18,19,17,25,3]);
```

$$M := \begin{bmatrix} 2 & 6 & 8 & 10 \\ 17 & 20 & 56 & 45 \\ 9 & 13 & 15 & 18 \\ 19 & 17 & 25 & 3 \end{bmatrix}$$

Selanjutnya adalah mencari dengan menggunakan perintah **inverse**

```
[ with(linalg):
[>inverse(M);
```

$$\begin{bmatrix} \frac{-1667}{3223} & \frac{203}{9669} & \frac{2311}{9669} & -\frac{241}{9669} \\ \frac{2685}{6446} & -\frac{185}{3223} & -\frac{312}{3223} & \frac{172}{3223} \\ \frac{823}{6446} & \frac{208}{9669} & -\frac{1252}{9669} & \frac{277}{9669} \\ -\frac{479}{3223} & \frac{42}{3223} & \frac{367}{3223} & -\frac{161}{3223} \end{bmatrix}.$$

Sedangkan untuk mencari harga determinan dari matrik **M**, kita lakukan dengan perintah **det**

```
[>with(linalg):
[>det(M);
```

dan hasilnya adalah,

$$-19338.$$

7.1.5 Mentranspose matrik

Jika baris dan kolom suatu Matrik **A** dipertukarkan, maka Matrik yang baru kita sebut dengan Matrik Transpose A^T dari Matrik semula.

Contoh 7.1.6. Suatu matrik **A**,

$$A = \begin{pmatrix} 4 & 6 \\ 7 & 9 \\ 2 & 5 \end{pmatrix}.$$

Jika ditranspose akan menghasilkan matrik :

$$A^T = \begin{pmatrix} 4 & 7 & 2 \\ 6 & 9 & 5 \end{pmatrix}.$$

Program Maple menyediakan fasilitas untuk mentranspose suatu Matrik. Perintah tersebut adalah **transpose**. Bentuk dari perintah ini adalah :

```
transpose(M);
```

Contoh 7.1.7. Tranposelah Matrik **H** dibawah ini:

$$\begin{pmatrix} 35 & 79 \\ 20 & 32 \end{pmatrix}.$$

Langkah pertama adalah, buat Matrik **H** yaitu :

```
[>H:=matrix(2,2,[35,79,20,32]);
```

$$H := \begin{bmatrix} 35 & 79 \\ 20 & 32 \end{bmatrix}.$$

Setelah itu, jalankan perintah **Transpose** :

```
[>with(linalg):  
[>transpose(H);
```

dan hasilnya adalah :

$$\begin{bmatrix} 35 & 20 \\ 79 & 32 \end{bmatrix}.$$

7.1.6 Penjumlahan dan Pengurangan Matrik

Supaya dua buah Matrik dapat dijumlahkan atau dikurangkan, maka orde keduanya haruslah sama. Untuk menjumlahkan Matrik **A** dan Matrik **B**, pada program Maple digunakan perintah :

```
evalm(A + B);
```

Sedangkan untuk pengurangan Matrik **A** dan Matrik **B**, digunakan perintah :

```
evalm(A - B);
```

Perintah `evalm` indentik dengan perintah `evalf`, lihat sub-bab 2.1.

Contoh 7.1.8. Dua buah Matrik **A** dan **B** :

$$\mathbf{A} = \begin{pmatrix} 3 & -4 & 5 \\ 8 & 0 & -3 \\ 5 & 2 & 1 \end{pmatrix} \quad \mathbf{B} = \begin{pmatrix} 10 & -6 & -9 \\ 6 & -5 & -7 \\ -10 & 9 & 12 \end{pmatrix}$$

Hitunglah $\mathbf{A} + \mathbf{B}$ dan $\mathbf{A} - \mathbf{B}$

Step pertama yang kita lakukan adalah membangun matrik **A** dan **B** dengan cara :

```
[>A:=matrix(3,3,[3,-4,5,8,0,-3,5,2,1]);
```

$$A := \begin{bmatrix} 3 & -4 & 5 \\ 8 & 0 & -3 \\ 5 & 2 & 1 \end{bmatrix},$$

```
[>B:=matrix(3,3,[10,-6,-9,6,-5,-7,-10,9,12]);
```

$$B := \begin{bmatrix} 10 & -6 & -9 \\ 6 & -5 & -7 \\ -10 & 9 & 12 \end{bmatrix}.$$

Lalu kita jumlahkan dengan cara :

```
[>evalm(A+B);
```

$$\begin{bmatrix} 13 & -10 & -4 \\ 14 & -5 & -10 \\ -5 & 11 & 13 \end{bmatrix}.$$

Untuk proses pengurangan, berikan Matrik **B** tanda *minus*, kemudian lakukan $\mathbf{A} + (-\mathbf{B})$. Seperti dibawah ini :

[>evalm(A - B);

dan hasilnya adalah :

$$\begin{bmatrix} -7 & 2 & 14 \\ 2 & 5 & 4 \\ 15 & -7 & -11 \end{bmatrix}.$$

7.1.7 Perkalian Matrik

Perkalian Skalar

Mengalikan suatu Matrik dengan sebuah bilangan (skalar) artinya kita mengalikan elemen - elemen Matrik dengan bilangan tersebut.

Contoh 7.1.9. Hitung perkalian $4.A$. Dimana A adalah suatu Matrik dari,

$$A = \begin{pmatrix} 7 & 9 \\ 8 & 2 \end{pmatrix}.$$

$4.A$

$$4. \begin{pmatrix} 7 & 9 \\ 8 & 2 \end{pmatrix} = \begin{pmatrix} 28 & 36 \\ 32 & 8 \end{pmatrix}.$$

Maple memberikan fasilitas perkalian sebuah Matrik A dengan skalar n . Perintah tersebut adalah :

`scalarmul(A,n),`

dimana n adalah sebuah bilangan sembarang.

Contoh 7.1.10. Dua buah Matrik A dan B :

$$A = \begin{pmatrix} 3 & -4 & 5 \\ 8 & 0 & -3 \\ 5 & 2 & 1 \end{pmatrix} \quad B = \begin{pmatrix} 10 & -6 & -9 \\ 6 & -5 & -7 \\ -10 & 9 & 12 \end{pmatrix}$$

Hitunglah $4B$ dan $A - 4B$

Pertama bangun Matrik A dengan cara,

[>A:=matrix(3,3,[3,-4,5,8,0,-3,5,2,1]);

$$A := \begin{bmatrix} 3 & -4 & 5 \\ 8 & 0 & -3 \\ 5 & 2 & 1 \end{bmatrix}.$$

Kemudian buat sebuah Matrik B ,

```
[>B:=matrix(3,3,[10,-6,-9,6,-5,-7,-10,9,12]);
```

$$B := \begin{bmatrix} 10 & -6 & -9 \\ 6 & -5 & -7 \\ -10 & 9 & 12 \end{bmatrix}.$$

Untuk mencari $4B$ lakukan,

```
[>scalarmul(B,4);
```

hasilnya seperti :

$$\begin{bmatrix} 40 & -24 & -36 \\ 24 & -20 & -28 \\ -40 & 36 & 48 \end{bmatrix}.$$

Untuk menghitung hasil dari $A - 4B$, dapat dicari dengan memasukkan perintah pada lembar kerja Maple sebagai berikut:

```
[>evalm(A - (scalarmul(B,4)));
```

$$\begin{bmatrix} -37 & 20 & 41 \\ -16 & 20 & 25 \\ 45 & -34 & -47 \end{bmatrix}.$$

Perkalian dua buah Matrik

Dua buah Matrik dapat dikalikan bilamana jumlah kolom pada Matrik pertama sama banyak dengan jumlah baris pada Matrik kedua. Yang perlu diperhatikan dalam perkalian Matrik tidak berlaku hukum komutatif yaitu :

$$\mathbf{A.B} \neq \mathbf{B.A}.$$

Perkalian dua buah Matrik A dan B dengan program Maple, dapat dilakukan dengan menggunakan perintah `multiply`, yang mana kaedah penulisannya adalah :

```
multiply(A,B);
```

atau perkalian Matrik disimbolkan dengan `&*`, dengan format penulisan sebagai berikut,

```
evalm(A&*B);.
```

Contoh 7.1.11. Kembali pada contoh [7.1.8](#); dua buah Matrik A dan B :

$$\mathbf{A} = \begin{pmatrix} 3 & -4 & 5 \\ 8 & 0 & -3 \\ 5 & 2 & 1 \end{pmatrix} \quad \mathbf{B} = \begin{pmatrix} 10 & -6 & -9 \\ 6 & -5 & -7 \\ -10 & 9 & 12 \end{pmatrix}$$

Hitunglah $A.B$.

Masukkan pada lembar kerja Maple

```
[>multiply(A,B);
```

$$\begin{bmatrix} -44 & 47 & 61 \\ 110 & -75 & -108 \\ 52 & -31 & -47 \end{bmatrix}.$$

atau perkalian tersebut dapat juga dilakukan dengan simbol `&*`, seperti dibawah ini :

```
[>evalm(A&*B);
```

$$\begin{bmatrix} -44 & 47 & 61 \\ 110 & -75 & -108 \\ 52 & -31 & -47 \end{bmatrix}.$$

Sekarang bandingkan dengan **B.A**,

```
[>multiply(B,A);
```

$$\begin{bmatrix} -63 & -58 & 59 \\ -57 & -38 & 38 \\ 102 & 64 & -65 \end{bmatrix}.$$

Catatan 7.1.1. Perkalian matrix adalah perkalian baris dan kolom. Maple tidak akan melakukan operasi perkalian, bilamana matrik tersebut baris dan kolomnya tidak sesuai.

7.2 Penyelesaian sistem persamaan linear

Sebuah sistem persamaan linear seperti dibawah ini,

$$\begin{aligned} ax + by + cz &= d \\ ex + fy + gz &= h \\ ix + jx + kz &= l \end{aligned}$$

dapat disusun kedalam bentuk Matrik menjadi sebagai berikut :

$$\begin{pmatrix} a & b & c \\ e & f & g \\ i & j & k \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} d \\ h \\ l \end{pmatrix}$$

$$\mathbf{A} \cdot \mathbf{x} = \mathbf{b}. \tag{7.1}$$

Untuk mencari **x**, ruas kanan dan kiri dari persamaan tersebut dikalikan dengan invers matrik **A**, maka :

$$\begin{aligned} \mathbf{A}^{-1} \cdot \mathbf{A} \cdot \mathbf{x} &= \mathbf{A}^{-1} \cdot \mathbf{b} \\ \mathbf{x} &= \mathbf{A}^{-1} \cdot \mathbf{b} \end{aligned} \tag{7.2}$$

Dengan bentuk seperti persamaan 7.2, Maple dapat langsung memecahkan persoalan diatas dengan perintah `linsolve`. Bentuk penulisan perintah `linsolve` adalah :

`linsolve(A,b).`

Contoh 7.2.1. Selesaikan satu sistem linear dengan tiga variabel yang tidak diketahui seperti dibawah ini :

$$\begin{aligned}x + 2y + 3z &= 1 \\x + 3y &= 2 \\x + 4x + 3z &= 3\end{aligned}$$

Untuk menyelesaikan sistem persamaan tersebut, langkah pertama adalah mengubah bentuk sistem persamaan tersebut dalam bentuk matrik seperti :

$$\begin{bmatrix} 1 & 2 & 3 \\ 1 & 3 & 0 \\ 1 & 4 & 3 \end{bmatrix} \begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}. \quad (7.3)$$

Dengan bentuk seperti persamaan 7.3, dengan mudah diketahui nilai matrik **A** dan **b**, kemudian aplikasikan perintah **linsolve** pada soal diatas,

```
[>with(linalg):  
[>A := matrix(3,3,[1, 2, 3,1, 3, 0, 1, 4, 3]):  
[>b:=matrix(3,1,[1,2,3]):  
[>linsolve(A,b);
```

$[-1, 1, 0]$.

Penyelesaian persamaan diatas adalah $x = -1$, $y = 1$, dan $z = 0$.

7.3 Program Linear

Program linear adalah bagian dari aplikasi matematika yang membahas masalah optimasi. Dalam persoalan program linear terdapat sebuah fungsi, yang dinamakan *fungsi objektif*. Disamping itu, dalam program linear terdapat sejumlah pertidaksamaan atau persamaan yang disebut sebagai kendala, *batasan*. Tujuan utama dalam menganalisa program linear ini adalah menentukan nilai dari variabel - variabel yang memenuhi *batasan* sehingga harga *fungsi objektif* menjadi optimum.

7.3.1 Menggambarkan daerah pertidaksamaan

Cara yang paling mudah memecahkan program linear yang hanya terdiri dari dua variabel, adalah dengan metoda grafik. Dengan menggambarkan semua fungsi batasannya, akan diketahui daerah himpunan penyelesaiannya. Untuk menggambarkan daerah yang memenuhi sebuah pertidaksamaan, pogram Maple mempunyai fasilitas yang disebut `inequal`. Perintah ini ada dalam paket `plots`, lihat sub-bab 5.1.4. Jadi sebelum menggunakan perintah `inequal`, paket **plots** harus dipanggil dahulu. Kaedah penulisan perintah `inequal` adalah :

```
inequal({pertidaksamaan}, batasan_x, batasan_y, opsional).
```


Pada bagian opsional perintah diatas, dapat digunakan untuk mengontrol warna, tebal garis, bentuk garis, dari daerah yang dibatasi oleh pertidaksamaan. Ada 4 jenis pilihan yang ditawarkan oleh Maple, mereka adalah:

1. `optionsfeasible`, daerah yang memenuhi dari semua pertidaksamaan;
2. `optionsexcluded`, daerah yang berada diluar dari paling sedikit satu pertidaksamaan;
3. `optionsopen`, menyatakan garis (batas) dari sebuah daerah, namun garis tersebut tidak masuk didalamnya;
4. `optionsclosed`, menyatakan garis (batas) dari sebuah daerah, dan garis tersebut masuk dalam daerah tersebut.

Contoh 7.3.1. Carilah penyelesaian dari daerah yang dibatasi $3x + 5y \leq 15, x > 0, y > 0$

```
[>with(plots):
[>inequal({3*x+5*y<=15,x>0, y>0}, x = -3..3, y = -3..3,
optionsexcluded=(color=white));
```

Pernyataan `optionsexcluded=(color=white)` mempunyai arti daerah yang tidak memenuhi dari ketiga pertidaksamaan tersebut diberi warna putih. Hasil dari perintah diatas dapat dilihat pada gambar 7.1.




Gambar 7.1: Himpunan penyelesaian, daerah yang dibatasi oleh $3x + 5y \leq 15, x > 0, y > 0$.


Contoh 7.3.2. Tentukan secara grafik, himpunan penyelesaian dari $2x + 5y > 10$.

```
[>with(plots):
[>inequal({2*x+5*y > 10}, x = -1..3, y = -1..3,
optionsexcluded=(color=white), optionsopen=(linestyle=DASHDOT,
thickness=3));
```

Indentik dengan contoh sebelumnya, pernyataan `optionsexcluded=(color=white)` mempunyai arti daerah yang tidak memenuhi dari ketiga pertidaksamaan tersebut diberi warna putih. Sedangkan perintah `optionsopen=(linestyle=DASHDOT, thickness=3)` mengatur garis yang membatasi daerah himpunan penyelesaian. Dalam hal ini garis batas mempunyai *style* garis putus (DASHDOT), dan mempunyai ketebalan 3 (`thickness=3`). Hasil dari perintah diatas dapat dilihat pada gambar 7.2(a).



(a) $2x + 5y > 10$



(b) $3x - y > -6$

Gambar 7.2: Daerah himpunan penyelesaian dari dua buah pertidaksamaan dengan menggunakan program Maple.

Contoh 7.3.3. Tentukan secara grafik, himpunan penyelesaian dari $3x - y > -6$.

```
[>with(plots):
[>inequal({3*x - y > -6}, x = -5..3, y = -2..8,
  optionexcluded=(color=white), optionsopen=(linestyle=DASHDOT,
  thickness=3));
```


Soal ini sama dengan soal sebelumnya, daerah yang memenuhi $3x - y > -6$ dapat dilihat pada gambar 7.2(b). Sekarang akan diperlihatkan bagaimana fasilitas `inequal` dapat menyelesaikan persoalan program linear dengan metoda grafik. Perhatikan contoh dibawah ini.

Contoh 7.3.4. Maksimumkan sebuah fungsi $z = 150x + 100y$, dengan batasan - batasan :

$$\begin{aligned} 200x + 100y &\leq 5000 \\ 20x + 15y &\leq 600 \\ x &\geq 0 \\ y &\geq 0 \end{aligned}$$

Langkah pertama adalah memvisualisasikan daerah daerah yang menjadi *batasan* yaitu $200x + 100y \leq 5000, 20x + 15y \leq 600, x \geq 0$ dan $y \geq 0$. Untuk menggambarkan daerah - daerah *batasan* tersebut digunakan perintah `inequal`. Perintah ini telah dijelaskan dalam subbab sebelumnya. Untuk memulai menggambarkan daerah - daerah batasan, pada lembar kerja Maple dimasukkan perintah sebagai berikut:

```
[> with(plots):
[> inequal({200*x +100*y <= 5000, 20*x+15*y <= 600, x>=0, y>=0},
  x=0..50, y=0..50, optionexcluded(color=black, thickness=3));
```



Gambar 7.3: Daerah yang berwarna abu - abu, merepresentasikan daerah yang memenuhi fungsi batasan.

Terlihat dari gambar 7.3 titik titik ekstrimnya dari daerah yang berwarna abu - abu yaitu pada titik **A**, **B** dan **C**. Untuk menentukan titik **A**, harga $x = 0$ disubstitusikan ke persamaan $20x + 15y = 600$, sehingga harga y diketahui. Untuk keperluan tersebut perintah `subs`, `fsolve`, atau `solve`, digunakan, seperti dibawah ini,

```
[>fsolve(subs(x=0, 20*x+15*y=600));
```

40.

Dari perhitungan diatas, program Maple menghasilkan 40, ini bearti koordinat titik **A** adalah (0,40).

Titik **B** adalah perpotongan garis antara garis $200x + 100y = 5000$ dan $10x + 15y = 600$. Untuk mencari titik potong kedua garis ini, aplikasikan perintah `solve` atau `fsolve`,

```
[> solve([200*x+100*y=5000, 20*x+15*y=600], {x,y});
```

$\{x = 15, y = 20\}$.

Koordinat titik **B** terletak pada (15,20).

Titik terakhir yang harus ditentukan adalah titik **C**, untuk menentukan titik **C**, harga $y = 0$ disubstitusikan ke persamaan $200x + 100y = 5000$, kemudian hasil dari substitusi tersebut dipecahkan dengan dengan menggunakan [erintah `fsolve` pada Maple.

```
[>fsolve(subs(y=0, 200*x+100*y=5000));
```

25.

Hasil ini menandakan bahwa koordinat titik **C** adalah $(25, 0)$. Tugas selanjutnya adalah; di antara ketiga titik tersebut, **A** $(0, 40)$, **B** $(15, 20)$, dan **C** $(25, 0)$ mana yang membuat $z = 15x + 100y$ menjadi maksimum. Untuk keperluan tersebut, fungsi $z(x, y) \Rightarrow 15x + 100y$ dibuat dahulu dengan menggunakan Maple,

```
[>z := (x, y) -> 15*x+100*y;
```

$$z := (x, y) \rightarrow 15x + 100y.$$

Kemudian, untuk mencari titik mana yang membuat fungsi z maksimum adalah dengan mensubstitusi satu - persatu koordinat **A**, **B**, dan **C**.

Koordinat titik **A**;

```
[>z (0, 40) ;
```

4000.

Koordinat titik **B**;

```
[>z (15, 20) ;
```

4250.

Koordinat titik **C**;

```
[>z (25, 0) ;
```

3750.

Dari hasil diatas, terlihat bahwa kordinat titik **B** $(15,20)$ membuat harga $z = 150x + 100y$ menjadi maksimum, yakni 4250.

7.3.2 Paket Simplex

Metode yang sering digunakan untuk memecahkan persoalan Program Linear selain metoda grafik, adalah metode Simplex. Metode ini sendiri tidak akan dibahas dalam subbab ini. Pada subbab ini hanya diperkenalkan paket metoda Simplex yang ada pada Maple, dan bagaimana menggunakan metoda tersebut untuk menyelesaikan Program Linear. Agar bisa bekerja dengan metoda Simplex, pengguna Maple harus memanggilnya terlebih dahulu, dengan perintah **with(simplex)**. Saat paket tersebut dipanggil, maka pada lembar kerja Maple akan keluar sejumlah perintah seperti dibawah ini:

```
[> with(simplex);
```

basis, convexhull, cterm, define_zero, display, dual, feasible, maximize, minimize, pivot, pivoteqn, pivotar, ratio, setup, standardize.

Dari sekian perintah yang menjadi fokus pada subbab ini adalah perintah `minimize` dan `maximize`. Karena kedua perintah inilah yang paling sering digunakan untuk memecahkan Program Linear tingkat SMU. Bentuk dari perintah `maximize` dan `minimize` adalah sebagai berikut :

```
minimize( Pernyataan, opsional ),
maximize( Pernyataan, opsional ).
```

Untuk lebih jelas penggunaan perintah diatas, perhatikan contoh dibawah ini.

Contoh 7.3.5. Berapakah harga maksimum fungsi $z = 6x + 8y$, jika fungsi batasannya adalah : $5x + 2y \leq 20, x + 2y \leq 10, x \geq 0$ dan $y \geq 0$.

Step pertama yang dilakukan untuk memecahkan persoalan diatas adalah, dengan mendefinisikan dahulu fungsi yang akan dimaksimumkan dan fungsi - fungsi batasannya.

```
[>z:=6*x + 8*y;
```

$$z := 6x + 8y$$

```
[>batasan:= {5*x+2*y<=20, x+2*y<=10, x>=0, y >=0};
```

$$\text{batasan} := 5x + 2y \leq 20, x + 2y \leq 10, x \geq 0, y \geq 0.$$

Langkah selanjutnya adalah mengaplikasikan perintah `maximize`, seperti yang telah dijelaskan diatas. Agar mudah digunakan, hasil dari perintah `maximize` tersebut diberi nama `solusi`. Jika pada lembar kerja Maple dimasukkan :

```
[>solusi:= maximize(z,batasan)
```

maka seketika Maple akan memberikan jawaban :


$$\text{solusi} := \{y = 15/4, x = 5/2\}.$$

Untuk mengetahui berapa harga z maksimum, substitusikan harga x dan y diatas kedalam persamaan z dengan perintah `subs`.

```
[>subs(solusi, z);
```

45.

Hasil akhir dari persoalan diatas adalah; harga z maksimum adalah 45 saat $x = 5/2$ dan $y = 15/4$. Apabila soal ini serta solusinya ingin ditampilkan dalam bentuk grafik, perintah Maple `inequal` digunakan seperti :



Gambar 7.4: Daerah yang berbayang (abu - abu) adalah daerah himpunan jawaban yang memenuhi, fungsi z dan batasan - batasannya.

```
[> with(plots):
[> inequal( {5*x+2*y<=20,x+2*y<=10,x>=0,y>=0},
  x=-1..10,y=-1..10, optionsexcluded=(color=white) );
```

Contoh 7.3.6. Tentukan nilai minimum fungsi objektif $z = 3x + 4y$, jika fungsi batasannya adalah :

$$\begin{aligned} 2x + 3y &\geq 12 \\ 5x + 2y &\geq 19 \\ x \geq 0, \quad y &\geq 0 \end{aligned}$$

Langkah pertama, definisikan semua fungsi kedalam sebuah variabel.

```
[> z:= 3*x+4*y;
```

$$z := 3x + 4y,$$

```
[> batasan:={2*x+3*y >=12, 5*x+2*y >= 19, x>=0, y >=0};
```

$$\text{batasan} := \{2x + 3y \geq 12, 5x + 2y \geq 19, x \geq 0, y \geq 0\}.$$

Cari harga minimum dengan metoda Simplex yang terdapat pada Maple. Berikan nama variabel untuk nilai x dan y pembuat minimum fungsi z . Namakan hasil.

```
[>with(simplex):
[>hasil:=minimize(z,batasan);
```

$hasil\{y = 2, x = 3\}$.


Substitusikan hasil diatas ke fungsi z , dengan menggunakan perintah `subs`.

```
[> subs(hasil, z);
```

17.

Untuk memvisualisasikan persoalan Program Linear ini, kita gunakan perintah `inequal` yang terdapat pada Maple.

```
[> with(plots):  
[> inequal( batasan, x=-1..10,y=-1..10, optionsexcluded=(color=white),  
optionsfeasible=(color=grey) );
```



Gambar 7.5: Daerah yang bewarna putih adalah daerah himpunan jawaban yang memenuhi, fungsi z dan batasan - batasannya. Titik **B** (3,2) adalah pembuat fungsi z minimum.

7.4 Vektor

Pada kenyataannya vektor adalah sebuah matrik yang hanya mempunyai satu baris atau satu kolom saja. Pada Maple untuk menuliskan vektor terdapat dua cara yaitu `Vector` (diawali dengan huruf kapital) atau `vector`. Cara penulisannya adalah sebagai berikut :

```
vector([elemen vektor]).
```

Perhatikan dua contoh dibawah ini untuk melihat perbedaan antara penulisan `Vector` dan `vector`.

```
[> Vector([1,2,3]);
```

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

Penulisan `vector` akan menghasilkan sebuah vektor kolom. Bandingkan sekarang dengan:

```
[> vector([1,2,3]);
```

$$[1 \ 2 \ 3]$$

Maple akan membuat sebuah vektor baris apabila penulisan memakai `vector`.

7.4.1 Memvisualkan Vektor

Bekerja dengan vektor akan lebih mudah, bila vektor tersebut dapat divisualisasikan. Pada subbab ini kita akan melihat bagaimana Maple memberikan kemudahan dalam memvisualkan vektor. Untuk dapat menggambar vektor, kita harus terlebih dahulu memanggil paket **plots**, hal ini sudah dijelaskan pada bab ???. Salah satu fasilitas yang ada pada paket **plots** adalah `arrow`, yang artinya dalam bahasa Indonesia anak panah. Dengan perintah `arrow` inilah sebuah vektor dapat divisualisasikan oleh Maple. Perintah untuk membuat sebuah vektor pada Maple adalah :

```
arrow(koordinat awal, arah vektor, opsional);
```

Vektor Berdimensi Dua

Koordinat awal vektor adalah tempat koordinat dimana awal vektor mulai bergerak. Koordinat tersebut dapat ditulis sebagai $[a, b]$ atau $\langle a, b \rangle$. Pernyataan arah vektor adalah besarnya vektor dalam arah sumbu x dan y , bilamana vektor tersebut dua dimensi. Arah vektor ditulis dalam bentuk $[c, d]$ atau $\langle c, d \rangle$.

Contoh 7.4.1. Gambarlah vektor $\mathbf{a} = \begin{pmatrix} 3 \\ 2 \end{pmatrix}$.


Pada lembar kerja Maple kita masukkan perintah :

```
[> with(plots):
[> arrow(<3,2>);
```

Perintah `arrow(<3,2>)` hanya menginstruksikan Maple untuk membuat vektor dengan arah 3 satuan pada sumbu x dan 2 satuan pada sumbu y , hasilnya dapat dilihat pada gambar 7.6(a). Jika tidak diberi perintah apa-apa lagi, maka Maple akan mengerjakan vektor dari titik $(0,0)$. Demikian juga bentuk vektor yang dibuat Maple sesuai dengan default-nya yakni seperti yang terlihat pada gambar 7.6(a).

Contoh 7.4.2. Gambarlah vektor $\mathbf{a} = \begin{pmatrix} 3 \\ 2 \end{pmatrix}$, yang berawal dari titik $(1,1)$.

Untuk membuat gambar yang dimaksud, pada lembar kerja Maple dimasukkan perintah :



Gambar 7.6:

```
[> with(plots):
[> arrow(<1,1>, <3,2>);
```

Hasilnya dapat dilihat pada gambar 7.6(b). Kita dapat lihat dengan adanya tambahan koordinat $\langle 1, 1 \rangle$ pada awal perintah `arrow`, maka vektor $\begin{pmatrix} 3 \\ 2 \end{pmatrix}$ dibuat oleh Maple dengan titik awal $(0,0)$.

Tampilan vektor akan menjadi lebih menarik, bila parameter - parameter tertentu pada bagian opsional dari perintah `arrow` diberikan. Pada bagian opsional ini, pengguna Maple dapat mengubah: warna vektor, bentuk vektor, ketebalan garis vektor dan "jendela pandang" (*view*) dari vektor yang kita buat.

Bentuk Vektor

Bentuk vektor didefinisikan dengan perintah `shape`. Untuk vektor dua dimensi, ada 3 jenis bentuk vektor yang dapat dibuat oleh Maple. bentuk bentuk tersebut adalah :

1. `harpoon`, \rightarrow .
2. `arrow`, \rightarrow .
3. `double_arrow`, \Rightarrow , seperti yang terlihat pada gambar 7.6(a) dan (b).

Warna Vektor

Warna vektor dapat dibuat dengan perintah `color`. Hal ini telah dibahas pada bab ???. Warna - warna yang dikenal Maple antara lain:

black	blue	cyan	green	gray	magenta	maroon	orange
navy	pink	red	sienna	violet	wheat	white	yellow

Ketebalan garis


Untuk mengatur ketebalan garis, Maple menyediakan perintah **thickness=n**. n adalah bilangan bulat yang menyatakan tebal garis. Semakin besar harga n , maka garis yang kita buat semakin tebal.

Jendela Pandang

Yang dimaksud "Jendela Pandang" adalah besarnya grafik yang kita buat. Kita bisa mengatur panjang dan lebar dari grafik yang kita buat dengan perintah `view=[xmin..xmax, ymin..ymax]`. Dimana x_{\min} adalah batasan sumbu x minimum, x_{\max} adalah batasan sumbu x maksimum, y_{\min} adalah batasan sumbu y minimum, dan y_{\max} adalah batasan sumbu y maksimum.

Contoh 7.4.3. Gambarlah vektor $\begin{pmatrix} 3 \\ 2 \end{pmatrix}$ yang bertitik awal dari titik $(1, 2)$.

```
[> with( plots ):  
[> arrow( [1,2], <3,2>, shape=arrow, color=black, thickness=3,  
view=[-5..5, -5..5] ):
```



Gambar 7.7: Gambar sebuah vektor yang dihasilkan oleh program Maple.

Dengan perintah diatas, kita akan membuat vektor dengan arah $\begin{pmatrix} 3 \\ 2 \end{pmatrix}$, dari titik $(1, 2)$, vektor yang kita buat berbentuk panah (`shape = arrow, →`), warna hitam (`color = black`), dan tebal garis 3 (`thickness = 3`). Jendela pandang yang didefinisikan oleh `view=[-5..5, -5..5]`, akan menghasilkan pandangan dengan sumbu x dari -5 sampai 5 dan sumbu y dari -5 sampai 5. Hasil dari perintah tersebut dapat dilihat pada gambar 7.7.


Sekarang bagaimana bila kita ingin menggambar beberapa vektor dalam satu gambar. Pada prinsipnya sama dengan bilamana kita hendak menggambar beberapa grafik dalam satu gambar. Hal ini telah dibahas dalam bab ???. Perintah yang digunakan untuk keperluan ini adalah `display`.

Contoh 7.4.4. Gambarlah vektor **a** dan **b** pada titik $(1, 2)$.

$$a = \begin{pmatrix} 3 \\ 2 \end{pmatrix}, \quad b = \begin{pmatrix} -2 \\ 3 \end{pmatrix}.$$

Kita masukkan perintah dibawah ini pada lembar kerja Maple,

```
[>with(plots):  
[>a := arrow( [1,2],<3,2>, shape=arrow, color=red ):  
[>b := arrow( [1,2],<-2,3>, shape=arrow, color=blue ):  
[> display( a, b, view=[-4..4,-6..6] );
```



Gambar 7.8: Gambar dua buah vektor yang dihasilkan oleh program Maple.

Vektor Dimensi Tiga

Membuat vektor yang berdimensi tiga, pada prinsipnya sama seperti membuat vektor berdimensi dua. Jika komponen sebuah vektor berjumlah tiga, maka Maple secara otomatis membacanya sebagai vektor yang berdimensi tiga.

Contoh 7.4.5. Gambarlah vektor berdimensi tiga $\begin{pmatrix} 2 \\ -1 \\ 2 \end{pmatrix}$.

```
[> with(plots):  
[> arrow(<2, -1, 2>, view=[-5..5,-5..5,-5..5],shape=arrow,  
thickness=3, axes normal);
```

Perintah opsi `axes = normal`, menyatakan bahwa vektor tersebut di-plot pada salib sumbu (x, y, z) biasa. Hasil dari perintah tersebut dapat dilihat pada gambar 7.9(a). Demikian halnya bila amana kita hendak memplot lebih dari satu vektor ke sebuah salib sumbu. Dalam hal ini perintah `display` pada Maple digunakan, lihat contoh 7.4.4.


Contoh 7.4.6. Gambarlah dua buah vektor dibawah ini, dalam dimensi tiga.

$$a = \begin{pmatrix} 2 \\ -1 \\ 2 \end{pmatrix}, \quad b = \begin{pmatrix} -2 \\ 4 \\ 1 \end{pmatrix}.$$

Sama halnya seperti contoh 7.4.4, pada lembar kerja Maple dimasukkan :

```
[> with(plots):
[> a := arrow(<2,-1,2 >, shape=arrow, color=red, thickness=3 ):
[> b := arrow(<-2,4,1>, shape=arrow, color=blue, thickness=3 ):
[> display( a, b, axes=normal,);
```

hasilnya terlihat pada gambar 7.9(b).



Gambar 7.9: Vektor - vektor yang berada pada ruang berdimensi tiga.

Gambar 7.9 dapat diekport kedalam bentuk JPG atau file gambar lainnya, untuk digunakan sebagai bahan presentasi atau lainnya. Cara - cara mengekspor gambar telah dijelaskan pada sub-bab ??.

7.4.2 Aljabar Vektor

Penjumlahan dan Pengurangan Vektor

Misalkan $\mathbf{a} = \begin{pmatrix} x_1 \\ y_1 \\ z_1 \end{pmatrix}$, dan $\mathbf{b} = \begin{pmatrix} x_2 \\ y_2 \\ z_2 \end{pmatrix}$. Jika $\mathbf{c} = \mathbf{a} + \mathbf{b}$, maka $\mathbf{c} = \begin{pmatrix} x_1 + x_2 \\ y_1 + y_2 \\ z_1 + z_2 \end{pmatrix}$. Pada Maple perintah untuk menjumlahkan atau mengurangkan Vektor adalah sama dengan proses penjumlahan dan pengurangan untuk aritmatika.


Contoh 7.4.7. Carilah $\mathbf{r} = \mathbf{a} + \mathbf{b}$, jika :

$$\mathbf{a} = \begin{pmatrix} 5 \\ 2 \end{pmatrix} \quad \mathbf{b} = \begin{pmatrix} 1 \\ 4 \end{pmatrix}$$

Lukiskan \mathbf{a} , \mathbf{b} dan \mathbf{r} dengan menggunakan Maple.

Langkah pertama; buat vektor \mathbf{a} dari titik (0,0). Kemudian buat vektor \mathbf{b} dengan titik awalnya adalah titik (5,2). Sedang vektor \mathbf{r} dibuat dengan menjumlahkan vektor \mathbf{a} dan vektor \mathbf{b} , dan digambar dengan titik awal (0,0). Terakhir ketiga vektor tersebut ditampilkan secara bersamaan dengan menggunakan perintah `display`. Langkah -langkah ini dituliskan pada lembar kerja Maple sebagai berikut :

```
[> with( plots ):
[> a := arrow( <5,2>, shape= arrow, color=blue ):
[> b := arrow( <5,2>, <1,4>, shape=arrow, color=blue ):
[> r := arrow( <5,2> + <1,4>, shape=arrow, color=red ):
[> display( a, b,r, scaling=constrained,
[>view=[-8..8,-8..8] );
```



Gambar 7.10: Grafik dari penjumlahan vektor $r = a + b$.

Untuk mengetahui arah r , maka pada lembar kerja Maple dimasukkan :

```
[>a := Vector([5,2]):
[>b := Vector([1,4]):
[>r := a + b;
```

$$r := \begin{pmatrix} 6 \\ 6 \end{pmatrix}.$$

Hal yang sama untuk kasus vektor yang berdimensi tiga.

Contoh 7.4.8. Diketahui vektor $\mathbf{a} = \begin{pmatrix} 3 \\ 2 \\ -1 \end{pmatrix}$, $\mathbf{b} = \begin{pmatrix} 2 \\ -3 \\ 4 \end{pmatrix}$. Nyatakan $\mathbf{c} = \mathbf{a} + \mathbf{b}$, dan $\mathbf{d} = \mathbf{a} - \mathbf{b}$ dalam bentuk kolom vektor.

Pertama buat vektor \mathbf{a} dan vektor \mathbf{b} , dengan cara :

```
[>a:=Vector([3,2,-1]);
```

$$a := \begin{bmatrix} 3 \\ 2 \\ -1 \end{bmatrix}.$$

```
[>b:=Vector([2,-3,4]);
```

$$b := \begin{bmatrix} 2 \\ -3 \\ 4 \end{bmatrix}.$$

Untuk mendapatkan $c = a + b$, maka pada Maple kita masukkan :


```
[>c:= a+b;
```

$$c := \begin{bmatrix} 5 \\ -1 \\ 3 \end{bmatrix}.$$

Untuk memvisualkan $c = a + b$, lakukan perintah ini pada lembar kerja Maple,

```
[>with(plots):  
[>a:= arrow( <3,2,-1>, shape=arrow, color=red,thickness=3):  
[>b:=arrow( [3, 2,-1],<2,-3,4>, shape=arrow, color=blue,thickness=3):  
[>c:=arrow( <3,2,-1>+<2,-3,4>, shape=arrow, color=black, thickness=3):  
[>display(a,b,c, axes=normal);
```

Hasilnya dapat dilihat pada gambar 7.11(a).



Gambar 7.11: Visualisasi penjumlahan dan pengurangan vektor yang berdimensi tiga.

Hal yang sama untuk $d = a - b$. Hanya saja sekarang vektor b diubah menjadi $-b$.

```
[>d:= a+(-b);
```

$$d := \begin{bmatrix} 1 \\ 5 \\ -5 \end{bmatrix}.$$

Untuk memvisualisasikan $\mathbf{d} = \mathbf{a} + (-\mathbf{b})$, pada lembar kerja Maple dimasukkan :

```
[>with(plots):
[>a:= arrow( <3,2,-1>, shape=arrow, color=red,thickness=3):
[>b:=arrow( [3, 2,-1],<-2,3,-4>, shape=arrow, color=blue,thickness=3):
[>c:=arrow( <3,2,-1>+<-2,3,-4>, shape=arrow, color=black, thickness=3):
[>display(a,b,c, axes=normal);
```

Hasil dari visualisasi ini dapat dilihat pada gambar 7.11(b).

7.4.3 Perkalian Vektor

Dalam aljabar linear dikenal dua jenis perkalian Vektor yaitu perkalian titik, *dot product* yang didefinisikan sebagai,

$$A \cdot B = |A| \cdot |B| \cos(\theta). \quad (7.4)$$

Hasil perkalian ini menghasilkan besaran skalar. Jenis perkalian yang kedua adalah perkalian silang, *cross product*, yang didefinisikan sebagai,

$$A \times B = |A| \cdot |B| \sin(\theta). \quad (7.5)$$

Hasil perkalian silang ini menghasilkan besaran vektor. Lambang θ pada kedua persamaan adalah sudut apit antara dua vektor A dan B . Program Maple telah menyediakan fasilitas perkalian vektor ini yakni dengan perintah **dotprod** untuk perkalian titik, dan **crossprod** untuk perkalian silang. Bentuk perintah **dotprod** dan **crossprod** pada Maple secara berturut turut adalah :

dotprod(A,B),

dan

crossprod(A,B).

Kedua perintah ini berada dalam paket **linalg**, jadi sebelum menggunakan perintah **dotprod** dan **crossprod**, paket **linalg** harus dipanggil terlebih dahulu dengan **with(linalg)**.

Contoh 7.4.9. Dua buah vektor yaitu A dengan komponen $[1, 2, 3]$ dan Vektor B dengan komponen $[2, -3, 5]$. Hitunglah $\mathbf{A} \cdot \mathbf{B}$ dan $\mathbf{A} \times \mathbf{B}$.

```
[>with(linalg):
[>A:=vector([1,2,3]);
```

$A := [1, 2, 3],$

```
[> B:=vector([2,-3,5]);
```

$$B := [2, -3, 5].$$

Untuk melakukan perkalian titik, dimana hasil perkalian ini adalah bentuk skalar, maka pada lembar kerja Maple dimasukkan :

```
[>with(linalg):  
[>s:=dotprod(A,B);
```

$$s := 11,$$

Hasil dari $\mathbf{A} \cdot \mathbf{B}$ adalah 11, sedang untuk melakukan perkalian silang,

```
[>with(linalg):  
[>C:=crossprod(A,B);
```

$$C := [19, 1, -7],$$

Hasil dari $\mathbf{A} \times \mathbf{B}$ adalah vektor $C = [19, 1, -7]$.

7.4.4 Sudut antara dua Vektor

Paket **linalg** pada Maple, menyediakan fasilitas untuk mencari sudut apit antara dua buah vektor. Perintah tersebut adalah `angle` dan kaedah penulisannya seperti :

$$\text{angle}(A, B).$$

\mathbf{A} dan \mathbf{B} adalah dua buah vektor yang akan dicari besar sudut antaranya.

Contoh 7.4.10. Carilah sudut apit dua buah vektor $\mathbf{P}=[2, 3, 4]$ dan $\mathbf{Q}=[4, -3, 2]$.

Langkah pertama adalah: buat vektor \mathbf{P} dan \mathbf{Q} dengan Maple:

```
[>P:=vector([2,3,4]);
```

$$\mathbf{P} := [2, 3, 4],$$

```
[> Q:=vector([4,-3,2]);
```

$$\mathbf{Q} := [4, -3, 2].$$

Untuk mencari besarnya sudut apit dari dua vektor ini (dalam radian),

```
[>sudut:=angle(P,Q);
```

$$\text{sudut} := \arccos\left(\frac{7}{29}\right)$$

Untuk menghitungnya dalam desimal, maka kita gunakan perintah **evalf**,

```
> evalf(sudut);
```

hasilnya adalah :

1.327009388.

Hasil diatas masih dalam bentuk radian, dan dapat dikonversikan kedalam derajat dengan mengalikannya $180/\pi$.

```
[>evalf(sudut*180/Pi);
```

dan hasilnya :

76.03203730.

Sudut antara vektor $\mathbf{P}=[2, 3, 4]$ dan $\mathbf{Q}=[4, -3, 2]$, adalah 76.03203730 derajat.

7.5 Soal - soal dan penyelesaiannya

1. Hitunglah matrik $\mathbf{A.B}$ dan $\mathbf{B.A}$ jika

$$A = \begin{pmatrix} -1 & -5 & -5 & -4 \\ -3 & 5 & 3 & -2 \\ -4 & 4 & 2 & -3 \end{pmatrix}, \quad B = \begin{pmatrix} 1 & -2 \\ -4 & 3 \\ 4 & -4 \\ -5 & -3 \end{pmatrix}$$

Karena \mathbf{A} adalah matrik 3×4 dan \mathbf{B} adalah matrik 4×2 , maka $\mathbf{A.B}$ adalah matrik 3×2 . Untuk mengerjakan soal ini dengan Maple, pertama buatlah matrik \mathbf{A} dan \mathbf{B} , dengan menggunakan paket `linalg`.

```
[> with(linalg):  
[> A:=matrix(3,4,[-1,-5,-5,-4,-3,5,3,-2,-4,4,2,-3]):  
[> B:=matrix(4,2,[1,-2,-4,3,4,-4,-5,-3]):
```

Kemudian kedua kalikan \mathbf{A} dan \mathbf{B} dengan perintah `evalm` dan `multiply`.

```
[>evalm(A&*B);
```

$$\begin{pmatrix} 19 & 19 \\ -1 & 15 \\ 3 & 21 \end{pmatrix}.$$

Atau soal diatas dapat juga dikerjakan dengan perintah `multiply`, seperti

```
[>multiply(A,B);
```

$$\begin{pmatrix} 19 & 19 \\ -1 & 15 \\ 3 & 21 \end{pmatrix}.$$

Perkalian $\mathbf{B.A}$ tidak dapat dikerjakan oleh Maple, hal ini dikarenakan baris dan kolom matrik \mathbf{B} \mathbf{A} tidak memenuhi persyaratan perkalian dua buah matrik. Jika kedua matrik ini dioperasikan perkalian dengan perintah `multiply`, maka hasilnya :

```
[> multiply(B,A);
```

Error, (in linalg:-multiply) non matching dimensions for vector/matrix product.

2. Hitung \mathbf{B}^3 untuk

$$B = \begin{pmatrix} -2 & 3 & 4 & 0 \\ -2 & 0 & 1 & 3 \\ -1 & 4 & -6 & 5 \\ 4 & 8 & 11 & -4 \end{pmatrix}$$

Bangun matrik B dengan Maple,

```
[>B:=matrix(4,4,[-2,3,4,0,-2,0,1,3,-1,4,-6,5,4,8,11,-4]);
```

Untuk mencari matrik dari B^3 ,

```
[> evalm(B^3);
```

$$\begin{pmatrix} 137 & 98 & 479 & -231 \\ -121 & 65 & -109 & 189 \\ -309 & 120 & -871 & 646 \\ 520 & 263 & 1381 & -738 \end{pmatrix}.$$

3. Selesaikan sistem persamaan dibawah ini. Hitunglah nilai x, y, z .

$$\begin{aligned} x - 2y + z &= -4 \\ 3x + 2y - z &= 8 \\ -x + 3y + 5z &= 0 \end{aligned}$$

Ada dua cara untuk menyelesaikan soal diatas, cara pertama adalah dengan memakai perintah `fsolve` atau `solve` seperti dibawah ini:

```
[>solve({x-2*y+z=-4,3*x+2*y-z=8,-x+3*y+5*z=0},{x,y,z});
```

$$\{z = -1, y = 2, x = 1\}.$$

Cara kedua adalah dengan membangun sistem persamaan tersebut kedalam bentuk matrix, seperti dibawah ini :

$$\begin{pmatrix} 1 & -2 & 1 \\ 3 & 2 & -1 \\ -1 & 3 & 5 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} -4 \\ 8 \\ 0 \end{pmatrix}$$

Bentuk ini indentik dengan bentuk :

$$Ax = b$$

Untuk mencari nilai x , dipakai perintah `linsolve`. Dari persamaan diatas, matrik A dapat dibuat dengan :

```
[>with(linalg):
[> A:=matrix(3,3,[1,-2,1,3,2,-1,-1,3,5]):
[> b:=Vector([-4,8,0]):
[> linsolve(A,b);
```

$$[1,2,-1].$$

Hasilnya adalah $x = 1, y = 2, \text{ dan } z = -1$.

4. Rokok **A** yang harganya 500 perbungkus dijual dengan keuntungan 50 perbungkus. Sedangkan rokok **B** yang harganya 1000 dijual dengan keuntungan 100 perbungkus. Seorang pedagang rokok mempunyai modal 500.000, dan mempunyai kios yang dapat menampung 600 bungkus rokok. Tentukan banyaknya rokok **A** dan **B** yang harus dia beli agar keuntungan pedagang rokok ini maksimum.

Rokok **A** dimisalkan x dan rokok **B** dimisalkan y . Dari soal diketahui fungsi objektifnya (keuntungan maksimum) adalah $z = 50x + 100y$. Sedangkan fungsi batasannya adalah :

$$\begin{aligned} 500x + 1000y &\leq 500.000 \\ x + y &\leq 600 \\ x \geq 0, \quad y &\geq 0 \end{aligned}$$

Soal ini akan dipecahkan dengan menggunakan Maple dengan fasilitas `simplex` nya. Langkah pertama definisikan semua fungsi nya.

```
[> z := 50*x + 100*y;
```

$$z := 50x + 100y.$$

```
[> batasan:={x + y <=600, 500*x + 1000*y <= 500000,
x >=0, y >=0};
```

$$\textit{batasan} := x + y \leq 600, 500x + 1000y \leq 500000, x \geq 0, y \geq 0$$

Langkah kedua panggil paket `simplex`, dan gunakan perintah `maximize`,

```
[> with(simplex):
[> hasil:=maximize(z, batasan);
```

$$\textit{hasil} := \{x = 200, y = 400\}.$$

Hasil ini memperlihatkan harga z akan menjadi maksimum jika $x = 200$ dan $y = 400$. Untuk mencari harga z , substitusikan x dan y ke fungsi objektifnya z .

```
[>subs(hasil, z);
```

50000.

Keuntungan maksimum yang dapat dicapai oleh pedagang rokok tersebut adalah 50000 dengan penjualan 200 rokok **A** dan 400 rokok **B**. Untuk memvisualkan Program Linear ini, perintah `inequal` dari Maple, digunakan kembali,

```
[>with(plots):
[>inequal(batasan, x=-100..800, y = -100..800,
optionsexcluded=(color=white))
```

5. Buktikan vektor $\mathbf{X} = (2,-4)$ dan $\mathbf{Y} = (4,2)$ saling tegak lurus.


Dalam perkalian titik, diketahui :

$$\begin{aligned} \mathbf{X} \cdot \mathbf{Y} &= \|\mathbf{X}\| \|\mathbf{Y}\| \cos(\theta) \\ \cos(\theta) &= \frac{\mathbf{X} \cdot \mathbf{Y}}{\|\mathbf{X}\| \|\mathbf{Y}\|} \end{aligned}$$

Jika \mathbf{X} dan \mathbf{Y} saling tegak lurus, maka $\theta = 90^\circ$, itu berarti $\cos(\theta) = 0$, dan

$$\mathbf{X} \cdot \mathbf{Y} = 0$$

Kita dapat mengerjakan hal ini dengan Maple, dengan perintah `dotprod`, seperti dibawah ini;



```
[>X :=vector([2,-4]);
```

$$X := [2, -4]$$

```
[>Y :=vector([4,2]);
```


$$Y := [4, 2]$$

```
[>dotprod(X,Y);
```

0.

Dari hasil perhitungan ini terlihat bahwa vektor **X** dan **Y** saling tegak lurus. Dengan Maple, kedua vektor tersebut dapat divisualisasikan :

```
[>with( plots ):
[> X := arrow( <2,-4>, shape= arrow, color=blue ):
[> Y := arrow( <4,2>, shape=arrow, color=blue ):
[> display( X, Y, scaling=constrained,
view=[-4..4,-4..4] );
```



7.6 Soal - soal latihan

1. Matrik **A** dan **E** didefinisikan sebagai berikut:

$$A = \begin{bmatrix} 3 & -1 & 3 \\ 4 & 1 & 5 \\ 2 & 1 & 3 \end{bmatrix}, \quad E = \begin{bmatrix} 2 & -4 & 5 \\ 0 & 1 & 4 \\ 3 & 2 & 1 \end{bmatrix}$$

Berapakah :

- $\mathbf{C} + \mathbf{E}$;
 - $(\mathbf{C} + \mathbf{E})^T$;
 - \mathbf{C}^{-1} ;
 - $\mathbf{C} \cdot \mathbf{E}$;
 - $2\mathbf{C} - 3\mathbf{E}$.
2. Hitunglah determinant matrik **A** (lihat soal nomor.1), apakah matrik tersebut *singular*.
3. Selesaikanlah sistem persamaan linear dibawah ini:

$$2x + 3y + z = 13$$

$$4x + 2y + z = 17$$

$$x + 3y + 2z = 11$$

4. Luas suatu daerah parkir adalah 100 m^2 . Luas rata - rata sebuah mobil 10 m^2 dan bus 20 m^2 . Daerah parkir tersebut tidak boleh menampung lebih dari 8 kendaraan. Jika tarif parkir mobil 2000 rupiah, dan tarif parkir bus 5000 rupiah. Berapakah pendapatan maksimum pengelola parkir di tempat tersebut.
5. Jika vektor $\mathbf{a} = 3\mathbf{i} + 2\mathbf{j} - \mathbf{k}$, $\mathbf{b} = 2\mathbf{i} - 3\mathbf{j} + 4\mathbf{k}$ dan $\mathbf{c} = -\mathbf{i} + 2\mathbf{j} + 3\mathbf{k}$. Berapakah :

- $\mathbf{a} + \mathbf{b}$;

(b) $\mathbf{a} + \mathbf{b} + \mathbf{c}$;

(c) $\mathbf{a} \cdot \mathbf{c}$;

(d) $\mathbf{b} \times \mathbf{c}$.

Buatlah juga gambar dari vektor - vektor diatas.

6. Dengan memakai data soal no.3, berapakah :

(a) $|\mathbf{a}|$;

(b) $|\mathbf{a} + 2\mathbf{b} - \mathbf{c}|$;

(c) $|2\mathbf{a} - \mathbf{b} + \mathbf{c}|$.


Bab 8

Aplikasi pada Fisika

Dalam bab ini akan diperlihatkan bagaimana program Maple dapat digunakan untuk memecahkan beberapa persoalan Fisika. Keuntungan dari program Maple untuk aplikasi Fisika adalah: Maple dapat memvisualisasikan (bahkan dapat dianimasikan) hasil hitungan Fisika secara cepat dan mudah. Hal ini akan membuat persoalan Fisika menjadi hidup. Siswa dan guru dapat berinteraktif dengan soal - soal Fisika. Dalam persoalan gerak peluru (parabola) misalnya, seseorang dapat dengan mudah mengganti - ganti sudut elevasinya dan dapat dengan cepat melihat pengaruhnya dalam gerak tersebut. Pembahasan bab ini dibatasi hanya pada contoh - contoh problem pada Fisika mekanika klasik.

8.1 Kasus Gerak Lurus

Dua buah benda yang bergerak dengan arah yang berlawanan berada pada jarak 50 [m] satu sama lain. Benda pertama mempunyai kecepatan 20 [m/s] dan benda kedua mempunyai kecepatan 15 [m/s] (lihat gambar 8.1). Untuk menghindari benturan, kedua benda ini diperlambat laju mereka sebesar 3 [m/s²]. Apakah kedua benda ini akan bertumbukkan, bila ya; dimana dan dimana terjadi tumbukkan.



Gambar 8.1: Gerak dua buah benda yang saling mendekat.

Untuk menyelesaikan masalah ini dengan Maple, diperlukan langkah - langkah sebagai berikut :

1. Bersihkan memori Maple dengan mengetik **restart**

```
[> restart;
```

2. Definisikan persamaan jarak fungsi dari waktu $x(t) = x_0 + v_0 t + \frac{1}{2}at^2$. Dimana x_0 adalah kedudukan awal satu benda, v_0 adalah kecepatan awal, a adalah percepatan benda dan t adalah waktu. Dalam Maple persamaan tersebut ditulis,

```
[> x:=x0+v0*t+(1/2)*a*t^2;
```

$$x := x_0 + v_0 t + \frac{at^2}{2}$$

3. Masukkan kondisi awal untuk kedua benda kedalam persamaan gerak diatas, dengan perintah **subs**. x_A adalah untuk benda yang bergerak dari titik **A**, sedang x_B adalah untuk benda yang bergerak dari titik **B**. Untuk benda yang bergerak dari titik **A**

```
[> xA:=subs({x0=0,v0=20,a=-3},x);
```

$$x_A := 20t - \frac{3}{2}t^2,$$

Untuk titik benda yang bergerak dari titik **B**

```
[> xB:=subs({x0=50,v0=-15,a=+3},x);
```

$$x_B := 50 - 15t + \frac{3}{2}t^2$$

4. Untuk menentukan bilamana dan dimana mereka berbenturan, kita dapat memakai bantuan grafik. Dengan memplot kedua persamaan gerak benda tersebut secara bersamaan dalam satu grafik jarak x sebagai fungsi dari waktu t , maka waktu dan jarak kedua benda berbenturan akan diketahui.

```
[> plot([xA,xB],t=0..12);
```

Gambar 8.2 memperlihatkan titik potong kedua kurva, yang mana menginformasikan waktu dan tempat mereka bertumbukan. Gambar tersebut mempunyai dua titik potong, yang artinya bahwa kedua benda tersebut berbenturan dua kali.


5. Untuk mengetahui secara numerik, bilamana dan kapan mereka bertumbukkan. Maka perintah **solve** digunakan untuk memecahkan persamaan x_A dan x_B . Syarat kedua benda tersebut bertumbukan adalah $x_A = x_B$.

```
[> tberbentur:=solve(xA=xB,t);
```

Hasil yang didapat adalah :

$$t_{berbentur} := \frac{5}{3}, 10.$$

Mereka bertemu dua kali yakni pada saat $t = \frac{5}{3}$ dan $t = 10$ [s].



Gambar 8.2: Kasus kinematika dua buah benda bertemu.

6. Setelah diketahui waktu mereka berbenturan, maka bisa diketahui tempat mereka bertabrakan. Caranya adalah dengan mensubstitusi salah satu waktu tersebut kedalam persamaan jarak baik x_A maupun x_B .

```
[>subs(t=tberbentur[1],xA);
```

Perintah diatas, $t=tberbentur[1]$ mempunyai arti : yang dicari adalah jarak dan waktu berbenturan yang pertama kali. Hasil dari substitusi diatas adalah :

$$\frac{175}{6}.$$

Untuk menghitungnya dalam bentuk desimal, digunakan perintah **evalf**

```
[> evalf(subs(t=tbertemu[1],xA));
```

29.16666667

Jadi mereka berbenturan pertama kali setelah $\frac{5}{3}$ [s] bergerak pada jarak 29.2 [m].

8.2 Gerak Vertikal

Ada dua orang yang sedang memanjat tebing. Kedua orang ini ada pada ketinggian yang berbeda. Orang pertama berada pada ketinggian 40 [m] dari permukaan tanah, sedang orang kedua berada 5 [m] lebih tinggi dari orang pertama. Jika orang pertama melempar sepotong sandwich kepada orang kedua dengan kecepatan 11 [m/s] arah keatas. Berapakah ketinggian maksimum yang dapat dicapai sandwich tersebut. Apakah orang kedua dapat menangkap sandwich tersebut. Jika tidak hitunglah bilamana sandwich itu jatuh ke tanah.

1. Yang pertama kali dilakukan adalah menulis persamaan gerak vertikal dengan memakai orang pertama sebagai titik acuan. Persamaan tersebut dapat ditulis :

```
[>y:=y0+v0*t+(1/2)*a*t^2;
```

$$y := y_0 + v_0 t + \frac{1}{2} a t^2;$$

Dimana y_0 adalah tinggi awal, v_0 adalah kecepatan awal, a adalah percepatan, dalam hal ini gravitasi $g = 9.8[m/s^2]$, dan t adalah waktu.

2. Langkah kedua adalah mendefinisikan kecepatan sebagai :

```
[>v:=v0+a*t;
```

$$v := v_0 + at;$$

3. Kemudian substitusikan kondisi awal yang diketahui kedalam persamaan jarak dan persamaan kecepatan seperti $y_0 = 0, v_0 = 11, a = -9.8$.

```
[>y:=subs({y0=0,v0=11,a=-9.8},y);
```

$$y := 11t - 4.9t^2.$$


```
[>v:=subs({y0=0,v0=11,a=-98/10},v);
```

$$v := 11 - \frac{49t}{5}.$$

4. Dari sini terlihat hubungan antara jarak (y) terhadap waktu (t).

```
[>plot(y,t=0..3);
```

Plot y sebagai fungsi dari waktu, dengan batas $t = 0$ sampai dengan 3 [sec].



Gambar 8.3: Jarak ketinggian sandwich sebagai fungsi dari waktu.

5. Untuk mengetahui apakah sandwich tersebut bisa digapai oleh orang kedua yang berjarak 5 [m] di atasnya, maka kita pakai special kondisi, yakni $y = 5$ [m]. Jika kondisi ini kita namakan *kond*, dengan maple kita dapat melihat bilamana sandwich itu mencapai titik ketinggian 5 [m].

```
[>kond:=solve(y=5,t);
```

```
kond := 0.6330784160, 1.611819543.
```

Terlihat bahwa sandwich bisa dua kali untuk mencapai titik 5 [m]. Yakni pada $t = 0.63$ dan $t = 1.611$. Artinya orang kedua dapat menggapai sandwich tersebut. Fenomena ini dapat dilihat Dari gambar 8.3.

6. Untuk mengetahui berapa kecepatan sandwich tersebut saat mencapai titik 5 [m], maka kita substitusikan harga *kond* kedalam persamaan kecepatan. Dalam hal ini kita memakai salah satu harga t yakni *kond*[1].

```
[> subs(t=kond[1],v);
```

```
4.795831523.
```

7. Sedang untuk menghitung berapa titik tertinggi yang dapat dicapai oleh sandwich tersebut, kita hitung harga t pada saat $v = 0$. Kita beri nama variabel t untuk kondisi ini adalah *ttinggi*.

```
[>ttinggi:=evalf(solve(v=0,t));
```

$$ttinggi := 1.122448980$$

Sandwich akan mencapai titik tertinggi saat $t = 1.122$ [s].

8. Untuk menghitung titik tertinggi yang dapat dicapai sandwich, adalah dengan mensubstitusikan harga $ttinggi$ kedalam persamaan jarak (y).

```
[> ytinggi:=subs(t=ttop,y);
```

$$ytinggi := 6.173469386$$

Titik tertinggi yang dapat dicapai oleh sandwich tersebut adalah 6.173 [m] dari orang pertama. Katakanlah orang kedua gagal menangkap sandwich tersebut karena satu dan lain hal. Akibatnya sandwich tersebut jatuh ke tanah (40 [m] dibawah orang pertama). Bilamana sandwich itu jatuh ketanah.

9. Untuk mengetahui kapan sandwich tersebut jatuh ketanah, maka syarat yang diperlukan adalah persamaan jarak, y harus -40 [m]. Harga ini kita substitusikan ke dalam persamaan jarak.

```
[>solve(y=-40,t);
```

$$-1.947267134, 4.192165093$$

Terdapat ada dua buah harga t , tentu kita akan mengambil yang positif yakni $t = 4.19$. Artinya sandwich akan jatuh ketanah setelah ≈ 4.2 [sec] setelah dilempar orang pertama.

8.3 Gerak Parabola

Sebuah gerak parabola mempunyai sudut elevasi 30° . Gerak tersebut mempunyai kecepatan awal 20 [m/s] , lihat gambar 8.4. Animasikan gerak parabola tersebut. Berapa titik tertinggi yang dapat dicapainya, bilamana dan dimana benda itu jatuh kembali ke bumi.



Gambar 8.4: Gerak benda dalam dua dimensi.

Seperti diketahui persamaan gerak parabolik dapat diuraikan menjadi dua gerak yang diproyeksikan pada sumbu vertikal (y) dan horisontal (x). Gerak pada sumbu vertikal adalah gerak lurus berubah beraturan dengan gravitasi (g) sebagai percepatannya, sedang gerak pada sumbu horisontal adalah gerak lurus beraturan. Persamaan gerak untuk sumbu x dan y dapat dijabarkan sebagai berikut:

$$\begin{aligned}x &= v_o \cos(\alpha)t \\y &= v_o \sin(\alpha)t - \frac{1}{2}gt^2.\end{aligned}$$

Dimana v_o adalah kecepatan awal $[\text{m/s}]$, α adalah sudut elevasi, dan t adalah waktu $[\text{s}]$. Komponen kecepatan pada gerak parabola pun dibagi dua komponen yakni komponen pada sumbu horisontal v_x dan komponen sumbu vertikal v_y . Dalam kalkulus diketahui bahwa kecepatan adalah turunan pertama jarak terhadap waktu, jadi $v_x = \dot{x}$ dan $v_y = \dot{y}$.


Dengan menggunakan Maple gerak parabola diatas dapat diselesaikan sebagai berikut :

1. Seperti biasa, bersihkan dahulu memory yang ada pada lembar kerja kita dengan perintah **restart**. Kemudian definisikan, kondisi awal seperti $v_o = 20 \text{ [m/s]}$ dan gravitasi $g = 9.8 \text{ [m/s}^2\text{]}$. Lalu definisikan persamaan gerak untuk sumbu horisontal dan vertikal.

```
[>restart:  
[>v0:=20:  
[>g:= 9.8:  
[>x:= v0*cos(Pi/6)*t:  
[>y:=v0*sin(Pi/6)*t-1/2*g*t^2:
```

- Lintasan gerak tersebut dapat dilihat dengan memplot persamaan gerak sumbu vertikal terhadap sumbu horisontal. Gunakan perintah plot untuk memvisualkan persamaan tersebut dalam range waktu $t = 0 - 3$ [s]. Agar lebih fokus pada lintasan gerak peluru tersebut, maka sumbu cartesian yang dipakai dibatasi, yakni untuk sumbu x mempunyai ranngge dari $0 - 40$ dan sumbu y antara $0 - 6$. Hasilnya dapat dilihat pada gambar 8.5.


```
[> plot([x,y,t=0..3],0..40,0..6,color=black,thickness=3);
```



Gambar 8.5: Lintasan gerak parabola untuk $t = 0 - 3$ [s].

- Untuk menganimasikan lintasan peluru tersebut, perintah pada Maple `animatecurve` di-aplikasikan. Hasil dari animasi ini dapat dilihat pada gambar 8.6.


```
[>animatecurve([x(t),y(t),t=0..3],view=[0..40,0..6]);
```



Gambar 8.6: Animasi dari satu gerak peluru, panel (a) gambar lintasan saat $t = 0.5$ [s], panel (b) menunjukkan listasan saat $t = 1.5$ [s]. Sedang panel (c) menunjukkan lintasan saat $t = 2.25$ [s].

8.4 Hukum Newton kedua

Sebuah benda meluncur (lihat gambar 8.7) dari satu bidang miring kasar. Bila sudut miring bidang tersebut $\theta = 36.87^\circ$, massa benda tersebut $m = 65$ [kg], harga gravitasi $g = 9.8$ [m/s²], dan gaya friksi adalah 75 [N]. Hitunglah percepatan dan gaya normal yang bekerja pada benda itu.



Gambar 8.7: Benda yang sedang meluncur dari satu bidang miring kasar.

Pada gambar 8.7 mengilustrasikan komponen gaya - gaya yang bekerja pada benda tersebut. Gaya - gaya tersebut dapat diuraikan dalam komponen x sejajar dengan arah bidang miring, dan sumbu y tegak lurus dengan bidang miring tersebut.

1. Pertama uraikan gaya - gaya yang bekerja pada sumbu x

$$ma_x = mg \sin(\theta) - F_{fric},$$

dan gaya - gaya yang bekerja pada sumbu y

$$ma_y = N - mg \cos(\theta).$$

Dimana m adalah massa benda, a_x dan a_y adalah percepatan dalam arah sumbu x dan y , g adalah untuk gravitasi, dengan harga 9.8. F_{fric} adalah gaya friksi yang bekerja pada sumbu x , dan N adalah gaya normal. Definisikan komponen - komponen gaya ini kedalam Maple,

```
[>restart:  
[> x_eq:=m*g*sin(theta)-Ffric=m*ax:
```

$$x_{eq} := mg \sin \theta - F_{fric} = max;$$

```
[> y_eq:=N - m*g*cos(theta)=m*ay:
```

$$y_{eq} := N - mg \sin \theta = may;$$

2. Untuk mencari ax percepatan arah sumbu x dan gaya normalnya N , persamaan x_{eq} dan y_{eq} dieliminasi dengan memakai perintah `solve`.

```
[>jawab :=solve({x_eq,y_eq}, {ax,N});
```

$$jawab := \{N = mg \cos(\theta) + may, ax = \frac{(mg \sin(\theta) - F_{fric})}{m}\};$$

3. Jawaban diatas berupa simbolik, terlihat ada dua jawaban yaitu N dan ax . Untuk mendapatkan jawaban numerik nilai N dan ax , substitusikan variabel - variabel yang kita ketahui seperti ; $m, g, \theta, F_{fric}, ay$. Karena benda bergerak searah sumbu x dan tidak pada sumbu y , maka $ay = 0$

```
[>evalf(subs({theta=36.87*3.1415/180,m=65,g=9.8,ay=0,Ffric=75},jawab));
```


hasil dari substitusi tersebut adalah :

$$\{N = 509.6065707, ax = 4.726019059\}$$

Jadi percepatan benda itu arah sumbu $x = 4.73 \text{ [m/s}^2\text{]}$ dan gaya normal yang bekerja pada benda itu adalah 510 [N] .

8.5 Katrol tetap

Sebuah katrol tetap tergantung pada satu titik dengan membawa beban A dan B , lihat gambar 8.8. Jika massa A besarnya $3[\text{kg}]$ dan massa $B = 5 [\text{kg}]$. Hitung percepatan sistem tersebut, dan berapa besar tegangan tali T . gravitasi $g = 9.8 [\text{m/s}^2]$.



Gambar 8.8: Sebuah katrol tetap dibebani dua benda A dan B .

Untuk menyelesaikan soal di atas dengan menggunakan Maple, kita lakukan langkah - langkah sebagai berikut:

1. Definisikan kesetimbangan pada massa A (m_A),

$$T - m_A \cdot g = m_A \cdot a_A y.$$

Dimana $a_A y$ adalah percepatan benda A arah y .

2. Kesetimbangan untuk massa B (m_B),

$$T - m_B \cdot g = m_B \cdot a_B y.$$

Dimana $a_B y$ adalah percepatan benda B arah y .

3. Karena massa benda A dan massa benda B terhubung oleh satu tali maka percepatan massa A arah y harus sama dengan percepatan massa B untuk arah y .
4. Pada Maple kita tulis :

```
[> A_pers := T - mA * g = mA * aAy;
```

$$A_pers := T - m_A \cdot g = m_A \cdot a_A y$$

```
[> B_pers := T - mB * g = mB * aBy;
```

$$B_pers := T - mB.g = mB.aBy$$

5. Karena percepatan massa benda A dan massa benda B adalah sama, (kita definisikan sebagai a). Harga a ini kita bisa substitusikan kedalam A_pers dan B_pers . Percepatan untuk massa benda A a_Ay adalah a dan untuk massa benda B a_By adalah $-a$. Tanda minus menyatakan arah gerak benda A berlawanan dengan arah gerak benda B . Cara mensubstitusikan harga percepatan ke dalam persamaan massa benda A dan B adalah;

```
[>dua_pers:=subs({aAy=a,aBy=-a},{A_eqn,B_eqn});
```

$$dua_pers := \{T - mA.g = mA.a, T - mB.g = -mB.a\}$$

kita menamakan fungsi yang telah tersubstitusi dan memuat dua persamaan massa benda A dan B sebagai dua_pers .

6. Untuk mendapatkan harga a dan T pada kedua persamaan tersebut, dengan Maple kita bisa melakukannya dengan perintah **solve**. Seperti yang kita lihat dibawah ini :

```
[>jawab :=solve(two_eqns, {T,a});
```

$$jawab := \left\{ a = -\frac{g.(mA - mB)}{mA + mB}, T = \frac{2.mA.g.mB}{mA + mB} \right\}$$

7. Sekarang kita dapat melihat jika harga a dan T hanya tergantung dari massa benda dan gravitasi g . Kita substitusikan harga - harga ini kedalam persamaan diatas, kemudian kita hitung nilai a dan T dengan perintah **evalf**.

```
[>evalf(subs({mA=3,mB=5,g=9.8},jawab));
```

$$\{a = 2.450000000, T = 36.75000000\}$$

8. Jadi percepatan sistem a adalah $2.45 \text{ [m/s}^2\text{]}$ dan tegangan tali T adalah 36.8 [N]

Bibliografi

- [1] Abell. Martha.L, Braselton. James.P., *Maple by Example*, Elsevier Academic Press,(2005).
- [2] Tampomas. Husein., *Matematika jilid 1 untuk SMA kelas X*, Penerbit Airlangga, (2006).
- [3] Soemartojo. Noeniek., Prof.Dra., *Matematika SMA jilid 3*, Penerbit Erlangga, (1990).
- [4] Alders.C.J., *Ilmu Aljabar*, Pradnya Paramita, (1976).
- [5] Don. Eugene., Ph.D, Don. Benay., M.S, *How to Solve Word Problems in Calculus*, McGraw Hill, (2001).
- [6] Wirodikromo. Sartono, Wandiyagiri. Dedi. D, *Matematika untuk SMA 3*, Penerbit Erlangga, (1990).
- [7] Stroud, K.A., *Engineering Mathematics*, The Macmilan Press Ltd, 1987.